U. S. DEPARTMENT OF THE INTERIOR

GEOLOGICAL SURVEY

BASIC programs for calculation of cation site occupancies and plotting of mineral data with applications for evaluation of metamorphic mineral assemblages

by

Marta J.K. Flohr

Open-File Report 83-905

1983

CONTENTS

CONTENTS (continued)

INTRODUCTION

Collection of large amounts of electron microprobe mineral analysis data from related samples neccessitated development of computer software to process these data efficiently. A system of computer programs was developed for calculation of formal cation site occupancies for several mineral groups (amphiboles, micas, garnets, feldspars, chlorites and cordierite) and for plotting mineral data. Listings of the computer programs and examples of output are included to illustrate specific applications. This system of programs was initially developed to facilitate the study of metamorphic rock assemblages. For example, compositional changes of a particular mineral as a function of metamorphic grade can be visually traced on a cation plot. On a larger scale, the changes in composition of minerals within an assamblage can be seen, and possible reaction relationships defined, on an ACFM diagram, a plot commonly used in metamorphic petrology. (These various plotting options and others will be discussed more fully below.) The programs are designed to be flexible so that they can be modified to suit the needs of other users.

The system of computer programs described herein was developed using the facilities available at the U. S. Geological Survey Reston Microprobe Facility. The microprobe is a SEMQ (Scanning Electron Microprobe Quantometer) manufactured by Applied Research Corporation (ARL). Raw count data are reduced to weight percent values on-line and may be selectively stored, at the user's option, on a floppy disk to be accessed at a later time (see McGee, 1983 for a detailed description of the data reduction and storage program used). All of the SEMQ software is written in ARL Extended Basic (ARLEB) and data are initially stored on an ARLEB-formatted disk (McGee, 1983). The programs described here are run on a Digital Equipment Corporation (DEC) PDP-11/23 microcomputer with 128 kbytes memory and a RX-02 floppy disk drive using MINC (Modular INstrument Computer) BASIC, a software package that incorporates a BASIC language interpreter and an operating system that is similar to the DEC RT-11 system with single-job monitor. Both the ARLEB system and the MINC system use the same hardware which facilitates transfer of the microprobe data. The data are transfered from the user's ARLEB disk to a MINC formatted disk by means of an interfacing program, RDARL4, described by Huebner (1983). RDARL4 contains several different data file formatting options; the one used in many of the programs described here is option 7, BASIC data file for silicate mineral recalculation. Data are plotted on a Bausch and Lomb, Houston Instruments (HI) DMP-7 plotter which uses high-level language.

Extra workspace is required when using the MINC system to run three of the programs described in this report - AMPHI, MICA and GARN. When the extra space command is executed, MINC adds 2048 words to the workspace. Additional workspace may not be necessary on other systems. Program GARN contains three overlay segments. The use of overlays could be eliminated or the number of overlays needed reduced on another system; lines would have to be renumbered sequentially and the "overlay" commands eliminated.

DESCRIPTION OF DATA FILE FORMAT

An example of the BASIC data file format generated by RDARL4 and used as input for most of the programs described in this report is given in Appendix I. The first record of the data file contains a numeric flag, originally generated by RDARL4 to indicate oxide or element weight percent data. Only oxide weight percent values can be used in the programs described in this report, so the value of the flag must be 1. Lines 2 through 19 are a list of the 18 elements contained in the file, in the order in which they appear in the following analyses. Two lines preceed each analysis and provide information about it. The first of these lines contains the analysis number, which is assigned to the analysis when it is initially stored on the user's ARLEB disk, and the analysis title or label, which is provided by the user also at the time the analysis is obtained and stored. The second of these two lines contains the name of the ARLEB file of analytical parameters used to collect the data (*HBBT in the example), followed by the number of oxygens used for stoichiometric cation calculation. The next 18 lines of each analysis contain pairs of oxide weight percent and calculated cations (obtained from the ARLEB file) for each of the 18 elements listed at the beginning of the file. A maximum of 14 elements can be obtained from a probe analysis and stored in an ARLEB file; as discussed below, 18 elements were chosen to cover a wide range of mineral groups.

Ferric iron, designated FC in the data file, is included even though the microprobe cannot distinguish between ferric and ferrous iron (all iron is reported as ferrous in microprobe analyses). The inclusion of Fe_2O_3 allows the user to construct files for data for which ferric iron is reported or calculated.

Since different minerals are often analyzed for different groups of elements, the 18 elements contained in the data file were chosen to include those elements most commonly analyzed in a wide range of silicate (and oxide) minerals. Data for elements not analyzed are reported as zeros by the RDARL4 program. Since the the name of the analysis package is given for each analysis there should be no confusion on the part of the user as to whether specific elements were analyzed or not.

This format is flexible in that it allows data from minerals analyzed by different analysis packages to be stored in one data file, i.e. all mineral data from one sample can be stored in one data file, helping to keep the number of files to a minimum. This is not a trivial matter when analyzing several different minerals in each of many samples. If analyses from one sample are collected during different microprobe shifts and are not stored in the same ARLEB data file, the data files generated by RDARL4 can easily be concatenated to create one "master data file" for that sample.

Since data from different mineral species can be stored in the same file, different analytical programs, each concerned with calculations for a specific mineral group, can read the same data file. Each mineral type can be identified by a two letter code which would be the first

two characters of the analysis label. Each program utilizes this identification system and has the ability to sort through the data file, accepting and processing only those analyses from a particular mineral group. For example, in the data file shown in Appendix I, the first analysis, an amphibole, is identified by the code "AM"; the second anlysis, a sphene, is identified by the code "SP". The amphibole site occupancy program, described below, searches for the string "AM". If an analysis is mislabelled when it is initially stored, the file created by RDARL4 can be corrected prior to running any of the recalculation programs, using a text editor.

It should be noted that users who wish to adopt the above described data file format, but whose data are not stored on disk, can create the data files by using a text editor. If the amount of data is small, it is less time consuming to enter the data directly in the format given in Appendix I. However, if a large volume of data is involved the following method of creating data files is suggested, for which an auxilary reformatting program written by the user will be needed. Create a data file containing, for each analysis, the two lines giving the analysis number, analysis label, analysis package name and number of oxygens for cation calculation, as described above. Follow these lines by the oxide weight percent for only those elements analyzed, in the order they appear in their original form (microprobe output, journal article, etc.). This data file can then be read by the auxiliary program which will calculate cations and reformat the data so that it is in the same format as that generated by RDARL4. This method helps to keep transcription errors to a minimum and the time spent writing the auxiliary reformatting program is small compared to the time spent reordering and entering large amounts of data by hand.

DESCRIPTION OF PROGRAMS

A brief description of the analytical programs developed and written for the MINC Basic system is presented in this section. Listings of the programs are included in Appendices II - X and XVIII - XIX. A flow chart (Fig. 1) provides a summary of what programs are currently available and what options they provide. Auxilary programs, such as those used to concatenate files or provide hard copies of the data files will not be described here because these operations are similar to those performed by standard RT-11 utilities. However, examples of such programs are available, upon request, from the author.

Programs for calculating cation site occupancies

Programs for calculating cation site occupancies are available for the following minerals (program names in parentheses): amphiboles (AMPHI), micas (MICA), chlorites (CHLOR), cordierite (CORD), feldspars (FELD) and garnets (GARN). The procedure used to calculate cations and oxygen equivalents of F and Cl is given in Deer, Howie and Zussman (1966). Each program generates a data table (examples are given in Appendicies II - VII) listing the oxide weight percents and calculated cations, formally partitioned into the appropriate sites for that mineral for each analysis. If the cations read in from the data file have been calculated on a different formula basis than the programs use, then in all programs, except GARN, the cations will be recalculated. The number of oxygens used for cation calculation is given below for each program. In programs AMPHI, MICA, CHLOR, CORD and FELD, the user has the option of calculating ferric iron by indicating what fraction of the total iron present in the analyses should be converted to ferric iron. In GARN ferric iron is calculated assuming ideal stoichiometry. All of these programs expect the input data file to be in the format illustrated in Appendix I.

Program AMPHI calculates cations on the basis of 23 oxygens for amphibole minerals and cation site occupancies following the method outlined by Leake (1978). If F and Cl are reported the oxygen equivalents of these anions are calculated and water is calculated by charge balance. A listing of the program is given in Appendix II, followed by an example of the data table it generates. An example of the input expected by AMPHI and used to generate the data table shown is also given. The input expected by the other site occupancy programs is similar and this example may be referenced for these program as well. Similar programs have been presented by Goff and Czamanske (1972) and J.H. Hammarstrom (oral communication, 1982) for calculating amphibole structural formulae.

Program MICA calculates cations and site occupancies for mica minerals on the basis of either 11 or 22 oxygens. Oxygen equivalents of F and Cl and water are calculated as in AMPHI. A listing of the program is given in Appendix III, followed by an example of the data table generated by MICA. J.H. Hammarstrom (oral communication, 1982) has developed a similar program for calculating structural formulae of mica minerals.

Program CHLOR, for chlorites, calculates cations on the basis of 28 oxygens and calculates a theoretical value for water from charge balance. A listing of the program and an example of the data table it produces are given in Appendix IV.

The program to calculate site occupancies for cordierite, CORD, is listed in Appendix V, along with an example of its data table. CORD uses 18 oxygens for cation calculations.

Program FELD gives the user the choice of calculating cations to either 8 or 32 oxygens. It also calculates mole percent of the feldspar end members $CaAl_2Si_2O_8$ (An), $NaAlSi_3O_8$ (Ab), $KAlSi_3O_8$ (Or) and $BaAl_2Si_2O_8$ (Cn). An example of the data table it generates and a listing of the program

are given in Appendix VI.

GARN, calculates cation site occupancies and end members for garnet minerals. It assumes that the cations read in have been calculated on the basis of 12 oxygens; the program will not recalculate cations if 12 was not used. Cations are normalized to a sum of 8. Ferric iron is then calculated, if warranted. Cations are then recalculated and the six most common garnet end members $Ca_3Cr_2Si_3O_{12}$ (Uvar), $Ca_3Fe_2^{+3}Si_3O_{12}$ (And), $Mg_3Al_2Si_3O_{12}$ (Pyr), $Mn_3Al_2Si_3O_{12}$ (Spes), $Ca_3Al_2Si_3O_{12}$ (Gros) and $Fe_3^{+2}Al_2Si_3O_{12}$ (Alm) are calculated from them, using the method and order of calculation presented by Rickwood (1968). Percent of the cations allocated to the six end members and the cation residuals are printed for each analysis, as shown in the example data table in Appendix VII. Analytical uncertainy may affect the achievement of charge balance, and adjustment of multivalent state cations may not be necessary. Since silica is the most abundant oxide in garnets, the amount of silica necessary to achieve ideal stoichiometry is given for each analysis (a detailed discussion of redox calculations and user evaluation of such recalculated data is given by Freeborn et al., unpub. data). This information can be used to evaluate the data. A listing of the program which consists of a main program, GARN, and three overlay segments, GARN2, GARN3 and GARN4, is given in Appendix VII.

Program structure

All of the cation site occupancy programs described above follow the same general format. Using AMPHI as an example, the general program structure will be reviewed. User input is described in the next section of this report. The programs consist of segments, each segment concerned with a specific calculation or operation. Sequentially, these are (line numbers as used in AMPHI, Appendix II):

lines 120-385: dimension arrays; fill arrays containing row labels (A$), formula weights of oxides (A), number of anions per oxide formula (B), number of cations per oxide formula (C).

lines 390-535: all user input is requested; files and line printer are opened if requested by the user.

lines 537-653: data file (defined as File#1, line 510) is read in. If the sort option is being used (user provided sort string, M2$, line 450) a test is made (line 570) to check if the first 2 characters, M8$, of the title, T$, are the same as M2$; if yes, the analysis is accepted and the oxide weight percent values are stored in array D; if no, the analysis is read in but the data are not stored. When the data table is printed, a maximum of 10 analyses fit on one computer page (print-header must be set at the top of the page -- reminder given in line 485). Therefore, only 10 analyses are accepted for processing at a time (counter N2). When the end-of-file is reached, counter C4 is set to 1.

lines 660-690: oxide weight percent values used in AMPHI are reordered and stored in array Y.

lines 720-850: Fe_2O_3 is calculated using Fe^{+3}/Fe total value (B7) supplied by the user in line 400; cations are calculated on the basis of 23 oxygens; oxygen equivalents of Cl and F, Y(16,J) Y(17,J) and a theoretical value of water, Y(19,J), are calculated.

lines 970: program branches to subroutine where formal cation site occupancies are calculated (GOSUB 1590).

lines 980-1040: site occupancy cations are written to output disk (File#4) if user responded yes (R3$=Y) to prompt in lines 455-460.

lines 1069-1520: data table is printed at line printer if user responded yes (Q2$=Y) to prompt in line 480; program branches to subroutine to choose row lables from array A$ (GOSUB 1930).

lines 1530-1580: test is made (line 1530) for end-of-file (C4=1?); if yes, files are closed and the program terminates; if no the program loops back to line 539 and, if the data table is being printed, a form-feed is executed (Print#2, Z$), counter N2 is reset to 0, more analyses are read and the above-described calculations are repeated.

lines 1590-1915: subroutine to calculate formal site occupancies.

lines 1930-2070: subroutine to choose row labels from A$.

User input

The following prompts, given below in capital letters, request user input. In general all of the cation site occupancy programs issue the same prompts and follow the same format for user input (see example, Appendix II).

ENTER RUN TITLE: This title is printed at the top of the data table. If no title is desired, input a carriage return.

ENTER FE+3/FE TOTAL: For example, 0.25 indicates that 25% of the total iron should be converted to ferric iron. Enter 0 (zero) if no adjustment is desired; all iron is then assumed to be ferrous. If an analysis is read in which contains ferric iron this variable is ignored by the program. This prompt is not used in GARN.

ENTER NAME OF DATA FILE AS SYn:XXXXXX.YYY: SY1:S6848.DAT is an example of the format expected.

ARE YOU SORTING? (Y OR N): The user has the option of sorting specific analyses from the data files by using a two character sort string as the first two characters of the analysis label. If the user responds no (N) to

this prompt, all analyses in the data file are accepted for processing. If he responds yes (Y), the following prompt is issued.

ENTER 2 CHARACTER SORT STRING: The user is now asked to input the sort string. In programs MICA and FELD the user can specify more than one sort string.

DO YOU WANT TO CREATE A DISK OUTPUT FILE CONTAINING SITE OCCUPANCY CATIONS? (Y OR N): The user has the option of creating a file containing the cations as they are calculated by the various programs. The prompt is given in AMPHI, MICA, CHLOR, and CORD. These files can then be used as the input files for the plotting programs PLAMPH, MICAPL and CHPLOT (PLCORD is in preparation) respectively, described later in this report. If the user responds yes to this question, the following prompt is issued.

ENTER NAME OF DATA FILE AS SYn:XXXXXX.YYY: The user is now asked to provide a name for the data file to be created.

GARN provides the option of creating two output files, one containing the end members it calculates and a second containing the recalculated oxides and cations. If the user chooses one or both of these options, he is asked to provide a name for the output file(s). These files can then be used as input files for the ternary plotting programs GAPLOT, which reads the end member file, and GRNPLT, which reads the oxide-cation file.

KEYPAD EDITOR ERROR CHECKING? (Y OR N): If the user has either edited his data file or created it using a text editor, there is the potential for format errors. If the user responds yes to this prompt, then each analysis number and title are printed on the video screen as the program reads the data file, whether or not an analysis is selected by the sort routine. Execution of the program will terminate if the program is unable to successfully read the file due to a format error, but the user will know at which analysis this occurred. He will then be able to correct the format error and reexecute the program.

DO YOU WANT DATA TABLE PRINTED? (Y OR N): The user has the option of obtaining a hardcopy of the data table (this prompt applies to all of the site occupancy programs except FELD, in which the data table is the only output). This option is included for the following reason. A user may first wish to generate the data table and evaluate the data, possibly deleting bad analyses, before generating a cation output file. The site occupancy program can then be run again to do this, without reprinting the data table.

In programs FELD and MICA the user is also asked to specify the number of oxygens for cation calculation (8 or 32 and 11 or 22, respectively).

Summary of HI plotter commands

Before the plotting programs are described, a brief summary of the plotter commands (which are specific to the HI plotter) will be given. This will allow potential users who have other systems to follow the sequence of plotting steps and adapt them to their own system. The HI plotter commands are as follows:

@ - deselects the plotter

;: - selects the plotter

U - pen up

D - pen down

H - send pen to "home" position (defined as the lower left-hand corner of the plotting surface)

A - indicates pen movement in "absolute" coordinates (that is, with respect to the origin (0,0), which can be the home position or any user-defined coordinate on the plotting surface)

R - indicates pen movement in "relative" coordinates (that is, with respect to the current pen position)

Ln - "L" indicates that a line is to be drawn; "n" is an integer, 0-9, which indicates line type, as follows:

————	line type 0	— — — —	line type 5
.....	1	—— —— ——	6
- - - - - -	2	— — - — - — -	7
— — — — — —	3	-- — -- — -- —	8
— — — — — —	4	- —— - —— - ——	9

Srhøsss... - "S" indicates characters are to be drawn and is used in the programs XYPLOT, TERPLT and TETPLT to label the axes.

"r" indicates the angle of rotation at which the characters are to be drawn:

r=1 for 0^{o} rotation r=2 for 90^{o} rotation
r=3 for 180^{o} rotation r=4 for 270^{o} rotation

"h" defines the height of the characters:

h=1 for 0.07" h=2 for 0.14" h=3 for 0.28"
h=4 for 0.56" h=5 for 1.12"

"ϕ" is a space or comma

"sss..." is the string of characters to be drawn

"_" the underscore indicates the end of the character string and exits the routine

Mhm - "M" indicates that a marker (or plotting) symbol is to be drawn

"h" is defined as above for drawing characters

"m" is the code for the symbol to be drawn:

m=0 for +	m=1 for ×	m=2 for □
m=3 for o	m=4 for △	m=5 for ⧖

When using the MINC system all plotter commands must be converted to strings before being sent to the plotter. For example, in line 520 of program XYPLOT (Appendix VIII), the command to raise the pen, "U", is converted to "U$" (U$='U'). Strings may be added together to form one longer string consisting of a sequence of commands.

Plotter commands are sent to the plotter using the "COUT" command. Its format is: COUT(mode,string-name,string-length,channel number). In all of the plotting routines described in this report, the standard mode is used. This is the default mode and allows program execution to continue while transfer to the plotter continues. String-length refers to the number of characters in the string; 255 is the maximum allowed by MINC. Channel 0 is used in the plotting programs as the serial channel for carrying output characters. For example, in XYPLOT, line 520, the commands to deselect the plotter, "@", and select the plotter, ";:", are converted to the string "A$" (A$='@;:'). In line 590, A$, consisting of 3 characters, is sent to the plotter, using the default mode, via Channel 0: COUT(,A$,3,0).

Programs for drawing and labelling axes

XYPLOT: Program XYPLOT, modelled after the ARLEB program "IPLOT" (McGee, 1983), draws and labels a Cartesian coordinate system. A listing of the program is given in Appendix VIII. The plot is 8 by 12 inches. User input consists of labels for the X and Y axes, a title for the plot and the ranges for both axes.

TERPLT: Program TERPLT, also modelled after the program "IPLOT" (McGee, 1983) draws and labels a ternary diagram. The triangle is 10 inches on a side. A listing of the program is given in Appendix IX. User input consists of labels for the three apices and a title for the plot.

TETPLT: Program TETPLT draws and labels a tetrahedron. A listing of the

program is given in Appendix X. User input consists of labels for the four apices and a title for the plot. In addition, the user must supply values for four parameters which control the orientation and size of the tetrahedron. These are: angle alpha, in degrees, which is the angle of rotation about the z-axis; angle theta, in degrees, which is the angle of rotation about the x-axis; parameter E, which is the perspective point for viewing the tetrahedron and controls the amount of foreshortening; and scaling factor S, which controls the size of the tetrahedron. The following initial values are suggested for a new user: $\alpha = -65^{o}$, $\Theta = 80^{o}$, E = 30, and S = 1200. For a more detailed explanation of the geometric projections used in tetrahedral plotting the reader is referred to Spear (1980, 1982).

Programs for plotting mineral data

The site occupancy programs AMPHI, MICA, CHLOR and CORD, at the user's option, generate output files containing the calculated site occupancy cations. These files can then be read, respectively, by the plotting programs PLAMPH, MICAPL and CHPLOT (PLCORD, in prep.) which are described below. Program GARN can generate two output files, one containg recalculated oxides and cations which can be read by the plotting program GRNPLT, and a second containing the calculated end members which can be read by the plotting program GAPLOT. Listing of the programs are given in Appendices XI - XV.

Format of plot files generated by site occupancy programs

Each plotting file has a different format, but the general file structures are the same. A copy of a plotting file generated by AMPHI is shown in Appendix XVI and will serve as an example. The first line of each file contains the name of the file originally read by the site occupancy program which generated it. The first line of each analysis contains the analysis number. This line in MICAPL also contains the sort string for that analysis. This is done so that the user may plot different mica minerals from the same sample using different symbols, again by sorting on the string. If the user did not sort when he ran MICA, a dummy sort string, MI, is substituted. This line is followed by the cation and anion data (for amphiboles, micas, chlorites and cordierite), one value per line. The order of the data is the same as reported in the corresponding data table and is given in Table 1.

Program structure - PLAMPH, MICAPL AND CHPLOT

PLAMPH, MICAPL and CHPLOT are used for plotting cations in X-Y space (ternary and tetrahedral plotting routines could be added relatively easily). The three programs are similar in structure; using PLAMPH as an example, this structure will be reviewed briefly below. Plotter commands have been defined in a preceeding section of this report and will not be discussed here. Line numbers are as used in PLAMPH (Appendix XI).

lines 130-178: plotter commands are converted to strings; line printer is accessed as File#2; Q4, counter for printing of plotting parameters and columm headings, is set equal to 0.

lines 180-280: user input is requested - name of data file, N$ and plotter symbol code, S; N$ is opened as File#1; the first record of the data file is read (A9$, the name of the data file initially read by AMPHI, from which the PLAMPH file was created); Z2, counter for relative plotting is set equal to 0 (the first data point is plotted in absolute coordinates; after it is plotted, Z2 is incremented by 1 (line 955) and subsequent points are plotted in relative coordinates, allowing for more rapid plotting).

lines 290-570: plotting options are chosen (parameters X and Y); the options are printed both on the video screen (lines 290-310) so that the user may choose the parameters of interest and at the line printer (lines 460-570) for later reference; user must input the minimum and maximum (i.e. ranges) for X (P3 and P4) and Y (P5 and P6); after the plotting options are chosen, Q4 is set to 1, so that if a second data file is read after data from the first file are plotted, this part (lines 290-570) of the program is skipped and execution continues at line 565.

lines 580-855: analyses are read from the file (and plotted) one at a time, with cation values being stored in array D; values for the plotting parameters are calculated from the cation values in lines 610-660; P1 and P2 are the X and Y coordinates and are defined in lines 710-855.

lines 900-975: the position of the point on the plot (P7,P8) is calculated and the point is plotted; the relative coordinate is defined in line 960.

lines 977-1010: if Z2=0 (for first analysis only), the column headings are printed; subsequently only the analysis# (N2) and coordinates are printed.

line 1030: program loops back to line 580; a check is made for the end-of-file; if negative, the next analysis is read in and the above described calculations are repeated.

lines 4000-5000: if the end-of-file is reached (line 580), File#1 is closed and the plotter pen returns to the home position; the user now has the option of plotting data from another file on the same graph; if he chooses not to plot (R$=N, line 4015), File#2 is closed and the program terminates; if R$=Y, the program loops back to line 180 and a new data-file name is requested.

Plotting options

In order to evaluate different parameters for different mineral species, each program contains different plotting options. Other options may be easily added. For X-Y plots the user specifies which parameter is to be plotted on each axis. An example of a plot and its accompanying printout, generated by PLAMPH, is shown in Appendix XVII. The input used to generate this plot is also given. The other plotting programs follow the same general format for user input. The plotting options are listed at the beginning of the printout with the two parameters chosen for that plot indicated. In the example AlIV + Fe+3 + Cr (option 6) and Na in A (option 13) were plotted. The data files accessed are listed for reference. The code for the plotting symbol is given for each group of analyses plotted. The coordinates for each analysis are then listed. Note that the first ten analyses from file SY1:S6954.DAT are the same as those in the data table in Appendix II. Output from the other plotting programs follows the same format.

PLAMPH - current X-Y plotting options:

AlIV + Fe+3 + 2Ti	Ca + Na in M4	AlIV	Mn
(Fe + Mn)/(Fe+2 + Mn + Mg)	100Na/(Na + Ca)	AlVI	Na in M4
AlVI + Fe+3 + Ti + Cr	Na in A + K	Ti	K
Mg/(Mg + Fe+2)	Na in A		

MICAPL - current X-Y plotting options:

AlVI	Fe+2 + Mg + Ti + (Si-3)	AlIV + AlVI + Fe+3
AlIV	(Fe+2 + Mn)/(Fe+2 + Mn + Mg)	Ti

CHPLOT - current X-Y plotting options:

AlIV	Mg/(Mg + Fe+2)	Total Al	Si
AlVI	Mn + Fe + Mg	Mn	

GAPLOT - current ternary plotting options:

Uvar	And	Pyr	Spes
Gros	Alm	Alm + Pyr	

GRNPLT - current ternary plotting options:

AFM	ACFm	AKFm	Mn-Fe-Mg

Molecular proportions are calculated and plotted for the first three options. Parameters are defined as follows:

AFM: $A=Al_2O_3+Fe_2O_3-K_2O-Na_2O$ (top component)
F=FeO+MnO (left component)
M=MgO (right component)

ACFm: A is the same as defined for AFM plots.
C=CaO (left component)
Fm=FeO+MnO+MgO (right compont)

AKFm: A is the same as defined for AFM and ACFm plots.
$K=K_2O$ (left component)
Fm is the same as defined for ACFm plots.

The fourth option plots the cations Mn (as the top component), Fe (as the left component) and Mg (as the right component).

User input

The following prompts, given below in capital letters, are issued by the programs described above.

ENTER NAME OF DATA FILE AS SYn:XXXXXX.YYY:

ENTER CODE FOR PLOTTER SYMBOL: 0,1,2,3,4 or 5. The symbols which correspond to each symbol code are given in the discussion of plotter commands, above.

CHOOSE X,Y: The parameters which can be plotted for each program are listed on the video screen, each with a numeric code. Here the user is asked to input the code for the parameter he wants plotted on each axis.

INPUT X MIN, MAX: and INPUT Y MIN, MAX: The user must supply the ranges for the X and Y axes.

In program MICAPL, the user may sort his analyses using the two character sort string as discussed above. User input for ternary plotting program GAPLOT is essentially the same, with the user indicating which three parameters he wants plotted. In GRNPLT, the user specifies which plot he wants - AFM, ACFm, AKFm or Mn-Fe-Mg.

Programs for plotting mineral assemblages

Two other plotting programs, PLDATA and TETPT, have also been developed. Both programs expect data to be input in the format specified in Appendix I,

i.e. the same format as is used for the site occupancy programs. The programs can therefore read the same files as the site occupancy programs.

Program PLDATA has several options for plotting data in ternary space. Molecular proportions are calculated for three options - AFM, ACFm and AKFm. Parameters are defined and positions of points are calculated as in GRNPLT described above. These options are useful for plotting metamorphic mineral assemblages. Two other options allow the user to plot pyroxene and feldspar end-members. For pyroxenes, mole percent of the end members $CaSiO_3$ (Wo), $MgSiO_3$ (En) and $FeSiO_3$ (Fs) are calculated, with no adjustments made for ferric iron or other pyroxene components. Positions of the points are calculated assuming that Wo is the top component, En the left component and Fs the right component. Similarly, feldspar end members (mole percent) are calculated and plotted using (Or + Cn) as the top component, Ab as the left component and An as the right component. A listing of the program is given in Appendix XVIII. An example of the output, an AFM plot and its accompanying printout is also included. The name of each data file accessed is printed. If the user is sorting, then the sort string, AM in the first group of analyses in the example, is printed. The number on the same line as the sort string is the plotter symbol code used for that group of analyses. After the amphibole data were plotted biotites (BI) and cordierites (CD) were plotted from the same data file. If the sort option had not been used then all analyses in the data file would have been plotted. (The plot shown in Appendix XVIII was drawn and labelled using the TERPLT, described above, and the points were plotted using the PLDATA.) Analogous output is produced using the other options.

TETPT calculates the position of points within a tetrahedron and plots them. The program was written using the algorithm presented by Spear (1980). A listing of the program is given in Appendix XIX. The program currrently has options for plotting molecular proportions calculated from the input oxide weight percent values on three diagrams: ACFM, AKFM and ACFmK, which are also useful diagrams for plotting mineral assemblages. The example given in Appendix XIX was generated using the ACFM option. As in PLDATA, TETPT generates an accompanying list of the coordinates for each point plotted. In the example, amphiboles (AM), biotites (BI), pyroxenes (PX) and plagioclase feldspars (PL) from the same data sample (and data file) were plotted, each with a different symbol. The code for the plotter symbol is printed on the same line as the sort string. (The plot shown in the example was drawn and labelled using the program TETPLT and the points were plotted using TETPT. Alpha = -65^{o}, theta = 80^{o}, E = 30 and S = 1200 were used; note that the figure in Appendix XIX is a photo reduction.) Components used in the plots and their positions on the diagrams are defined as follows.

ACFM: A=Al_2O_3+Fe_2O_3-K_2O-Na_2O (top component)
C=CaO (left component)
F=FeO+MnO (bottom component)
M=MgO (right component)

AKFM: A, F and M are defined as for ACFM plots.
K=K_2O (left component)

ACFmK: A and C are defined as for ACFM plots.
K=K_2O (right component)
Fm=FeO+MnO+MgO (bottom component)

User input

PLDATA and TETPT request similar user input. The prompts are given below in capital letters.

CHOOSE PLOT: The user is asked to specify the plotting option desired.

ENTER NAME OF DATA FILE AS SYn:XXXXXX.YYY:

KEYPAD EDITOR ERROR CHECKING? (Y OR N): This prompt is explained in the section describing the site occupancy programs and will not be repeated here.

ARE YOU SORTING? (Y OR N): This prompt is also explained in the preceeding section. As in the other programs, if the user chooses to use this option, he is asked to provide the two character sort string.

CHOOSE CODE FOR PLOTTING SYMBOL 0,1,2,3,4 or 5: This prompt is explained above.

After the analyses from the first data file are plotted the user has the option, if he has plotted only one mineral type from the data file (i.e. he has used the sort option), to plot additional mineral data from the same file. If the user responds yes to the prompt 'DO YOU WANT TO PLOT MORE DATA FROM THIS FILE ON THIS GRAPH? (Y OR N)' he is asked to input the code for the plotting symbol desired and a new sort string. The user may plot as many different minerals on one diagram as he desires. If the user did not use the sort option (i.e. he plotted all data from one data file) or chooses not to plot additional data from the same data file, he now has the option of reading a new data file and plotting data from it. The user may do this by responding yes to the prompt 'DO YOU WANT TO PLOT DATA FROM ANOTHER FILE ON THIS GRAPH? (Y OR N)'. He is then asked to provide the name of the new file, a sort string if he is sorting and a plotting symbol code. The user may read in as many data files as he wishes. A 'no' response to this prompt terminates the program.

Program TETPT also requests the four parameters alpha, theta, E and S which control the size and orientation of the tetrahedron and allow the program to calculate the positions of the points within the tetrahedron. These parameters are defined in the discussion of the program TETPLT.

Acknowledgements

D.B. Stewart of the U.S.G.S. (Reston) suggested the development of the package of programs described in this report and provided useful suggestions and support during its evolution.

REFERENCES

Deer, W.A., Howie, R.A. and Zussman, J. (1966). An Introduction to the Rock Forming Minerals. New York: John Wiley and Sons. 528 pages.

Huebner, J.S. (1983). RDARL4, a FORTRAN interface for transferring chemical data from an Applied Research Laboratories electron microprobe to a PDP-11 computing system. U.S. Geological Survey Open File Report 83-713, 40 pages.

Goff, F.E. and Czamanske, G.K. (1972). Calculation of amphibole structural formulae. U.S. Geological Computer Contribution 16.

Leake, B.E. (1978). Nomenclature of amphiboles. Mineral. Mag. 42, 533-563.

McGee, J.J. (1983). $ANBA - a rapid combined data acquisition and correction program for the SEMQ electron microprobe. U.S. Geological Survey Open File Report 83-817, 47 pages.

Rickwood, P.C. (1968). On recasting analyses of garnets into end-member molecules. Contr. Min. Pet. 18, 175-198.

Spear, F.S. (1980). Plotting stereoscopic phase diagrams. Am. Min. 65, 1291-1293.

--------- (1982). in Reviews of Mineralogy, Vol. 10. Characterization of Metamorphism through Mineral Equilibrium, J.M. Ferry, ed. pp.99-103.

Appendix I. Format of data files for site occupancy programs

```
 1
SI
TI
AL
FC
CR
FE
ZN
MN
NI
MG
CA
SR
BA
NA
K
P
F
CL
  1,AM S69-54 D-1
*HBBT ,23.000
56.81, 7.774
 0.39, 0.040
 1.80, 0.290
 0.00, 0.000
 0.03, 0.003
 0.52, 0.060
 0.00, 0.000
 0.34, 0.039
 0.00, 0.000
23.93, 4.880
13.15, 1.928
 0.00, 0.000
 0.00, 0.000
 0.13, 0.034
 0.07, 0.012
 0.00, 0.000
 0.29, 0.126
 0.00, 0.000
  2,SP S69-54 D-2
*HBBT , 5.000
30.14, 0.998
38.92, 0.969
 0.88, 0.034
 0.00, 0.000
 0.01, 0.000
 0.52, 0.014
 0.00, 0.000
 0.40, 0.011
 0.00, 0.000
 0.23, 0.011
27.48, 0.975
 0.00, 0.000
 0.00, 0.000
```

Appendix II. Listing of AMPHI

```
100 PRINT'PROGRAM AMPHI FOR RECALCULATION OF AMPHIBOLE ANALYSES'
102 PRINT'NUMBER OF OXYGENS = 23'
105 REM RECALCULATION OF MICROPROBE AMPHIBOLE DATA-M.FLOHR 9/82
110 REM MODIFICATIONS BY J.S.HUEBNER 9/82!FLOHR MODIFICATIONS 5/26/83
120 DIM A$(214),A(14),B(14),C(14),D(18,10),T$(48),W(14,10)
130 DIM Q(10),X(10),Z(14,10),Y(48,10),F(10)
170 READ A$
180 DATA /SIO2/AL2O3/FE2O3/FEO/MGO/CAO/NA2O/K2O/TIO2/MNO/
190 DATA BAO/CL/F/CR2O3/SUM/CL=O/F=O/SUM/H2O CALC/SUM/
210 DATA SI/ALIV/ T/ALVI/FE+3/FE+2/MG/TI/MN/CR/ M1-M3/MN/
230 DATA FE+2/MG/CA/NA/ M4/NA/K/CA/ A/CATSUM/F/CL/OH CALC/
240 DATA ANSUM/FE:FE+MG/MG:MG+FE/
260 FOR I=1 TO 4
270 READ D$
280 A$=A$+D$
290 NEXT I
300 REM A=FORMULA WT OXIDES,  B=# ANIONS/OXIDE FORMULA
310 REM C= # CATIONS/OXIDE FORMULA OR ANIONS
315 FOR I=1 TO 14\READ A(I)
330 DATA 60.09,101.94,159.7,71.85,40.32,56.08,61.982,94.2,
340 DATA 79.9,70.94,153.36,35.457,19,152.02
345 NEXT I
347 FOR I=1 TO 14\READ B(I)
360 DATA 2,3,3,1,1,1,1,2,1,1,1,3
365 NEXT I
367 FOR I=1 TO 14\READ C(I)
380 DATA 1,2,2,1,1,1,2,2,1,1,1,1,2
385 NEXT I
390 PRINT'ENTER RUN TITLE: '\INPUT B$
400 PRINT'ENTER FE+3/FE TOTAL= '!\INPUT B7
410 PRINT'ENTER NAME OF DATA FILE AS SYn:XXXXXX.YYY'\INPUT N$
420 PRINT'ARE YOU SORTING?(Y OR N): '!\INPUTS8$
430 IF ASC(S8$)<>89 GOTO455
450 PRINT'ENTER 2 CHARACTER SORT STRING '\INPUT M2$
455 PRINT'DO YOU WANT TO CREATE A DISK OUTPUT FILE WITH SITE OCCUPANCY'
460 PRINT'CATIONS? (Y OR N)'\INPUT R3$\IF ASC(R3$)<>89 GOTO475
465 PRINT'ENTER OUTPUT FILE NAME AS SYN:XXXXXX:YYY'\INPUT N3$
470 OPEN N3$ FOR OUTPUT AS FILE#4\PRINT#4,N$
475 PRINT'KEYPAD EDITOR ERROR CHECKING? (Y OR N)'\INPUT R$
480 PRINT'DO YOU WANT DATA TABLE PRINTED? (Y OR N)'\INPUTQ2$
485 IF ASC(Q2$)=89 THEN PRINT'ALIGN PRINT-HEAD TO VERY TOP OF PAPER'
510 OPEN N$ FOR INPUT AS FILE #1
520 IF ASC(Q2$)=89 THEN OPEN 'LP:' FOR OUTPUT AS FILE #2
525 Z$=CHR$(12)\Q$='#.###'
533 IF ASC(Q2$)<>89 GOTO 536
535 PRINT#2,'AMPHIBOLE PROGRAM',DAT$,CLK$\PRINT#2,B$
536 C4=0\N=0\INPUT#1,A8
537 FORJ=1TO18\INPUT#1,A5$\NEXTJ
538 GOTO545
539 IF ASC(Q2$)=89 THEN PRINT#2,Z$\IF ASC(Q2$)=89 THEN PRINT#2
545 N2=0
550 IF END#1 THEN GOTO 650
554 N=N+1
557 IF ASC(R$)=89 THEN PRINT 'N= '!N
560 INPUT#1,F8,T$\INPUT#1,T5$,F9
563 IF ASC(R$)=89 THEN PRINTF8!T$
567 IF ASC(S8$)<>89 GOTO575
570 M8$=SEG$(T$,1,2)\IF M8$<>M2$ GOTO595
575 N2=N2+1\F(N2)=F8
580 FOR J=1TO18\INPUT#1,D(J,N2),D3\NEXTJ
585 IF ASC(Q2$)=89 THEN PRINT#2,F(N2)!T$
590 IF N2<10 GOTO550\GOTO622
595 FOR J=1TO18\INPUT#1,H5,H3\NEXTJ
600 GOTO550
622 IF ASC(Q2$)=89 THEN PRINT#2
630 GOTO660
650 PRINT\PRINT 'END OF DATA FILE REACHED AT ANALYSIS '!N
651 C4=1\CLOSE#1
653 IF N2=0 GOTO 1570
660 FORJ=1TON2\Y(1,J)=D(1,J)\Y(2,J)=D(3,J)
665 Y(3,J)=D(4,J)\Y(4,J)=D(6,J)\Y(5,J)=D(10,J)
670 Y(6,J)=D(11,J)\Y(7,J)=D(14,J)\Y(8,J)=D(15,J)
675 Y(9,J)=D(2,J)\Y(10,J)=D(8,J)\Y(11,J)=D(13,J)
680 Y(12,J)=D(18,J)\Y(13,J)=D(17,J)\Y(14,J)=D(5,J)
690 NEXT J\REM CALCULATE FE2O3 AND FEO
720 FORJ=1TON2\IF Y(3,J)<>0 GOTO750
730 Y(3,J)=Y(4,J)*B7*1.111\Y(4,J)=Y(4,J)-Y(4,J)*B7
740 REM CALCULATE OXIDE SUM AND CATIONS!A1=#OXYGENS
750 A1=23\Q(J)=0\Y(15,J)=0
760 FOR I=1TO14\Y(15,J)=Y(15,J)+Y(I,J)
770 W(I,J)=Y(I,J)/A(I)*B(I)\Q(J)=Q(J)+W(I,J)\NEXTI
```

```
790 Q(J)=Q(J)-W(12,J)-W(13,J)
800 X(J)=A1/Q(J)\REM X=NORMALIZATION FACTOR
810 FOR I=1 TO 14\Z(I,J)=X(J)*(Y(I,J)/A(I))*C(I)\NEXTI
820 Y(16,J)=Y(12,J)*.226\Y(17,J)=Y(13,J)*.421
830 Y(18,J)=Y(15,J)-Y(16,J)-Y(17,J)
840 P(J)=2-Z(12,J)-Z(13,J)\Y(19,J)=P(J)/2/X(J)*18.016
850 Y(20,J)=Y(18,J)+Y(19,J)\NEXTJ
950 REM SUBROUTINE TO CALCULATE SITE OCCUPANCIES
970 GOSUB 1590
980 IF ASC(R3$)<>89 GOTO 1060
990 FOR I=1TON2\PRINT#4,USING"####",F(I)
1000 FOR J=21TO22\PRINT#4,USING Q$,Y(J,I)\NEXTJ
1010 FOR J=24TO30\PRINT#4,USING Q$,Y(J,I)\NEXTJ
1020 FOR J=32TO36\PRINT#4,USING Q$,Y(J,I)\NEXTJ
1030 FOR J=38TO41\PRINT#4,USING Q$,Y(J,I)\NEXTJ
1040 FOR J=43TO44\PRINT#4,USING Q$,Y(J,I)\NEXTJ\NEXTI
1060 IF ASC(Q2$)<>89 GOTO1530
1360 PRINT #2, 'ANALYSIS';\FOR K=1TON2
1380 PRINT #2, USING "############";F(K);\NEXTK
1400 PRINT #2\PRINT #2
1410 FOR J=1 TO 48\IF J=21GOTO1460
1450 GOTO 1470
1460 PRINT #2
1470 GOSUB 1930
1480 PRINT #2, USING "'LLLLLLLL";X$;
1485 IF J>=21 THEN GOTO1515
1490 FOR L=1 TO N2
1500 PRINT #2, USING "########.##";Y(J,L);
1510 NEXT L\PRINT#2\GOTO1520
1515 FOR L=1TON2\PRINT#2,USING"#######.###";Y(J,L);
1518 NEXTL\PRINT#2
1520 NEXT J
1530 IF C4=1 GOTO 1570
1560 GOTO 539
1570 IF ASC(Q2$)<>89 GOTO 1575
1572 PRINT#2,Z$\CLOSE #2
1575 IF ASC(R3$)=89 THEN CLOSE#4
1580 GOTO 4000
1590 REM SUBROUTINE TO CALCULATE SITE OCCUPANCIES
1600 FOR J=1 TO N2
1610 Y(21,J)=Z(1,J)\IF Z(1,J)+Z(2,J)>8 GO TO 1670
1620 Y(21,J)=Z(1,J)\Y(22,J)=Z(2,J)\Y(24,J)=0
1660 GOTO 1690
1670 Y(22,J)=8-Z(1,J)\Y(24,J)=Z(2,J)-Y(22,J)
1690 Y(23,J)=Y(21,J)+Y(22,J)
1700 Y(25,J)=Z(3,J)\Y(26,J)=Z(4,J)\Y(27,J)=Z(5,J)
1710 Y(28,J)=Z(9,J)\Y(29,J)=Z(10,J)\Y(30,J)=Z(14,J)
1760 Y(31,J)=Y(24,J)+Y(25,J)+Y(26,J)+Y(27,J)+Y(28,J)+Y(29,J)+Y(30,J)
1762 IF Y(31,J)>5 GOTO 1783
1770 Y(32,J)=0\Y(33,J)=0\Y(34,J)=0
1780 Y(35,J)=Z(6,J)\Y(36,J)=Z(7,J)\GOTO1836
1783 IF Y(31,J)-Y(29,J)>5 GOTO 1806
1785 Y(29,J)=5-(Y(24,J)+Y(25,J)+Y(26,J)+Y(27,J)+Y(28,J)+Y(30,J))
1790 Y(32,J)=Z(10,J)-Y(29,J)\Y(33,J)=0\Y(34,J)=0
1795 Y(35,J)=Z(6,J)\Y(36,J)=Z(7,J)\GOTO1836
1806 IF Y(31,J)-(Z(10,J)+Y(26,J))>5 GOTO 1824
1810 Y(29,J)=0\Y(32,J)=Z(10,J)
1812 Y(26,J)=5-(Y(24,J)+Y(25,J)+Y(27,J)+Y(28,J)+Y(29,J)+Y(30,J))
1815 Y(33,J)=Z(4,J)-Y(26,J)\Y(34,J)=0\Y(35,J)=Z(6,J)
1820 Y(36,J)=Z(7,J)\GOTO1836
1824 Y(29,J)=0\Y(26,J)=0
1828 Y(27,J)=5-(Y(24,J)+Y(25,J)+Y(26,J)+Y(28,J)+Y(29,J)+Y(30,J))
1830 Y(34,J)=Z(5,J)-Y(27,J)\Y(32,J)=Z(10,J)\Y(33,J)=Y(26,J)
1836 Y(37,J)=Y(32,J)+Y(33,J)+Y(34,J)+Y(35,J)+Y(36,J)\IFY(37,J)>2GOTO1852
1840 Y(35,J)=Z(6,J)\Y(36,J)=Z(7,J)\Y(38,J)=0
1845 Y(40,J)=0\Y(39,J)=Z(8,J)\GOTO1890
1852 IF Y(37,J)-Z(7,J)>2 GOTO 1866\Y(40,J)=Z(6,J)
1855 Y(36,J)=2-(Y(32,J)+Y(33,J)+Y(34,J)+Y(35,J))\Y(38,J)=Z(7,J)-Y(36,J)
1860 Y(39,J)=Z(8,J)\Y(40,J)=Z(6,J)-Y(35,J)\GOTO1890
1866 IF Y(37,J)-(Z(7,J)+Z(6,J))>2 GOTO 1880
1870 Y(38,J)=Z(7,J)\Y(36,J)=0\Y(40,J)=Y(37,J)-(Z(7,J)+2)
1875 Y(35,J)=Z(6,J)-Y(40,J)\Y(39,J)=Z(8,J)\GOTO1890
1880 Y(36,J)=0\Y(38,J)=Z(7,J)
1885 Y(40,J)=Z(6,J)\Y(39,J)=Z(8,J)\Y(35,J)=0
1890 Y(31,J)=Y(24,J)+Y(25,J)+Y(26,J)+Y(27,J)+Y(28,J)+Y(29,J)+Y(30,J)
1892 Y(37,J)=Y(32,J)+Y(33,J)+Y(34,J)+Y(35,J)+Y(36,J)
1894 Y(41,J)=Y(38,J)+Y(39,J)+Y(40,J)
1896 Y(42,J)=Y(31,J)+Y(37,J)+Y(41,J)+Y(23,J)
```

```
1896 Y(42,J)=Y(31,J)+Y(37,J)+Y(41,J)+Y(23,J)
1900 Y(43,J)=Z(13,J)\Y(44,J)=Z(12,J)\Y(45,J)=P(J)
1904 Y(46,J)=Y(43,J)+Y(44,J)+Y(45,J)
1910 Y(47,J)=(Z(4,J)+Z(3,J))/(Z(5,J)+Z(4,J)+Z(3,J))
1912 Y(48,J)=Z(5,J)/(Z(5,J)+Z(4,J)+Z(3,J))
1915 NEXT J\RETURN
1930 REM SUBROUTINE TO CHOOSE ROW LABELS FROM A$
1940 B1=0
1950 N1=0
1960 FOR I=1 TO LEN(A$)
1970 S$=SEG$(A$,I,I)
1980 IF S$<>'/' GOTO 2060
1990 N1=N1+1
2000 IF N1=J GOTO 2020
2010 IF N1=J+1 GOTO 2040
2020 B1=I+1
2030 GOTO 2060
2040 E1=I-1
2050 X$=SEG$(A$,B1,E1)
2060 NEXT I
2070 RETURN
4000 END
```

Appendix II. Example of output from AMPHI

AMPHIBOLE PROGRAM 26-SEP-83 14:40:26
S69-54 AMPHIBOLES
1 AM S69-54 D-1
4 AM S69-54 D-4-1
5 AM S69-54 D-4-2
6 AM S69-54 D-5
7 AM S69-54 B-1
8 AM S69-54 B-2
9 AM S69-54 B-3
10 AM S69-54 B-4
11 AM S69-54 B-4-2
12 AM S69-54 C-1

ANALYSIS	1	4	5	6	7	8	9	10	11	12
SIO2	56.81	47.02	46.40	46.12	50.42	51.06	48.84	51.08	49.61	51.38
AL2O3	1.80	9.80	10.97	11.20	6.61	7.56	8.36	7.94	7.41	7.33
FE2O3	0.00	0.00	0.00	0.00	0.00	0.00	0.00	0.00	0.00	0.00
FEO	0.52	14.84	13.82	13.34	10.63	4.68	9.54	4.06	11.82	4.87
MGO	23.93	11.57	11.39	12.52	15.18	18.52	14.99	18.85	14.13	18.92
CAO	13.15	11.60	11.69	11.74	11.90	12.25	11.67	12.40	11.70	11.87
NA2O	0.13	0.71	0.75	0.95	0.46	0.57	0.65	0.69	0.53	0.55
K2O	0.07	0.17	0.16	0.17	0.08	0.10	0.14	0.13	0.10	0.13
TIO2	0.39	0.35	0.38	0.37	0.21	0.37	0.28	0.29	0.21	0.18
MNO	0.34	1.07	1.30	1.19	1.48	1.52	1.54	1.32	1.29	1.58
BAO	0.00	0.00	0.00	0.00	0.00	0.00	0.00	0.00	0.00	0.00
CL	0.00	0.05	0.00	0.03	0.01	0.05	0.00	0.00	0.02	0.05
F	0.29	0.02	0.00	0.12	0.03	0.03	0.00	0.00	0.00	0.13
CR2O3	0.03	0.01	0.00	0.00	0.03	0.00	0.00	0.00	0.00	0.00
SUM	97.46	97.21	96.86	97.75	97.04	96.71	96.01	96.76	96.82	96.99
CL=O	0.00	0.01	0.00	0.01	0.00	0.01	0.00	0.00	0.00	0.01
F=O	0.12	0.01	0.00	0.05	0.01	0.01	0.00	0.00	0.00	0.05
SUM	97.34	97.19	96.86	97.69	97.03	96.69	96.01	96.76	96.82	96.92
H2O CALC	2.05	2.01	2.03	1.98	2.05	2.09	2.06	2.13	2.05	2.05
SUM	99.39	99.20	98.89	99.68	99.08	98.78	98.07	98.89	98.87	98.97
SI	7.774	6.946	6.852	6.755	7.298	7.223	7.122	7.195	7.231	7.253
ALIV	0.226	1.054	1.148	1.245	0.702	0.777	0.878	0.805	0.769	0.747
T	8.000	8.000	8.000	8.000	8.000	8.000	8.000	8.000	8.000	8.000
ALVI	0.065	0.653	0.761	0.689	0.426	0.484	0.559	0.514	0.505	0.473
FE+3	0.000	0.000	0.000	0.000	0.000	0.000	0.000	0.000	0.000	0.000
FE+2	0.012	1.760	1.690	1.538	1.273	0.554	1.153	0.478	1.403	0.528
MG	4.880	2.547	2.507	2.733	3.275	3.904	3.258	3.957	3.070	3.980
TI	0.040	0.039	0.042	0.041	0.023	0.039	0.031	0.031	0.023	0.019
MN	0.000	0.000	0.000	0.000	0.000	0.019	0.000	0.020	0.000	0.000
CR	0.003	0.001	0.000	0.000	0.003	0.000	0.000	0.000	0.000	0.000
M1-M3	5.000	5.000	5.000	5.000	5.000	5.000	5.000	5.000	5.000	5.000
MN	0.039	0.134	0.163	0.148	0.181	0.163	0.190	0.137	0.159	0.189
FE+2	0.048	0.073	0.017	0.097	0.014	0.000	0.011	0.000	0.038	0.047
MG	0.000	0.000	0.000	0.000	0.000	0.000	0.000	0.000	0.000	0.000
CA	1.913	1.793	1.821	1.756	1.804	1.837	1.799	1.863	1.803	1.764
NA	0.000	0.000	0.000	0.000	0.000	0.000	0.000	0.000	0.000	0.000
M4	2.000	2.000	2.000	2.000	2.000	2.000	2.000	2.000	2.000	2.000
NA	0.034	0.203	0.215	0.270	0.129	0.156	0.184	0.188	0.150	0.151
K	0.012	0.032	0.030	0.032	0.015	0.018	0.026	0.023	0.019	0.023
CA	0.015	0.043	0.029	0.087	0.041	0.020	0.024	0.009	0.025	0.031
A	0.062	0.279	0.274	0.388	0.185	0.194	0.234	0.221	0.193	0.205
CATSUM	15.062	15.279	15.274	15.388	15.185	15.195	15.234	15.221	15.193	15.205
F	0.126	0.009	0.000	0.056	0.014	0.013	0.000	0.000	0.000	0.058
CL	0.000	0.013	0.000	0.007	0.002	0.012	0.000	0.000	0.005	0.012
OH CALC	1.874	1.978	2.000	1.937	1.984	1.975	2.000	2.000	1.995	1.930
ANSUM	2.000	2.000	2.000	2.000	2.000	2.000	2.000	2.000	2.000	2.000
FE:FE+MG	0.012	0.419	0.405	0.374	0.282	0.124	0.263	0.108	0.319	0.126
MG:MG+FE	0.988	0.581	0.595	0.626	0.718	0.876	0.737	0.892	0.681	0.874

Appendix II. Example of input for AMPHI.

The following are the prompts given by program AMPHI and the responses (indicated by a box [ ]) which were given to generate the data table shown in this appendix. Note that the cation output file created during this run (SY1:PA6954.DAT) was used later as one of the input files to create the plot shown in Appendix XVII (output from the program PLAMPH).

ENTER RUN TITLE: ? [S69-54 AMPHIBOLES]

ENTER FE+3/FE TOTAL= ? [0]

ENTER NAME OF DATA FILE AS SYn:XXXXXX.YYY: ? [SY1:S6954.DAT]

ARE YOU SORTING? (Y OR N) ? [Y]

ENTER 2 CHARACTER SORT STRING ? [AM]

DO YOU WANT TO CREATE A DISK OUTPUT FILE WITH SITE OCCUPANCY

CATIONS? (Y OR N) ? [Y]

ENTER OUTPUT FILE NAME AS SYN:XXXXXX.YYY ? [SY1:PA6954.DAT]

KEYPAD EDITOR ERROR CHECKING? (Y OR N) ? [N]

DO YOU WANT DATA TABLE PRINTED OUT? (Y OR N) ? [Y]

ALIGN PRINTER HEAD TO VERY TOP OF PAGE

Appendix III. Listing of MICA

```
100 PRINT'PROGRAM MICA FOR RECALCULATION OF MICA ANALYSES'
110 REM M.FLOHR 9/82; MODIFIED BY J.S.HUEBNER;MODIFIED 6/83-FLOHR
120 DIM A$(192),A(14),B(14),C(14),D(18,10),T$(48),W(14,10)
130 DIM Q(10),X(10),Z(14,10),Y(43,10),F(10)
170 READ A$
180 DATA /SIO2/AL2O3/FE2O3/FEO/MGO/CAO/NA2O/K2O/TIO2/MNO/
190 DATA BAO/CL/F/CR2O3/SUM/CL=O/F=O/SUM/H2O CALC/SUM/
210 DATA SI/ALIV/ T/ALVI/FE+3/FE+2/MG/TI/MN/CR/ M/CA/NA/
230 DATA K/BA/ A/CATSUM/F/CL/OH CALC/ANSUM/FE:FE+MG/
250 DATA MG:MG+FE/
260 FOR I=1TO4
270 READ D$
280 A$=A$+D$
290 NEXT I
300 REM A=FORMULA WT OXIDES, B=#ANIONS/OXIDE FORMULA
310 REM C=#CATIONS/OXIDE FORMULA OF ANIONS
315 FOR I=1TO14\READ A(I)
330 DATA 60.09,101.94,159.7,71.85,40.32,56.08,61.982,94.2,
340 DATA 79.9,70.94,153.36,35.457,19,152.02
345 NEXT I
347 FOR I=1TO14\READ B(I)
360 DATA 2,3,3,1,1,1,1,1,2,1,1,1,1,3
365 NEXT I
367 FOR I=1TO14\READ C(I)
380 DATA 1,2,2,1,1,1,2,2,1,1,1,1,1,2
385 NEXT I
390 Z$=CHR$(12)\Q$='$,###'\CB$=CHR$(44)
395 PRINT'ENTER RUN TITLE:'\INPUTB$
400 PRINT'ENTER FE+3/FE TOTAL= ';\INPUTR7
405 PRINT'ENTER FORMULA BASIS (11 OR 22 OXYGENS): ';\INPUTA1
410 PRINT'ENTER NAME OF DATA FILE AS SY#:XXXXXX.YYY'\INPUTN$
415 PRINT'DO YOU WANT TO CREATE A DISK OUTPUT FILE WITH SITE OCCUPANCY'
420 PRINT'CATIONS? (Y OR N)';\INPUTR3$
425 IF ASC(R3$)<>89 GOTO440
430 PRINT'ENTER OUTPUT FILE NAME AS SYN:XXXXXX.YYY'\INPUTN3$
435 OPEN N3$ FOR OUTPUT AS FILE#4\PRINT#4,N$
440 PRINT'KEYPAD EDITOR ERROR CHECKING?(Y OR N)'\INPUTR$
445 OPEN N$ FOR INPUT AS FILE #1
450 PRINT'PRINT OUT DATA TABLE?(Y OR N)';\INPUTQ2$
455 IF ASC(Q2$)=89 THEN OPEN 'LP:' FOR OUTPUT AS FILE #2
460 PRINT'ARE YOU SORTING?(Y OR N): ';\INPUT S8$
465 IF ASC(S8$)<>89 GOTO520
470 PRINT'YOU CAN ENTER UP TO TWO (2) SORT STRINGS'
475 PRINT'ENTER FIRST 2 CHARACTER SORT STRING ';\INPUT M1$
480 PRINT'ENTER SECOND 2 CHARACTER SORT STRING -- ENTER ZERO (0) IF'
485 PRINT'YOU DO NOT WANT TO SORT OUT A 2ND GROUP OF ANALYSES';\INPUT M2$
520 INPUT #1,I9\FOR J=1TO18\INPUT#1,A5$\NEXTJ
528 IF ASC(Q2$)<>89 GOTO 536
530 PRINT#2,'MICA PROGRAM',DAT$,CLK$
535 PRINT#2\PRINT#2,B$\PRINT#2
536 N=0\C4=0\REM N IS COUNTER FOR TOTAL #ANALYSES IN FILE;C4 FOR END OF FILE
537 GOTO 539
538 IF ASC(Q2$)=89 THEN PRINT#2,Z$\IF ASC(Q2$)=89 THEN PRINT#2
539 N2=0\REM COUNTER FOR #ANALYSES ACCEPTED FOR CALCULATION AT ONE TIME
550 IF END#1 THEN GOTO 650
555 N=N+1
557 IF ASC(R$)=89 THEN PRINT 'N= ';N
558 INP
567 IF ASC(R$)=89 THEN PRINT #5;T$
568 IF ASC(S8$)<>89 GOTO605
570 M8$=SEG$(T$,1,2)\IF M8$=M1$ GOTO595
573 IF M2$='0' GOTO580
575 IF M8$=M2$ GOTO605
580 FOR J=1TO18\INPUT#1,D6,D7\NEXTJ
585 IF N2<10 GOTO550\GOTO625
595 N2=N2+1\F(N2)=F8\IF ASC(Q2$)=89 THEN PRINT#2,F(N2);T$
600 V2(N2)=1\GOTO610
605 N2=N2+1\F(N2)=F8\IF ASC(Q2$)=89 THEN PRINT#2,F(N2);T$
608 V2(N2)=0\GOTO610
609 N2=N2+1\F(N2)=F8\V2(N2)=2\IF ASC(Q2$)=89 THEN PRINT#2,F(N2);T$
610 FOR J=1TO18\INPUT#1,D(J,N2),D3\NEXTJ
615 IF N2<10 GOTO550
625 IF ASC(Q2$)<>89 GOTO640
630 PRINT#2\PRINT#2
640 GOTO670
650 PRINT\PRINT'END OF DATA FILE REACHED AT ANALYSIS ';N
655 C4=1\CLOSE#1
660 IF N2=0 GOTO1570
670 FOR J=1TON2
675 Y(1,J)=D(1,J)\Y(2,J)=D(3,J)\Y(3,J)=D(4,J)
680 Y(4,J)=D(6,J)\Y(5,J)=D(10,J)\Y(6,J)=D(11,J)
```

```
685 Y(7,J)=D(14,J)\Y(8,J)=D(15,J)\Y(9,J)=D(2,J)
690 Y(10,J)=D(8,J)\Y(11,J)=D(13,J)\Y(12,J)=D(18,J)
700 Y(13,J)=D(17,J)\Y(14,J)=D(5,J)
705 NEXTJ
710 REM CALCULATE FE2O3 AND FEO
720 FOR J=1TON2\IF Y(3,J)<>0 GOTO760
730 Y(3,J)=Y(4,J)*B7*1.111\Y(4,J)=Y(4,J)-Y(4,J)*B7
760 REM CALCULATE OXIDE SUM AND CATIONS
780 Q(J)=0\Y(15,J)=0
790 FOR I=1TO14
800 Y(15,J)=Y(15,J)+Y(I,J)
810 W(I,J)=Y(I,J)/A(I)*B(I)
820 Q(J)=Q(J)+W(I,J)
830 NEXT I
840 Q(J)=Q(J)-W(12,J)-W(13,J)
850 X(J)=A1/Q(J)\REM X=NORMALIZATION FACTOR
870 FOR I=1TO14\Z(I,J)=X(J)*(Y(I,J)/A(I)*C(I))\NEXTI
880 Y(16,J)=Y(12,J)*.266\Y(17,J)=Y(13,J)*.421
890 Y(18,J)=Y(15,J)-Y(16,J)-Y(17,J)
905 IF A1=22 GOTO 915
910 P(J)=2-Z(12,J)-Z(13,J)
913 GOTO 920
915 P(J)=4-Z(12,J)-Z(13,J)
920 Y(19,J)=P(J)/2/X(J)*18.016
930 Y(20,J)=Y(18,J)+Y(19,J)\NEXTJ
950 GOSUB1590\REM SUBROUTINE TO CALCULATE SITE OCCUPANCIES
980 IF ASC(R3$)<>89 GOTO1060
995 FOR I=1TON2\PRINT#4,USING"####";F(I);\PRINT#4,C8$;
1000 IF V2(I)=0 THEN PRINT#4,USING"LL";M2$
1010 IF V2(I)=1 THEN PRINT#4,USING"LL";M1$
1015 IF V2(I)=2 THEN PRINT#4,USING"LL";"MI"
1020 FOR J=21TO22\PRINT#4,USING Q$,Y(J,I)\NEXTJ
1030 FOR J=24TO30\PRINT#4,USING Q$,Y(J,I)\NEXTJ
1040 FOR J=32TO35\PRINT#4,USING Q$,Y(J,I)\NEXTJ
1050 FOR J=38TO39\PRINT#4,USING Q$,Y(J,I)\NEXTJ
1055 NEXTI\K8=K8+1
1060 IF ASC(Q2$)<>89 GOTO1530
1340 REM PRINT DATA TABLE
1360 PRINT #2, "ANALYSIS";
1370 FOR K=1TON2
1380 PRINT #2, USING "##########";F(K);\NEXTK
1400 PRINT #2\PRINT #2
1410 FOR J=1TO43
1420 IF J=21 GOTO1460
1430 IF J=42 GOTO1460
1440 IF J=43 GOTO1460
1450 GOTO 1470
1460 PRINT#2
1470 GOSUB1930
1480 PRINT#2,USING"LLLLLLLL";X$;
1485 IF J>=21 GOTO1513
1490 FOR L=1TON2
1500 PRINT #2,USING"########.##",Y(J,L);
1510 NEXT L\PRINT#2\GOTO1520
1513 FOR L=1TON2\PRINT#2,USING"#######.###",Y(J,L);
1515 NEXTL\PRINT#2
1520 NEXTJ
1530 IF C4=1 GOTO1570
1560 GOTO538
1570 IF ASC(Q2$)=89 THEN PRINT#2,Z$
1573 IF ASC(Q2$)=89 THEN CLOSE#2
1575 IF ASC(R3$)=89 THEN CLOSE#4
1580 GOTO4000
1590 REM SUBROUTINE TO CALCULATE SITE OCCUPANCIES
1600 FOR J=1TON2
1610 Y(21,J)=Z(1,J)
1615 IF A1=22 GOTO 1625
1620 IF Z(1,J)+Z(2,J)>4 GOTO1670
1622 GOTO 1630
1625 IF Z(1,J)+Z(2,J)>8 GOTO1675
1630 Y(21,J)=Z(1,J)\Y(22,J)=Z(2,J)\Y(24,J)=0
1660 GOTO 1690
1670 Y(22,J)=4-Z(1,J)
1672 GOTO 1680
1675 Y(22,J)=8-Z(1,J)
1680 Y(24,J)=Z(2,J)-Y(22,J)
1690 Y(23,J)=Y(21,J)+Y(22,J)
```

```
1700 Y(25,J)=Z(3,J)\Y(26,J)=Z(4,J)
1720 Y(27,J)=Z(5,J)\Y(28,J)=Z(9,J)
1740 Y(29,J)=Z(10,J)\Y(30,J)=Z(14,J)
1760 Y(31,J)=Y(24,J)+Y(25,J)+Y(26,J)+Y(27,J)+Y(28,J)+Y(29,J)+Y(30,J)
1770 Y(32,J)=Z(6,J)\Y(33,J)=Z(7,J)
1790 Y(34,J)=Z(8,J)\Y(35,J)=Z(11,J)
1810 Y(36,J)=Y(32,J)+Y(33,J)+Y(34,J)+Y(35,J)
1820 Y(37,J)=Y(23,J)+Y(31,J)+Y(36,J)
1830 Y(38,J)=Z(13,J)\Y(39,J)=Z(12,J)\Y(40,J)=P(J)
1860 Y(41,J)=Y(38,J)+Y(39,J)+Y(40,J)
1890 Y(42,J)=(Y(25,J)+Y(26,J))/(Y(25,J)+Y(26,J)+Y(27,J))
1900 Y(43,J)=Y(27,J)/(Y(25,J)+Y(26,J)+Y(27,J))
1910 NEXT J
1920 RETURN
1930 REM SUBROUTINE TO CHOOSE ROW LABELS FROM A$
1940 B1=0\N1=0
1960 FOR I=1 TO LEN(A$)
1970 S$=SEG$(A$,I,I)
1980 IF S$<>'/' GOTO 2060
1990 N1=N1+1
2000 IF N1=J GOTO 2020
2010 IF N1=J+1 GOTO 2040
2020 B1=I+1
2030 GOTO 2060
2040 E1=I-1
2050 X$=SEG$(A$,B1,E1)
2060 NEXT I
2070 RETURN
4000 END
```

Appendix III. Example of output from MICA

MICA PROGRAM 26-SEP-83 14:34:32

S68-51 BIOTITES

2 BI S68-51 3-1CORE
3 BI S68-51 3-1RIM
4 BI S68-51 3-2CORE
5 BI S68-51 3-2RIM
31 BI S68-51 5-4
32 BI S68-51 5-5
33 BI S68-51 5-6-7-8
48 BI S68-51 7-6

ANALYSIS	2	3	4	5	31	32	33	48
SIO2	37.72	37.92	36.92	37.54	37.35	37.37	37.44	36.63
AL2O3	15.58	15.71	15.88	15.97	15.58	15.54	15.65	15.83
FE2O3	0.00	0.00	0.00	0.00	0.00	0.00	0.00	0.00
FEO	17.72	17.82	17.71	18.05	18.36	17.82	17.90	18.01
MGO	13.23	12.88	12.86	12.75	12.42	12.32	12.43	12.28
CAO	0.03	0.01	0.04	0.05	0.06	0.13	0.10	0.09
NA2O	0.15	0.18	0.15	0.17	0.16	0.16	0.15	0.12
K2O	8.54	9.02	9.04	8.77	9.10	8.89	8.98	8.91
TIO2	2.92	3.15	2.99	3.21	3.22	3.13	2.99	3.18
MNO	0.01	0.01	0.05	0.05	0.04	0.05	0.06	0.04
BAO	0.00	0.00	0.00	0.00	0.00	0.00	0.00	0.00
CL	0.00	0.00	0.00	0.00	0.00	0.00	0.00	0.00
F	0.00	0.00	0.00	0.00	0.00	0.00	0.00	0.00
CR2O3	0.00	0.00	0.00	0.00	0.00	0.00	0.00	0.00
SUM	95.90	96.70	95.64	96.56	96.29	95.41	95.70	95.09
CL=O	0.00	0.00	0.00	0.00	0.00	0.00	0.00	0.00
F=O	0.00	0.00	0.00	0.00	0.00	0.00	0.00	0.00
SUM	95.90	96.70	95.64	96.56	96.29	95.41	95.70	95.09
H2O CALC	4.02	4.04	3.99	4.04	4.01	3.98	4.00	3.96
SUM	99.92	100.75	99.63	100.60	100.30	99.39	99.70	99.05
SI	2.812	2.811	2.774	2.788	2.795	2.812	2.810	2.773
ALIV	1.188	1.189	1.226	1.212	1.205	1.188	1.190	1.227
T	4.000	4.000	4.000	4.000	4.000	4.000	4.000	4.000
ALVI	0.181	0.183	0.181	0.187	0.169	0.190	0.194	0.185
FE+3	0.000	0.000	0.000	0.000	0.000	0.000	0.000	0.000
FE+2	1.105	1.105	1.113	1.121	1.149	1.121	1.123	1.140
MG	1.470	1.423	1.440	1.411	1.385	1.381	1.390	1.385
TI	0.164	0.176	0.169	0.179	0.181	0.177	0.169	0.181
MN	0.001	0.001	0.003	0.003	0.003	0.003	0.004	0.003
CR	0.000	0.000	0.000	0.000	0.000	0.000	0.000	0.000
M	2.920	2.887	2.906	2.902	2.886	2.873	2.880	2.894
CA	0.002	0.001	0.003	0.004	0.005	0.010	0.008	0.007
NA	0.022	0.026	0.022	0.024	0.023	0.023	0.022	0.018
K	0.812	0.853	0.867	0.831	0.869	0.853	0.860	0.860
BA	0.000	0.000	0.000	0.000	0.000	0.000	0.000	0.000
A	0.836	0.880	0.892	0.859	0.897	0.887	0.890	0.885
CATSUM	7.757	7.767	7.798	7.761	7.783	7.760	7.770	7.779
F	0.000	0.000	0.000	0.000	0.000	0.000	0.000	0.000
CL	0.000	0.000	0.000	0.000	0.000	0.000	0.000	0.000
OH CALC	2.000	2.000	2.000	2.000	2.000	2.000	2.000	2.000
ANSUM	2.000	2.000	2.000	2.000	2.000	2.000	2.000	2.000
FE:FE+MG	0.429	0.437	0.436	0.443	0.453	0.448	0.447	0.451
MG:MG+FE	0.571	0.563	0.564	0.557	0.547	0.552	0.553	0.549

Appendix IV. Listing of CHLOR

```
100 PRINT'PROGRAM CHLOR.BAS FOR RECALCULATION OF CHLORITE ANALYSES'
110 REM M.FLOHR 10/82! MODIFIED 6/24/83
120 DIM A$(156),A(12),B(12),C(12),D(18,10),T$(48),W(12,10)
130 DIM Q(10),X(10),Z(12,10),Y(34,10),F(10)
170 READ A$
180 DATA /SIO2/AL2O3/FE2O3/FEO/MGO/CAO/NA2O/K2O/TIO2/MNO/
190 DATA CR2O3/NIO/SUM/H2O CALC/SUM/SI/AL/ Z/AL/FE+3/FE+2/MG/
200 DATA NI/TI/MN/CR/CA/NA/K/ Y/CATSUM/(OH)/FE:FE+MG/MG:MG+FE/
210 FOR I=1 TO 2
270 READ D$
280 A$=A$+D$
290 NEXT I
300 REM A=FORMULA WT OXIDES,  B=# ANIONS/OXIDE FORMULA
310 REM C= # CATIONS/OXIDE FORMULA OR ANIONS
315 FOR I=1 TO 12\READ A(I)
330 DATA 60.09,101.94,159.7,71.85,40.32,56.08,61.982,94.2,
340 DATA 79.9,70.94,152.02,74.71
345 NEXT I
347 FOR I=1 TO 12\READ B(I)
360 DATA 2,3,3,1,1,1,1,1,2,1,3,1
365 NEXT I
367 FOR I=1 TO 12\READ C(I)
380 DATA 1,2,2,1,1,1,2,2,1,1,2,1
385 NEXT I
390 PRINT'ENTER RUN TITLE:'\INPUTB$
400 PRINT'ENTER FE+3/FE TOTAL '\INPUT B7
410 PRINT'ENTER NAME OF DATA FILE AS SYn:XXXXXX.YYY'\INPUTN$
420 PRINT'ARE YOU SORTING?(Y OR N): '!\INPUT S8$
425 IF ASC(S8$)<>89 GOTO440
430 PRINT'ENTER 2 CHARACTER SORT STRING'\INPUT M2$
440 PRINT'DO YOU WANT TO CREATE A DISK OUTPUT FILE WITH SITE OCCUPANCY'
450 PRINT'CATIONS?(Y OR N)'\INPUTR3$
460 IF ASC(R3$)<>89 GOTO490
470 PRINT'ENTER OUTPUT FILE NAME AS SYN:XXXXXX.YYY'\INPUTN3$
480 OPEN N3$ FOR OUTPUT AS FILE#4\PRINT#4,N$
490 PRINT'KEYPAD EDITOR ERROR CHECKING? (Y OR N)'\INPUTR$
500 OPEN N$ FOR INPUT AS FILE #1
510 PRINT'PRINT OUT DATA TABLE?(Y OR N)'\INPUTQ2$
520 IF ASC(Q2$)=89 THEN OPEN 'LP:' FOR OUTPUT AS FILE #2
525 INPUT#1,I9\FORJ=1TO18\INPUT#1,A5$\NEXTJ
530 Z$=CHR$(12)\Q$='##.###'
532 IF ASC(Q2$)<>89 GOTO536
534 PRINT#2,'CHLORITE PROGRAM',DAT$,CLK$
535 PRINT#2\PRINT#2,B$\PRINT#2
536 N=0\C4=0\REM N IS COUNTER FOR TOTAL #ANALYSES IN FILE!C4 FOR END OF FILE
537 GOTO 539
538 IF ASC(Q2$)=89 THEN PRINT#2,Z$
539 N2=0\REM N2 COUNTER FOR #ANALYSES ACCEPTED FOR CALCULATION AT ONE TIME
550 IF END#1 THEN GOTO640
555 N=N+1
557 IF ASC(R$)=89 THEN PRINT 'N= '!N
560 INPUT #1,F8,T$\INPUT#1,T5$,F9
565 IF ASC(R$)=89 THEN PRINT F8!T$
568 IF ASC(S8$)<>89 GOTO585
570 M8$=SEG$(T$,1,2)\IF M8$=M2$ GOTO585
575 FOR J=1TO18\INPUT#1,D6,D7\NEXTJ
580 IFN2<10 GOTO550\GOTO600
585 N2=N2+1\F(N2)=F8\IF ASC(Q2$)=89 THEN PRINT#2,F(N2)!T$
590 FOR J=1TO18\INPUT#1,D(J,N2),D3\NEXTJ
595 IF N2<10 GOTO550
600 IF ASC(Q2$)<>89 GOTO610
605 PRINT#2\PRINT#2
610 GOTO660
640 PRINT\PRINT'END OF DATA FILE REACHED AT ANALYSIS '!N
650 C4=1\CLOSE#1
655 IF N2=0 GOTO1550
660 FOR J=1TON2
665 Y(1,J)=D(1,J)\Y(2,J)=D(3,J)\Y(3,J)=D(4,J)
670 Y(4,J)=D(6,J)\Y(5,J)=D(10,J)\Y(6,J)=D(11,J)
675 Y(7,J)=D(14,J)\Y(8,J)=D(15,J)\Y(9,J)=D(2,J)
680 Y(10,J)=D(8,J)\Y(11,J)=D(5,J)\Y(12,J)=D(9,J)
685 NEXTJ
710 REM CALCULATE FE2O3 AND FEO
720 FOR J=1TON2\IF Y(3,J)<>0 GOTO760
740 Y(3,J)=Y(4,J)*B7*1.111\Y(4,J)=Y(4,J)-Y(4,J)*B7
760 REM CALCULATE OXIDE SUM AND CATIONS
765 A1=28\Y(13,J)=0\Q(J)=0
790 FOR I=1 TO 12
800 Y(13,J)=Y(13,J)+Y(I,J)
```

```
810 W(I,J)=Y(I,J)/A(I)*B(I)
820 Q(J)=Q(J)+W(I,J)\NEXTI
850 X(J)=A1/Q(J)\REM X=NORMALIZATION FACTOR
870 FOR I=1 TO 12
880 Z(I,J)=X(J)*(Y(I,J)/A(I))*C(I)\NEXTI
920 Y(14,J)=16/2/X(J)*18.016\Y(15,J)=Y(13,J)+Y(14,J)
940 NEXT J
950 REM SUBROUTINE TO CALCULATE SITE OCCUPANCIES
970 GOSUB 1590
980 IF ASC(R3$)<>89 GOTO1340
1000 FOR I=1TON2\PRINT#4,USING"####",F(I)
1010 FOR J=16TO17\PRINT#4,USING Q$,Y(J,I)\NEXTJ
1020 FOR J=19TO29\PRINT#4,USING Q$,Y(J,I)\NEXTJ
1030 NEXTI
1040 IF ASC(Q2$)<>89 GOTO1535
1340 REM PRINT DATA TABLE
1360 PRINT #2,'ANALYSIS';
1370 FOR K=1TON2
1380 PRINT#2,USING '############',F(K);\NEXTK
1400 PRINT #2\PRINT #2
1410 FOR J=1 TO 34
1420 IF J=16 GOTO 1460
1425 IF J=32 GOTO 1460
1430 IF J=33 GOTO 1460
1440 IF J=34 GOTO 1460
1450 GOTO 1470
1460 PRINT #2
1470 GOSUB 1930
1480 PRINT #2, USING "'LLLLLLLLL";X$;
1485 IF J>=16 GOTO1515
1490 FOR L=1 TO N2
1500 PRINT#2,USING "########.##",Y(J,L);\NEXTL
1510 PRINT#2\GOTO1530
1515 FOR L=1TON2\PRINT#2,USING"#######.###",Y(J,L);
1520 NEXTL\PRINT#2
1530 NEXT J
1535 IF C4=1 GOTO1550
1540 GOTO 538
1550 IF ASC(Q2$)=89 THEN PRINT#2,Z$
1560 IF ASC(Q2$)=89 THEN CLOSE #2
1570 IF ASC(R3$)=89 THEN CLOSE#4
1580 GOTO 4000
1590 REM SUBROUTINE TO CALCULATE SITE OCCUPANCIES
1600 FOR J=1 TO N2
1610 Y(16,J)=Z(1,J)
1620 IF Z(1,J)+Z(2,J)>8 GOTO1660
1630 Y(17,J)=Z(2,J)\Y(19,J)=0
1650 GOTO 1680
1660 Y(17,J)=8-Z(1,J)\Y(19,J)=Z(2,J)-Y(17,J)
1680 Y(18,J)=Y(16,J)+Y(17,J)
1690 Y(20,J)=Z(3,J)\Y(21,J)=Z(4,J)
1710 Y(22,J)=Z(5,J)\Y(23,J)=Z(12,J)
1730 Y(24,J)=Z(9,J)\Y(25,J)=Z(10,J)
1750 Y(26,J)=Z(11,J)\Y(27,J)=Z(6,J)
1770 Y(28,J)=Z(7,J)\Y(29,J)=Z(8,J)
1780 Y(30,J)=Y(19,J)+Y(20,J)+Y(21,J)+Y(22,J)+Y(23,J)+Y(24,J)+Y(25,J)
1785 Y(30,J)=Y(30,J)+Y(26,J)+Y(27,J)+Y(28,J)+Y(29,J)
1790 Y(31,J)=Y(18,J)+Y(30,J)\Y(32,J)=16
1820 Y(33,J)=(Y(20,J)+Y(21,J))/(Y(20,J)+Y(21,J)+Y(22,J))
1830 Y(34,J)=Y(22,J)/(Y(20,J)+Y(21,J)+Y(22,J))
1840 NEXT J
1850 RETURN
1930 REM SUBROUTINE TO CHOOSE ROW LABELS FROM A$
1940 B1=0\N1=0
1960 FOR I=1 TO LEN(A$)
1970 S$=SEG$(A$,I,I)
1980 IF S$<>'/' GOTO 2060
1990 N1=N1+1
2000 IF N1=J GOTO 2020
2010 IF N1=J+1 GOTO 2040
2020 B1=I+1
2030 GOTO 2060
2040 E1=I-1
2050 X$=SEG$(A$,B1,E1)
2060 NEXT I
2070 RETURN
4000 END
```

Appendix IV. Example of output from CHLOR

CHLORITE PROGRAM 26-SEP-83 14:55:39

ME-185 CHLORITES

45 CH ME-185 1-1-1
46 CH ME-185 1-2-1
47 CH ME-185 1-3-1
48 CH ME-185 1-5-1
49 CH ME-185 1-6-1
50 CH ME-185 1-7-1

ANALYSIS	45	46	47	48	49	50
SIO2	26.47	26.32	25.03	25.15	25.50	25.26
AL203	20.62	21.24	21.16	21.10	21.96	21.48
FE203	0.00	0.00	0.00	0.00	0.00	0.00
FEO	25.33	25.00	26.32	25.67	24.98	25.54
MGO	14.58	13.87	14.23	14.53	13.60	14.65
CAO	0.00	0.04	0.00	0.00	0.06	0.02
NA2O	0.01	0.02	0.02	0.01	0.01	0.04
K2O	0.00	0.00	0.00	0.00	0.00	0.00
TIO2	0.05	0.03	0.00	0.02	0.07	0.07
MNO	0.63	0.62	0.68	0.63	0.64	0.61
CR2O3	0.00	0.00	0.00	0.00	0.00	0.00
NIO	0.00	0.00	0.00	0.00	0.00	0.00
SUM	87.69	87.14	87.44	87.11	86.82	87.67
H2O CALC	11.39	11.34	11.25	11.25	11.28	11.34
SUM	99.08	98.48	98.69	98.36	98.10	99.01
SI	5.575	5.566	5.338	5.363	5.421	5.343
AL	2.425	2.434	2.662	2.637	2.579	2.657
Z	8.000	8.000	8.000	8.000	8.000	8.000
AL	2.696	2.861	2.658	2.667	2.924	2.699
FE+3	0.000	0.000	0.000	0.000	0.000	0.000
FE+2	4.462	4.421	4.694	4.578	4.441	4.518
MG	4.577	4.371	4.523	4.618	4.309	4.618
NI	0.000	0.000	0.000	0.000	0.000	0.000
TI	0.008	0.005	0.000	0.003	0.011	0.011
MN	0.112	0.111	0.123	0.114	0.115	0.109
CR	0.000	0.000	0.000	0.000	0.000	0.000
CA	0.000	0.009	0.000	0.000	0.014	0.005
NA	0.004	0.008	0.008	0.004	0.004	0.016
K	0.000	0.000	0.000	0.000	0.000	0.000
Y	11.859	11.786	12.006	11.984	11.818	11.976
CATSUM	19.859	19.786	20.006	19.984	19.818	19.976
(OH)	16.000	16.000	16.000	16.000	16.000	16.000
FE:FE+MG	0.494	0.503	0.509	0.498	0.508	0.495
MG:MG+FE	0.506	0.497	0.491	0.502	0.492	0.505

Appendix V. Listing of CORD

```
100 PRINT'PROGRAM CORD.BAS FOR RECALCULATION OF CORDIERITE ANALYSES'
110 REM M.FLOHR 10/82!MODIFIED 6/29/83
120 DIM A$(129),A(18),B(18),C(18),D(18,10),T$(48),W(18,10)
130 DIM S(10),Q(10),X(10),Z(18,10),Y(28,10),F(10)
170 READ A$
180 DATA /SIO2/AL2O3/FE2O3/FEO/MGO/CAO/NA2O/K2O/TIO2/MNO/
210 DATA SUM/SI/AL/ Z/AL/TI/FE+3/ CY]4/MG/FE+2/MN/CA/
230 DATA NA/K/ CXY]6/CATSUM/FE:FE+MG/MG:MG+FE/
260 FOR I=1 TO 2
270 READ D$
280 A$=A$+D$
290 NEXT I
300 REM A=FORMULA WT OXIDES,B=#ANIONS/OXIDE FORMULA
310 REM C=#CTIONS/OXIDE FORMULA OR ANIONS
320 FOR I=1TO18\READ A(I)
330 DATA 60.09,79.9,101.96,159.7,151.99,71.85,81.37,70.94,74.69,40.31,
340 DATA 56.08,103.62,153.34,61.98,94.2,141.944,19,35.45
345 NEXT I
350 FOR I=1TO18\READ B(I)
360 DATA 2,2,3,3,3,1,1,1,1,1,1,1,1,1,1,5,1,1
365 NEXT I
367 FOR I=1TO18\READ C(I)
380 DATA 1,1,2,2,2,1,1,1,1,1,1,1,1,2,2,2,1,1
385 NEXTI
390 PRINT'ENTER RUN TITLE!'\INPUTB$
400 PRINT'ENTER FE+3/FE TOTAL= '!\INPUTB7
410 PRINT'ENTER NAME OF DATA FILE AS SYn!XXXXXX.YYY'\INPUTN$
420 OPEN N$ FOR INPUT AS FILE#1
430 PRINT'ARE YOU SORTING?(Y OR N): '!\INPUT S8$
435 IF ASC(S8$)<>89 GOTO450
440 PRINT'ENTER 2 CHARACTER SORT STRING '!\INPUTM2$
450 PRINT'DO YOU WANT TO CREATE A DISK OUTPUT FILE WITH SITE OCCUPANCY'
460 PRINT'CATIONS?(Y OR N)'\INPUTR3$
470 IF ASC(R3$)<>89 GOTO485
475 PRINT'ENTER OUTPUT FILE NAME AS SYN:XXXXXX.YYY'\INPUTN3$
480 OPEN N3$ FOR OUTPUT AS FILE#4\PRINT#4,N$
485 PRINT'KEYPAD EDITOR ERROR CHECKING?(Y OR N)'\INPUTR$
490 PRINT'PRINT DATA TABLE?(Y OR N)'\INPUT Q2$
495 IF ASC(Q2$)=89 THEN OPEN 'LP:' FOR OUTPUT AS FILE#2
500 INPUT#1,I9\FORJ=1TO18\INPUT#1,A5$\NEXTJ
510 Z$=CHR$(12)\Q$='##.###'
520 IF ASC(Q2$)<>89 GOTO535
525 PRINT#2,'CORDIERITE PROGRAM',DAT$,CLK$
530 PRINT#2\PRINT#2,B$\PRINT#2
535 N=0\C4=0\REM N IS COUNTER FOR TOTAL #ANALYSESE IN FILE;C4 FOR END OF FILE
536 GOTO540
538 IF ASC(Q2$)=89 THEN PRINT#2,Z$
540 N2=0\REM N2 COUNTER FOR #ANALYSES ACCEPTED FOR CALCULATION AT ONE TIME
550 IF END#1 GOTO640
555 N=N+1
560 IF ASC(R$)=89 THEN PRINT 'N= '!N
565 INPUT#1,F8,T$\INPUT#1,T5$,F9
570 IF ASC(R$)=89 THEN PRINT F8!T$
573 IF ASC(S8$)<>89 GOTO590
575 M8$=SEG$(T$,1,2)\IF M8$=M2$ GOTO590
580 FOR J=1TO18\INPUT#1,D6,D7\NEXTJ
585 IF N2<10 GOTO550\GOTO620
590 N2=N2+1\F(N2)=F8\IF ASC(Q2$)=89 THEN PRINT#2,F(N2)!T$
595 S(N2)=F9
600 FOR J=1TO18\INPUT#1,D(J,N2),Z(J,N2)\NEXTJ
610 IF N2<10 GOTO550
620 IF ASC(Q2$)<>89 GOTO630
625 PRINT#2\PRINT#2
630 GOTO660
640 PRINT\PRINT'END OF DATA FILE REACHED AT ANALYSIS '!N
645 C4=1\CLOSE#1
650 IF N2=0 GOTO1550
660 FOR J=1TON2
670 Y(1,J)=D(1,J)\Y(2,J)=D(3,J)\Y(3,J)=D(4,J)
675 Y(4,J)=D(6,J)\Y(5,J)=D(10,J)\Y(6,J)=D(11,J)
680 Y(7,J)=D(14,J)\Y(8,J)=D(15,J)\Y(9,J)=D(2,J)
685 Y(10,J)=D(8,J)\NEXTJ
690 FOR J=1TON2\REM CALCULATE FE2O3!OXIDE SUM
700 IF Y(3,J)<>0 GOTO720
710 Y(3,J)=Y(4,J)*B7*1.111\Y(4,J)=Y(4,J)-Y(4,J)*B7
720 Y(11,J)=0\FOR I=1TO10
730 Y(11,J)=Y(11,J)+Y(I,J)\NEXTI
740 NEXTJ
750 REM CALCULATE SITE OCCUPANCIES
760 FOR J=1TON2
```

```
770 IF $(J)<>18 GOTO940
780 IF B7<>0 GOTO940
790 Y(12,J)=Z(2,J)\IF Z(1,J)+Z(3,J)>6 GOTO820
800 Y(13,J)=Z(3,J)\Y(15,J)=0
810 GOTO 830
820 Y(13,J)=6-Z(1,J)\Y(15,J)=Z(3,J)-Y(13,J)
830 Y(14,J)=Y(12,J)+Y(13,J)
840 Y(16,J)=Z(2,J)\Y(17,J)=Z(4,J)
850 Y(18,J)=Y(15,J)+Y(16,J)+Y(17,J)
860 Y(19,J)=Z(10,J)\Y(20,J)=Z(6,J)
870 Y(21,J)=Z(8,J)\Y(22,J)=Z(11,J)
880 Y(23,J)=Z(14,J)\Y(24,J)=Z(15,J)
890 Y(25,J)=Y(19,J)+Y(20,J)+Y(21,J)+Y(22,J)+Y(23,J)+Y(24,J)
900 Y(26,J)=Y(14,J)+Y(18,J)+Y(25,J)
910 Y(27,J)=Y(17,J)+Y(20,J)/(Y(17,J)+Y(20,J)+Y(19,J))
920 Y(28,J)=Y(19,J)/(Y(17,J)+Y(20,J)+Y(19,J))
930 GOTO1000
935 REM CALCULATE CATIONS IF $OXY READ FROM DATA FILE NOT EQUAL TO 18
936 REM OR IF CALCULATING FE+3
940 Q(J)=0\FOR I=1TO18
950 W(I,J)=D(I,J)/A(I)*B(I)
960 Q(J)=Q(J)+W(I,J)\NEXTI
970 X(J)=18/Q(J)\REM X=NORMALIZATION FACTOR
980 FOR I=1TO18\Z(I,J)=X(J)*(D(I,J)/A(I))*C(I)
990 NEXTI\GOTO790
1000 NEXTJ
1050 IF ASC(RJ$)<>89 GOTO1330
1070 FOR I=1TON2
1080 PRINT#4,USING"####",F(I)
1090 FOR J=12TO13\PRINT#4,USING Q$,Y(J,I)\NEXTJ
1100 FOR J=15TO17\PRINT#4,USING Q$,Y(J,I)\NEXTJ
1110 FOR J=19TO24\PRINT#4,USING Q$,Y(J,I)\NEXTJ
1120 NEXTI
1330 IF ASC(Q2$)<>89 GOTO1530
1340 REM PRINT DATA TABLE
1360 PRINT#2\PRINT#2,'ANALYSIS';
1370 FOR K=1TON2
1380 PRINT#2,USING "##########",F(K);\NEXTK
1400 PRINT#2\PRINT#2
1410 FOR J=1TO28
1420 IF J=12 GOTO1460
1430 IF J=27 GOTO1460
1440 IF J=28 GOTO1460
1450 GOTO1470
1460 PRINT#2
1470 GOSUB1930
1480 PRINT#2,USING"LLLLLLLL",X$;
1485 IF J>=12 GOTO1515
1490 FOR L=1TON2
1500 PRINT#2,USING"########.##",Y(J,L);\NEXTL
1505 GOTO1525
1515 FOR L=1TON2
1520 PRINT#2,USING"#######.###",Y(J,L);\NEXTL
1525 PRINT#2\NEXTJ
1530 IF C4=1 GOTO1550
1535 GOTO538
1550 IF ASC(Q2$)=89 THEN PRINT#2,Z$
1560 IF ASC(Q2$)=89 THEN CLOSE #2
1570 IF ASC(R3$)=89 THEN CLOSE#4
1580 GOTO4000
1930 REM SUBROUTINE TO CHOOSE ROW LABELS FROM A$
1940 B1=0\N1=0
1960 FOR I=1 TO LEN(A$)
1970 S$=SEG$(A$,I,I)
1980 IF S$<>'/' GOTO2060
1990 N1=N1+1
2000 IF N1=J GOTO2020
2010 IF N1=J+1 GOTO2040
2020 B1=I+1
2030 GOTO2060
2040 E1=I-1
2050 X$=SEG$(A$,B1,E1)
2060 NEXTI
2070 RETURN
4000 END
```

Appendix V. Example of output from CORD

CORDIERITE PROGRAM 26-SEP-83 14:13:04

S67-52 CORDIERITE

15 CD S67-52 5-8-1
16 CD S67-52 5-9-1
17 CD S67-52 5-10-1
26 CD S67-52 6-4-1

ANALYSIS	15	16	17	26
SIO2	50.80	49.93	50.87	50.44
AL2O3	33.04	33.49	32.99	33.33
FE2O3	0.00	0.00	0.00	0.00
FEO	4.03	3.72	3.82	3.93
MGO	10.84	10.88	10.65	10.89
CAO	0.00	0.01	0.02	0.00
NA2O	0.35	0.25	0.49	0.32
K2O	0.00	0.00	0.00	0.00
TIO2	0.02	0.04	0.00	0.14
MNO	0.46	0.49	0.44	0.49
SUM	99.54	98.81	99.28	99.54
SI	5.071	5.018	5.087	5.038
AL	0.929	0.982	0.913	0.962
Z	6.000	6.000	6.000	6.000
AL	2.958	2.985	2.975	2.961
TI	0.002	0.003	0.000	0.010
FE+3	0.000	0.000	0.000	0.000
[Y]4	2.960	2.988	2.975	2.971
MG	1.613	1.629	1.587	1.621
FE+2	0.337	0.313	0.320	0.328
MN	0.039	0.041	0.037	0.041
CA	0.000	0.001	0.002	0.000
NA	0.068	0.049	0.096	0.061
K	0.000	0.000	0.000	0.000
[XY]6	2.057	2.033	2.042	2.051
CATSUM	11.017	11.021	11.017	11.022
FE:FE+MG	0.173	0.161	0.168	0.168
MG:MG+FE	0.827	0.839	0.832	0.832

Appendix VI. Listing of FELD

```
100 PRINT'PROGRAM FELD FOR RECALCULATION OF FELDSPAR ANALYSES'
110 REM M.FLOHR-10/82!MOFIFIED 5/26/83!8/1/83
120 DIM A$(122),A(18),B(18),C(18),D(18,10),T$(48),W(18,10)
130 DIM Y(32,10),Q(10),X(10),Z(18,10),F4(10),R(10),F(10)
170 READ A$
180 DATA /SIO2/AL2O3/FE2O3/FEO/MGO/CAO/NA2O/K2O/TIO2/BAO/SRO/
210 DATA P2O5/SUM/SI/AL/TI/P/FE+3/ T/FE+2/MG/CA/NA/K/BA/SR/ M/
230 DATA CATSUM/AN/AB/OR/CN/
260 FOR I=1 TO 2
270 READ D$
280 A$=A$+D$
290 NEXT I
300 REM A=FORMULA WT OXIDES,  B=# ANIONS/OXIDE FORMULA
310 REM C= # CATIONS/OXIDE FORMULA OR ANIONS
315 FOR I=1TO18\READ A(I)
330 DATA 60.09,79.9,101.96,159.7,151.99,71.85,81.37,70.94,74.69,40.31,
340 DATA 56.08,103.62,153.34,61.98,94.2,141.944,19,35.45
345 NEXT I
347 FOR I=1TO18\READ B(I)
360 DATA 2,2,3,3,3,1,1,1,1,1,1,1,1,1,5,1,1
365 NEXT I
367 FOR I=1TO18\READ C(I)
380 DATA 1,1,2,2,2,1,1,1,1,1,1,1,1,2,2,2,1,1
385 NEXT I
390 Z$=CHR$(12)
395 PRINT'ENTER RUN TITLE:'\INPUTB$
400 PRINT'ENTER FE+3/FE TOTAL= '!\INPUTB7
405 PRINT'ENTER FORMULA BASIS (8 OR 32 OXYGENS): '!\INPUTA1
410 PRINT'ENTER NAME OF DATA FILE AS SYN:XXXXXX:YYY'\INPUT N$
415 PRINT'KEYPAD EDITOR ERROR CHECKING?(Y OR N)'\INPUTR$
420 OPEN N$ FOR INPUT AS FILE #1
425 OPEN 'LP:' FOR OUTPUT AS FILE #2
430 PRINT'ARE YOU SORTING?(Y OR N): '\INPUT S8$
435 IF ASC(S8$)<>89 GOTO520
440 PRINT'YOU CAN ENTER UP TO THREE (3) SORT STRINGS'
445 PRINT'ENTER FIRST 2 CHARACTER SORT STRING '!\INPUT M1$
450 PRINT'ENTER SECOND 2 CHARACTER SORT STRING -- ENTER ZERO (0)'
455 PRINT'IF YOU DO NOT WANT TO SORT A 2ND GROUP OF ANALYSES: '!
460 INPUT M2$\IF M2$<>'0' THEN GOTO 475
465 M3$='0'\GOTO520
475 PRINT'ENTER THIRD 2 CHARACTER SORT STRING -- ENTER ZERO (0)'
480 PRINT'IF YOU DO NOT WANT TO SORT A 3RD GROUP OF ANALYSES: '!\INPUTM3$
520 INPUT#1,Q2
525 FOR J=1TO18\INPUT#1,A$\NEXTJ
534 PRINT#2,'FELDSPAR PROGRAM',DAT$,CLK$
535 PRINT#2\PRINT#2,B$\PRINT#2
536 N=0\C4=0\REM N IS COUNTER FOR TOTAL #ANALYSES IN FILE; C4 FOR END OF FILE
537 GOTO 539
538 PRINT#2,Z$\PRINT#2
539 N2=0\REM N2 IS COUNTER FOR #ANALYSES ACCEPTED FOR CALCULATION AT ONE TIME
550 IF END #1 THEN GOTO 650
555 N=N+1\IF ASC(R$)=89 THEN PRINT 'N= ';N
560 INPUT #1,F8,T$\IF ASC(R$)=89 THEN PRINT F8;T$
565 IF ASC(S8$)<>89 GOTO593
570 M8$=SEG$(T$,1,2)\IF M8$=M1$ GOTO593
572 IF M2$='0' GOTO580
574 IF M8$=M2$ GOTO593
576 IF M3$='0' GOTO580
578 IF M8$=M3$ GOTO593
580 INPUT#1,T5$,F9
585 FOR J=1TO18\INPUT#1,D6,D7\NEXTJ
590 IF N2<10 GOTO550\GOTO625
593 N2=N2+1\INPUT#1,T5$,F4(N2)\F(N2)=F8
595 PRINT#2,F(N2);T$\FOR J=1TO18\INPUT#1,D(J,N2),Z(J,N2)
600 NEXTJ\IF N2<10 GOTO 550
625 PRINT#2\PRINT#2\GOTO670
650 PRINT\PRINT'END OF DATA FILE REACHED AT ANALYSIS ';N
651 C4=1\CLOSE#1\IF N2=0 GOTO1570
670 FOR J=1TON2\Y(1,J)=D(1,J)\Y(2,J)=D(3,J)
673 Y(3,J)=D(4,J)\Y(4,J)=D(6,J)\Y(5,J)=D(10,J)
675 Y(6,J)=D(11,J)\Y(7,J)=D(14,J)\Y(8,J)=D(15,J)
678 Y(9,J)=D(2,J)\Y(10,J)=D(13,J)\Y(11,J)=D(12,J)
680 Y(12,J)=D(16,J)\NEXTJ\FORJ=1TON2
682 IF F4(J)=A1 GOTO740
685 IF D(4,J)<>0 GOTO700\REM CALCULATE FE2O3
690 D(4,J)=D(6,J)*B7*1.111\D(6,J)=D(6,J)-D(6,J)*B7
695 Y(3,J)=D(4,J)\Y(4,J)=D(6,J)
700 Q=0\REM CALCULATE CATIONS IF A1<>F4(N2)
705 FOR I=1TO18\Z(I,J)=0\NEXTI
710 FOR I=1TO18\W(I,J)=D(I,J)/A(I)*B(I)
715 Q(J)=Q(J)+W(I,J)\NEXTI
720 X(J)=A1/Q(J)\REM X=NORMALIZATION FACTOR
```

```
725 FOR I=1TO18\Z(I,J)=X(J)*(D(I,J)/A(I))*C(I)\NEXT I
730 GOTO 750
740 IF B7<>0 GOTO685
750 Y(13,J)=0\FOR I=1TO12\Y(13,J)=Y(13,J)+Y(I,J)\NEXTI
770 Y(14,J)=Z(1,J)\Y(15,J)=Z(3,J)
780 Y(16,J)=Z(2,J)\Y(17,J)=Z(16,J)\Y(18,J)=Z(4,J)
790 Y(19,J)=Y(14,J)+Y(15,J)+Y(16,J)+Y(17,J)+Y(18,J)
800 Y(20,J)=Z(6,J)\Y(21,J)=Z(10,J)\Y(22,J)=Z(11,J)
810 Y(23,J)=Z(14,J)\Y(24,J)=Z(15,J)\Y(25,J)=Z(13,J)\Y(26,J)=Z(12,J)
820 Y(27,J)=Y(20,J)+Y(21,J)+Y(22,J)+Y(23,J)+Y(24,J)+Y(25,J)+Y(26,J)
830 Y(28,J)=Y(27,J)+Y(19,J)\R(J)=Y(22,J)+Y(23,J)+Y(24,J)+Y(25,J)
840 Y(29,J)=(Y(22,J)/R(J))*100\Y(30,J)=(Y(23,J)/R(J))*100
850 Y(31,J)=(Y(24,J)/R(J))*100\Y(32,J)=(Y(25,J)/R(J))*100\NEXTJ
1340 REM PRINT DATA TABLE
1360 PRINT #2, 'ANALYSIS';\FOR K=1TON2
1380 PRINT #2, USING "############",F(K);\NEXTK
1400 PRINT #2\PRINT #2\FOR J=1TO32
1420 IF J=14 GOTO 1460\IF J=29 GOTO1460
1450 GOTO 1470
1460 PRINT #2
1470 GOSUB 1930
1480 PRINT #2, USING "LLLLLLLL";X$;
1485 IF J<=13 GOTO1510\IF J>=29 GOTO1510
1490 FOR L=1 TO N2
1500 PRINT#2,USING"#######.###",Y(J,L);
1505 NEXTL\GOTO 1520
1510 FOR L=1TON2
1515 PRINT#2,USING"########.##",Y(J,L);\NEXTL
1520 PRINT#2\NEXTJ
1530 IF C4=1 GOTO 1570
1560 GOTO 538
1570 PRINT#2,Z$\CLOSE #2
1580 GOTO 4000
1930 REM SUBROUTINE TO CHOOSE ROW LABELS FROM A$
1940 B1=0
1950 N1=0
1960 FOR I=1 TO LEN(A$)
1970 S$=SEG$(A$,I,I)
1980 IF S$<>"/" GOTO 2060
1990 N1=N1+1
2000 IF N1=J GOTO 2020
2010 IF N1=J+1 GOTO 2040
2020 B1=I+1
2030 GOTO 2060
2040 E1=I-1
2050 X$=SEG$(A$,B1,E1)
2060 NEXT I
2070 RETURN
4000 END
```

Appendix VI. Example of output from FELD

FELDSPAR PROGRAM 26-SEP-83 14:29:52

S68-51 PLAGIOCLASE

34 PL S68-51 GR.1
35 PL S68-51 GR.2
36 PL S68-51 GR.3
37 PL S68-51 GR.4
38 PL S68-51 GR.5
39 PL S68-51 GR.6
40 PL S68-51 GR.7
41 PL S68-51 GR.8
42 PL S68-51 GR.9
43 PL S68-51 GR.10

ANALYSIS	34	35	36	37	38	39	40	41	42	43
SIO2	50.78	51.92	52.35	51.73	50.60	51.74	54.21	52.12	51.54	53.23
AL2O3	31.72	31.24	30.24	30.42	31.71	30.78	29.67	30.70	31.52	30.24
FE2O3	0.00	0.00	0.00	0.00	0.00	0.00	0.00	0.00	0.00	0.00
FEO	0.00	0.00	0.00	0.00	0.00	0.00	0.00	0.00	0.00	0.00
MGO	0.00	0.00	0.00	0.00	0.00	0.00	0.00	0.00	0.00	0.00
CAO	14.53	13.59	12.57	12.88	14.51	12.98	11.59	12.50	13.93	12.68
NA2O	3.17	3.76	4.45	3.96	3.37	4.14	4.79	4.25	3.51	4.47
K2O	0.03	0.01	0.02	0.05	0.08	0.03	0.02	0.06	0.00	0.02
TIO2	0.02	0.01	0.01	0.01	0.00	0.02	0.00	0.01	0.00	0.00
BAO	0.00	0.00	0.00	0.00	0.00	0.00	0.00	0.00	0.00	0.00
SRO	0.00	0.00	0.00	0.00	0.00	0.00	0.00	0.00	0.00	0.00
P2O5	0.00	0.00	0.00	0.00	0.00	0.00	0.00	0.00	0.00	0.00
SUM	100.25	100.53	99.64	99.05	100.27	99.69	100.28	99.64	100.50	100.64
SI	9.215	9.371	9.520	9.464	9.193	9.415	9.749	9.471	9.311	9.577
AL	6.784	6.645	6.481	6.558	6.790	6.601	6.288	6.576	6.711	6.413
TI	0.003	0.001	0.002	0.001	0.000	0.002	0.000	0.001	0.000	0.000
P	0.000	0.000	0.000	0.000	0.000	0.000	0.000	0.000	0.000	0.000
FE+3	0.000	0.000	0.000	0.000	0.000	0.000	0.000	0.000	0.000	0.000
T	16.002	16.017	16.003	16.023	15.983	16.018	16.037	16.048	16.022	15.990
FE+2	0.000	0.000	0.000	0.000	0.000	0.000	0.000	0.000	0.000	0.000
MG	0.000	0.000	0.000	0.000	0.000	0.000	0.000	0.000	0.000	0.000
CA	2.826	2.628	2.449	2.525	2.825	2.530	2.233	2.434	2.696	2.444
NA	1.115	1.316	1.568	1.404	1.188	1.459	1.670	1.498	1.230	1.561
K	0.007	0.003	0.005	0.013	0.020	0.008	0.005	0.015	0.000	0.005
BA	0.000	0.000	0.000	0.000	0.000	0.000	0.000	0.000	0.000	0.000
SR	0.000	0.000	0.000	0.000	0.000	0.000	0.000	0.000	0.000	0.000
M	3.948	3.947	4.022	3.942	4.033	3.997	3.908	3.947	3.926	4.010
CATSUM	19.950	19.964	20.025	19.965	20.016	20.015	19.945	19.995	19.948	20.000
AN	71.58	66.58	60.89	64.05	70.05	63.30	57.14	61.67	68.67	60.95
AB	28.24	33.34	38.99	35.62	29.46	36.50	42.73	37.95	31.33	38.93
OR	0.18	0.08	0.12	0.33	0.50	0.20	0.13	0.38	0.00	0.12
CN	0.00	0.00	0.00	0.00	0.00	0.00	0.00	0.00	0.00	0.00

Appendix VII. Listing of GARN

```
100 PRINT'PROGRAM GARN.BAS FOR NORM CALCULATION OF GARNET ANALYSES'
105 PRINT'MAIN PROGRAM'\REM M.FLOHR 3/83-MODIFIED 6/83
110 PRINT'NO.OXYGENS=12 FOR CATION CALCULATIONS'
112 REM ORDER OF END MEMBER CALC. FR. RICKWOOD(1968)
115 REM N-COUNTER FOR TOTAL# ANALYSES IN FILE!C4 CHECK FOR END OF FILE
120 DIM A$(107),A(10),B(10),C(4),D(18,4),T$(48),W(26,4),P(4),F6(4)
130 DIM M(4),H(4),N(4),D2(18,4),O(26,4),Q$(39),I$(35)
140 DIM R(10,4),Y(4),F(4),U(4),E(10,4),G(10,4),V(4)
160 READ A$
170 DATA /SIO2/TIO2/AL2O3/CR2O3/FE2O3/FEO/MNO/MGO/CAO/NA2O/
180 DATA SUM/SI/AL/ Z/AL/CR/FE+3/TI/ Y/FE+2/MN/MG/CA/NA/ X/
190 DATA CATSUM/
200 FOR I=1TO2\READ D$
220 A$=A$+D$\NEXTI
240 READ Q$
250 DATA /%CAT ALLOC/UVAR/AND/PYR/SPES/GROS/ALM/
260 READ I$
270 DATA /SI/TI/AL/CR/FE+3/FE+2/MN/MG/CA/NA/
300 REM A=FORM. WT. OX.!B=ANIONS/OX. FORM.
315 FOR I=1TO10\READ A(I)
330 DATA 60.09,79.9,50.98,75.995,79.85,71.85,70.94,40.32,56.08,30.99
345 NEXT I
350 FOR I=1TO10\READ B(I)
355 DATA 2,2,1.5,1.5,1.5,1,1,1,1,0.5
360 NEXT I
365 C4=0\N=0\Q5$='#.###'\Q4$='##.##'\Q3$=CHR$(44)
370 PRINT'ENTER RUN TITLE:'\INPUTB$
380 PRINT'ENTER NAME OF DATA FILE AS SYn:XXXXXX.YYY'\INPUTN$
390 PRINT'CHECK FOR KEYPAD EDITOR ERRORS? (Y OR N)'!\INPUTR$
392 PRINT'ARE YOU SORTING?(Y OR N)'!\INPUTS8$
395 IF ASC(S8$)<>89 GOTO400
397 PRINT'ENTER 2 CHARACTER SORT STRING! '!\INPUTM8$
400 PRINT'PRINT DATA TABLE? (Y OR N)'!\INPUT R7$
405 PRINT'DO YOU WANT TO CREATE A DISK FILE CONTAINING CALCULATED'
410 PRINT'END MEMBERS? (Y OR N)'\INPUTP$
415 IF ASC(P$)<>89 GOTO425
420 PRINT'ENTER OUTPUT FILE NAME AS SYN:XXXXXX:YYY'\INPUTP8$
423 OPEN P8$ FOR OUTPUT AS FILE#4\PRINT#4,N$
425 PRINT'DO YOU WANT TO CREATE A DISK FILE CONTAINING RECALCULATED'
430 PRINT'OXIDES + CATIONS? (Y OR N)'\INPUTP9$
435 IF ASC(P9$)<>89 GOTO450
440 PRINT'ENTER OUTPUT FILE NAME AS SYN:XXXXXX:YYY'\INPUTP7$
445 OPEN P7$ FOR OUTPUT AS FILE#3\PRINT#3,N$
450 OPEN N$ FOR INPUT AS FILE #1
455 IF ASC(R7$)=89 THEN OPEN 'LP:' FOR OUTPUT AS FILE #2
457 INPUT#1,B6
460 Z$=CHR$(12)\FORJ=1TO18\INPUT#1,A5$\NEXTJ
465 IF ASC(R7$)=89 THEN PRINT#2,B$\GOTO480
470 IF ASC(R7$)=89 THEN PRINT#2,Z$
480 N2=0
485 IF END#1 GOTO540\N=N+1
490 INPUT#1,Q5,T$\INPUT#1,T5$,F9
492 IF ASC(R$)=89 THEN PRINT Q5!T$
495 IF ASC(S8$)<>89 GOTO520
500 M5$=SEG$(T$,1,2)\IF M5$=M8$ GOTO520
505 FOR J=1TO18\INPUT#1,H5,H3\NEXTJ
510 IF N2<4 GOTO 485\GOTO550
520 N2=N2+1\F(N2)=Q5\IF ASC(R7$)=89 THEN PRINT#2, F(N2)!T$
525 FOR J=1TO18\INPUT#1,D(J,N2),D2(J,N2)\NEXTJ
530 IF N2<4 GOTO485\GOTO550
540 PRINT\PRINT'END OF DATA FILE REACHED AT ANALYSIS '!N
545 C4=1\CLOSE#1
547 IF N2=0 GOTO4095\REM MATRIX D=OXIDES+CATIONS IN SITES TO BE PRINTED
550 FOR J=1TON2\O(1,J)=D(1,J)\O(2,J)=D(2,J)\O(3,J)=D(3,J)
555 O(4,J)=D(5,J)\O(5,J)=D(4,J)\O(6,J)=D(6,J)\O(7,J)=D(8,J)
560 O(8,J)=D(10,J)\O(9,J)=D(11,J)\O(10,J)=D(14,J)\NEXTJ
565 FORJ=1TON2\O(11,J)=0\FOR I=1TO10\O(11,J)=O(11,J)+O(I,J)
570 NEXTI\NEXTJ\FOR J=1TON2\O(12,J)=D2(1,J)\IF O(12,J)>3 GOTO580
575 O(13,J)=3-D2(1,J)\O(15,J)=D2(3,J)-O(13,J)\GOTO585
580 O(13,J)=0\O(15,J)=D2(3,J)
585 O(14,J)=O(12,J)+O(13,J)\O(16,J)=D2(5,J)\O(17,J)=D2(4,J)
590 O(18,J)=D2(2,J)\O(19,J)=O(15,J)+O(16,J)+O(17,J)+O(18,J)
595 O(20,J)=D2(6,J)\O(21,J)=D2(8,J)\O(22,J)=D2(10,J)
600 O(23,J)=D2(11,J)\O(24,J)=D2(14,J)
605 O(25,J)=O(20,J)+O(21,J)+O(22,J)+O(23,J)+O(24,J)
610 O(26,J)=O(14,J)+O(19,J)+O(25,J)\NEXTJ
615 IF ASC(R7$)<>89 GOTO706
620 PRINT#2,'ANALYSIS      '!\FOR I=1TON2\IF I=1 GOTO630
625 PRINT#2,USING"##########################",F(I)!\GOTO635
630 PRINT#2,USING"##############",F(I)!
```

```
635 NEXT I
640 PRINT#2\PRINT#2,'                '
645 FOR I=1TON2\PRINT#2,'        INPUT        NORMALIZED'\NEXTI\PRINT#2
706 OVERLAY '$Y0:GARN2.BAS' LINE 707
2050 PRINT'BACK IN MAIN PROGRAM'
4085 IF C4=1 GOTO 4095
4087 IF ASC(R7$)=89 THEN PRINT#2,Z$\GOTO 480
4095 IF ASC(P9$)=89 THEN CLOSE#3
4096 IF ASC(P$)=89 THEN CLOSE#4
4098 IF ASC(R7$)<>89 GOTO9000
5000 PRINT#2,Z$\CLOSE#2
9000 END
```

```
707 PRINT'GARN2 -- SEGMENT #2 FOR GARNET PROGRAM'
708 REM M.FLOHR 3/83--NORMALIZE CATIONS TO 8
710 FOR J=1TON2\C1=0\C1=8/O(26,J)\H(J)=0\E(2,J)=O(18,J)*C1
730 E(1,J)=O(12,J)*C1\E(3,J)=(O(13,J)+O(15,J))*C1
750 FORI=4TO5\E(I,J)=O(I+12,J)*C1\NEXTI
770 FORI=6TO10\E(I,J)=O(I+14,J)*C1\NEXTI
780 FOR I=1TO10\G(I,J)=E(I,J)*B(I)\H(J)=H(J)+G(I,J)\NEXTI
800 L5(J)=((2*(12-H(J)))/E(1,J))*O(1,J)\E2(J)=E(6,J)
810 IF O(5,J)<>0 GOTO 850\IF O(26,J)<=8 GOTO850
830 REM TEST IF ENOUGH FE TO FORM FE+3
840 IF E(6,J)>=2*(12-H(J)) GOTO860
850 W(5,J)=O(5,J)\W(6,J)=O(6,J)\GOTO890
860 E(5,J)=2*(12-H(J))\E(6,J)=E(6,J)-E(5,J)
870 C(J)=0\C(J)=O(6,J)/E2(J)
880 W(6,J)=C(J)*E(6,J)\W(5,J)=C(J)*(E(5,J)*1.111)
890 H5(J)=0\FOR I=1TO10\G(I,J)=E(I,J)*B(I)
910 H5(J)=H5(J)+G(I,J)\NEXT I\REM CALCULATE NEW SITE OCCUPANCIES
930 W(12,J)=E(1,J)\IF E(1,J)>=3GOTO980
940 W(13,J)=3-E(1,J)\W(14,J)=W(12,J)+W(13,J)
960 W(15,J)=E(3,J)-W(13,J)
970 GOTO 1010
980 W(13,J)=0
1000 W(14,J)=W(12,J)+W(13,J)\W(15,J)=E(3,J)
1010 W(16,J)=E(4,J)\W(17,J)=E(5,J)\W(18,J)=E(2,J)
1030 W(19,J)=W(15,J)+W(16,J)+W(17,J)+W(18,J)
1040 W(20,J)=E(6,J)\W(21,J)=E(7,J)
1050 W(22,J)=E(8,J)\W(23,J)=E(9,J)
1060 W(24,J)=E(10,J)
1070 W(25,J)=W(20,J)+W(21,J)+W(22,J)+W(23,J)+W(24,J)
1080 W(26,J)=W(14,J)+W(19,J)+W(25,J)
1090 NEXT J
1100 REM CALC. NEW OXIDES AND CATION PROPORTIONS
1110 FOR J=1 TO N2\W(11,J)=0
1120 FOR I=1 TO 4\W(I,J)=O(I,J)
1140 W(11,J)=W(11,J)+W(I,J)\NEXTI
1150 FOR I=7 TO 10
1170 W(I,J)=O(I,J)
1180 W(11,J)=W(11,J)+W(I,J)\NEXTI
1200 REM
1210 W(11,J)=W(11,J)+W(5,J)+W(6,J)\C(J)=0
1230 FOR I=1 TO 10
1240 R(I,J)=W(I,J)/A(I)
1250 C(J)=C(J)+R(I,J)
1260 NEXT I\NEXT J
1270 REM MATRICES O AND W CONTAIN ALL INPUT AND CALCULATED
1280 REM OXIDE AND CATION VALUES
1290 REM-OUTPUT CALCULATED OXIDES + CATIONS TO DISK FILE
1300 IF ASC(P9$)<>89 GOTO1420\FOR J=1TON2\PRINT#3,USING"####",F(J)
1310 PRINT#3,USING Q4$,W(1,J);\PRINT#3,Q3$;\PRINT#3,USING Q5$,W(12,J)
1320 PRINT#3,USING Q4$,W(2,J);\PRINT#3,Q3$;\PRINT#3,USING Q5$,W(18,J)
1340PRINT#3,USING Q4$,W(3,J);\PRINT#3,Q3$;\PRINT#3,USING Q5$,(W(13,J)+W(15,J))
1350 FOR I=4TO5\PRINT#3,USING Q4$,W(I,J);\PRINT#3,Q3$;
1370 PRINT#3,USING Q5$,W(12+I,J)\NEXTI
1380 FOR I=6TO10\PRINT#3,USING Q4$,W(I,J);\PRINT#3,Q3$;
1390 PRINT#3,USING Q5$,W(14+I,J)\NEXTI\NEXTJ
1410 REM
1420 IF ASC(R7$)<>89GOTO1600
1440 REM PRINT ALL OXIDES+CATIONS
```

```
1450 FOR J=1 TO 26
1460 IF J=12 GOTO 1480
1470 GOTO 1490
1480 PRINT#2
1490 GOSUB 1630
1500 PRINT#2,USING"'LLLLLLLLLL",X$;\IF J>=12 GOTO1550
1520 FOR K=1TON2\PRINT#2,USING"##########.##",O(J,K);
1540 PRINT#2,USING"##########.##",W(J,K);\NEXTK\PRINT#2\GOTO1570
1550 FOR K=1TON2\PRINT#2,USING"##########.###",O(J,K);
1560 PRINT#2,USING"##########.###",W(J,K);\NEXTK\PRINT#2
1570 NEXT J
1600 OVERLAY "SY0:GARN3.BAS" LINE 707
1620 REM SUBROUTINE TO CHOOSE ROW LABELS FROM A$
1630 B1=0\N1=0
1640 FOR I=1 TO LEN(A$)
1650 S$=SEG$(A$,I,I)
1660 IF S$<>'/' GOTO1800
1670 N1=N1+1
1680 IF N1=J GOTO1720
1700 IF N1=J+1 GOTO1750
1720 B1=I+1
1740 GOTO1800
1750 E1=I-1
1780 X$=SEG$(A$,B1,E1)
1800 NEXT I
1810 RETURN
1820 REM
1830 REM
1840 REM
1850 REM
1860 REM
1870 REM
1880 REM
1890 REM
1900 REM
1910 REM
1920 REM
1930REM
```

```
707 PRINT'GARN3-SEGMENT #3 FOR GARNET PROGRAM'
708 REM M.FLOHR 3/83--CALCULATE END MEMBERS
710 FOR J=1 TO N2\REM FORM UVAROVITE V(J)
730 IF R(4,J)<>0 GOTO 750\V(J)=0\GOTO890
750 IF R(9,J)<>0 GOTO 770\V(J)=0\GOTO1080
770 H1=R(4,J)/2\I1=R(9,J)/3\L1=R(1,J)/2
780 IF H1>I1 GOTO810\IF H1>L1 GOTO830
800 V(J)=H1\GOTO 860
810 IF I1>L1 GOTO 850\V(J)=I1\GOTO860
830 IF I1>L1 GOTO 850
840 PRINT'LINE 840 SOMETHING IS WRONG'\REM CHECK IF TESTS FAIL
850 V(J)=L1
860 R(4,J)=R(4,J)-(2*V(J))\R(9,J)=R(9,J)-(3*V(J))
870 R(1,J)=R(1,J)-(3*V(J))
880 REM FORM ANDRADITE Y(J)
890 IF R(9,J)<>0 GOTO 910\Y(J)=0\GOTO1090
910 IF R(5,J)<>0 GOTO 930\Y(J)=0\GOTO1090
930 IF R(1,J)<>0 GOTO 970
940 Y(J)=0\X(J)=0\Q(J)=0\M(J)=0\U(J)=0
960 GO TO 1860
970 H1=R(9,J)/3\K1=R(5,J)/2\L1=R(1,J)/3
980 IF H1>K1 GOTO 1010\IF H1>L1 GOTO1030
1000 Y(J)=H1\GOTO 1060
1010 IF K1>L1 GOTO 1050\Y(J)=K1\GOTO1060
1030 IF K1>L1 GOTO 1050
1040 PRINT'LINE 1040 SOMETHING IS WRONG'\REM CHECK IF TESTS FAIL
1050 Y(J)=L1
```

```
1060 R(9,J)=R(9,J)-(3*Y(J))\R(5,J)=R(5,J)-(2*Y(J))
1070 R(1,J)=R(1,J)-(3*Y(J))
1080 REM FORM PYROPE X(J)
1090 X(J)=0\IF R(8,J)<>0 GOTO 1110
1100 X(J)=0\GOTO 1290
1110 IF R(3,J)<>0 GOTO 1140
1120 X(J)=0\Q(J)=0\M(J)=0\U(J)=0\GOTO1860
1140 IF R(1,J)<>0 GOTO 1170
1150 X(J)=0\Q(J)=0\M(J)=0\U(J)=0\GOTO1860
1170 R1=R(8,J)/3\P1=R(3,J)/2\L1=R(1,J)/3
1180 IF R1>P1 GOTO 1210\IF R1>L1 GOTO1250
1200 X(J)=R1\GOTO 1260
1210 IF R1>L1 GOTO 1250\X(J)=P1\GOTO1260
1230 IF P1>L1 GOTO 1250
1240 PRINT'LINE 1240 SOMETHING IS WRONG'\REM CHECK IF TESTS FAIL
1250 X(J)=L1
1260 R(8,J)=R(8,J)-(3*X(J))
1270 R(3,J)=R(3,J)-(2*X(J))\R(1,J)=R(1,J)-(3*X(J))
1280 REM FORM SPESSERTINE Q(J)
1290 Q(J)=0\IF R(7,J)<>0 GOTO 1310
1300 Q(J)=0\GOTO 1500
1310 IF R(3,J)<>0 GOTO 1340
1320 Q(J)=0\U(J)=0\M(J)=0\GOTO1860
1340 IF R(1,J)<>0 GOTO 1370
1350 Q(J)=0\U(J)=0\M(J)=0\GOTO1860
1370 S1=R(7,J)/3
1380 P1=R(3,J)/2\L1=R(1,J)/3
1390 IF S1>P1 GOTO1420\IF S1>L1 GOTO1440
1410 Q(J)=S1\GOTO1470
1420 IF P1>L1 GOTO1460\Q(J)=P1\GOTO1470
1440 IF P1>L1 GOTO1460
1450 PRINT'LINE 1450 SOMETHING IS WRONG'\REM CHECK IF TESTS FAIL
1460 Q(J)=L1
1470 R(7,J)=R(7,J)-(3*Q(J))
1480 R(3,J)=R(3,J)-(2*Q(J))\R(1,J)=R(1,J)-(3*Q(J))
1490 REM CALCULATE GROSSULAR M(J)
1500 IF R(9,J)<>0 GOTO 1520\M(J)=0\GOTO1670
1520 IF R(3,J)<>0 GOTO 1540\M(J)=0\U(J)=0\GOTO1860
1540 IF R(1,J)<>0 GOTO 1560
1550 M(J)=0\U(J)=0\GOTO 1860
1560 I1=R(9,J)/3\P1=R(3,J)/2\L1=R(1,J)/3
1570 IF I1>P1 GOTO1600\IF I1>L1 GOTO1620\M(J)=I1\GOTO1650
1600 IF P1>L1 GOTO1640\M(J)=P1\GOTO1650
1620 IF P1>L1 GOTO 1640
1630 PRINT'LINE 1630 SOMETHING IS WRONG'\REM CHECK IF TESTS FAIL
1640 M(J)=L1
1650 R(9,J)=R(9,J)-(3*M(J))
1660 R(3,J)=R(3,J)-(2*M(J))\R(1,J)=R(1,J)-(3*M(J))
1670 REM FORM ALMANDINE U(J)
1680 IF R(6,J)<>0 GOTO1700\U(J)=0\GOTO1860
1700 IF R(3,J)<>0 GOTO1720\U(J)=0\GOTO1860
1720 IF R(1,J)<>0 GOTO1740\U(J)=0\GOTO1860
1740 T1=R(6,J)/3\P1=R(3,J)/2\L1=R(1,J)/3
1750 IF T1>P1 GOTO 1780\IF T1>L1 GOTO1800\U(J)=T1\GOTO1830
1780 IF P1>L1 GOTO 1820\U(J)=P1\GOTO1830
1800 IF P1>L1 GOTO 1820
1810 PRINT'LINE 1810 SOMETHING IS WRONG'\REM CHECK IF TESTS FAIL
1820 U(J)=L1
1830 R(6,J)=R(6,J)-(3*U(J))
1850 R(3,J)=R(3,J)-(2*U(J))\R(1,J)=R(1,J)-(3*U(J))
1860 F8(J)=V(J)+Y(J)+X(J)+Q(J)+M(J)+U(J)
1870 REM CALC. %CATIONS ALLOCATED
1880 N(J)=(8*F8(J)*100)/C(J)\A2=100/F8(J)
1890 REM CALC. GARNET MOLECULES
1900 V(J)=V(J)*A2\Y(J)=Y(J)*A2\X(J)=X(J)*A2
1910 Q(J)=Q(J)*A2\M(J)=M(J)*A2\U(J)=U(J)*A2
1920 NEXTJ
1930 OVERLAY"SY0:GARN4.BAS" LINE 707


 707 PRINT'GARN4-SEGMENT#4 FOR GARNET PROGRAM'
 708 REM M.FLOHR 4/83-MODIFIED 6/83
 710 REM OUTPUT END MEMBERS TO DISK FILE
 730 IF ASC(P$)<>89 GOTO800
 750 FOR J=1TON2\PRINT#4,USING"####",F(J)
 770 PRINT#4,USING Q4$,V(J);Y(J);X(J);Q(J);M(J);U(J)
 780 NEXTJ
 800 IF ASC(R7$)<>89 GOTO2050
 810 PRINT#2,'          '\FOR I=1 TO (2*N2)
 830 PRINT#2,'      CATIONS/'\NEXT I
```

```
840 PRINT#2\PRINT#2,'
850 FOR I=1 TO N2
860 PRINT#2,'   12ANIONS     ';
870 PRINT#2,USING"##.###",H5(I);
880 PRINT#2,'ANIONS';
890 NEXT I\PRINT#2
910 Q6$='##########################.##'
930 Q7$='##########################.###'
940 PRINT#2
960 FOR J=1TON2\Z1(1,J)=N(J)\Z1(2,J)=V(J)
970 Z1(3,J)=Y(J)\Z1(4,J)=X(J)\Z1(5,J)=Q(J)
980 Z1(6,J)=M(J)\Z1(7,J)=U(J)\NEXTJ
1000 FOR J=1TO7
1010 GOSUB 1350
1030 PRINT#2,USING"'LLLLLLLLLL",H$;
1040 FOR I=1TON2
1050 PRINT#2,USING Q6$,Z1(J,I);\NEXTI
1060 PRINT#2\NEXTJ
1070 PRINT#2,'CAT. RESIDUALS'
1080 FOR I=1TON2\FOR J=1TO10
1090 IF R(J,I)<0 THEN LET R(J,I)=0\NEXT J
1100 NEXT I
1110 FOR J=1TO10
1120 GOSUB1550
1140 PRINT#2,USING"'LLLLLLLLLL",H$;
1150 FOR I=1 TO N2
1170 PRINT#2,USING Q7$,R(J,I);
1180 NEXT I
1200 PRINT#2\NEXTJ\PRINT#2,'FE+2/MG    ';
1210 FOR J=1TON2\IF W(22,J)<>0 GOTO1250
1230 F6(J)=0\GOTO1260
1240 REM
1250 F6(J)=W(20,J)/W(22,J)
1260 PRINT#2,USING Q7$,F6(J);\NEXTJ\PRINT#2
1270 FOR J=1 TO N2\PRINT#2,'ANALYSIS #';
1280 PRINT#2,USING"####",F(J);
1290 PRINT#2,' ADDITION OF ';\PRINT#2,USING"##.###",2*(12-H(J));
1300 PRINT#2,' MOLES SI WOULD BALANCE THE CATION TO ANION RATIO. ';
1310 PRINT#2,'THIS EQUALS ';\PRINT#2,USING Q4$,L5(J);
1320 PRINT#2,' WT.% SIO2'
1340 NEXT J\GOTO 2050
1350 REM SUBROUTINE TO CHOOSE ROW LABELS FROM Q$
1370 Q1=0\M1=0
1380 FOR I=1 TO LEN(Q$)
1390 V$=SEG$(Q$,I,I)
1410 IF V$<>'/' GOTO1500
1420 M1=M1+1
1440 IF M1=J GOTO1460
1450 IF M1=J+1 GOTO1480
1460 Q1=I+1
1470 GOTO1500
1480 G1=I-1
1490 H$=SEG$(Q$,Q1,G1)
1500 NEXT I
1520 RETURN
1540 REM SUBROUTINE TO CHOOSE ROW LABELS FROM I$
1550 Q2=0\M2=0
1560 FOR I=1 TO LEN(I$)
1570 C$=SEG$(I$,I,I)
1600 IF C$<>'/' GOTO1700
1620 M2=M2+1
1630 IF M2=J GOTO1650
1640 IF M2=J+1 GOTO1670
1650 Q2=I+1
1660 GOTO1700
1670 G2=I-1
1680 H$=SEG$(I$,Q2,G2)
1700 NEXT I
1720 RETURN
1740 REM
1750 REM
1780 REM
1800 REM
1810 REM
1820 REM
1830 REM
1850 REM
1860 REM
1870 REM
1880 REM
1890 REM
1900 REM
1910 REM
1920 REM
1930 REM
```

Appendix VII. Example of output from GARN

S69-32 GARNETS
28 GA CORE A-1-1 2PT AVE S69-52
29 GA RIM 2PT AVE A-1-2 S69-52
30 GA CORE 2PT AVE A-2-1 S69-52
31 GA RIM 2PT AVE A-2-2 S69-52

ANALYSIS	28		29		30		31	
	INPUT	NORMALIZED	INPUT	NORMALIZED	INPUT	NORMALIZED	INPUT	NORMALIZED
SIO2	36.13	36.13	37.15	37.15	36.11	36.11	37.05	37.05
TIO2	0.45	0.45	0.06	0.06	0.41	0.41	0.18	0.18
AL2O3	4.65	4.65	6.88	6.88	5.23	5.23	7.44	7.44
CR2O3	0.00	0.00	0.00	0.00	0.00	0.00	0.00	0.00
FE2O3	0.00	24.47	0.00	21.42	0.00	23.90	0.00	21.12
FEO	23.91	1.88	19.99	0.71	23.20	1.69	19.39	0.38
MNO	3.44	3.44	0.97	0.97	2.92	2.92	0.92	0.92
MGO	0.22	0.22	0.16	0.16	0.25	0.25	0.16	0.16
CAO	29.55	29.55	33.17	33.17	30.02	30.02	33.46	33.46
NA2O	0.00	0.00	0.00	0.00	0.00	0.00	0.00	0.00
SUM	98.35	100.80	98.38	100.52	98.14	100.53	98.60	100.71
SI	3.187	2.985	3.187	3.013	3.175	2.978	3.162	2.993
AL	0.000	0.015	0.000	0.000	0.000	0.022	0.000	0.007
Z	3.187	3.000	3.187	3.013	3.175	3.000	3.162	3.000
AL	0.483	0.437	0.696	0.658	0.542	0.487	0.748	0.700
CR	0.000	0.000	0.000	0.000	0.000	0.000	0.000	0.000
FE+3	0.000	1.522	0.000	1.308	0.000	1.484	0.000	1.284
TI	0.030	0.028	0.004	0.004	0.027	0.025	0.012	0.011
Y	0.513	1.987	0.700	1.970	0.569	1.996	0.760	1.996
FE+2	1.764	0.130	1.435	0.048	1.706	0.116	1.384	0.026
MN	0.257	0.241	0.070	0.066	0.217	0.204	0.067	0.063
MG	0.029	0.027	0.021	0.020	0.033	0.031	0.021	0.020
CA	2.792	2.615	3.049	2.883	2.828	2.653	3.059	2.895
NA	0.000	0.000	0.000	0.000	0.000	0.000	0.000	0.000
X	4.842	3.013	4.575	3.017	4.784	3.004	4.531	3.004
CATSUM	8.542	8.000	8.462	8.000	8.528	8.000	8.453	8.000
	CATIONS/ 12ANIONS	CATIONS/ 12.000ANIONS	CATIONS/ 12ANIONS	CATIONS/ 12.000ANIONS	CATIONS/ 12ANIONS	CATIONS/ 12.000ANIONS	CATIONS/ 12ANIONS	CATIONS/ 12.000ANIONS
XCAT ALLOC		98.70		98.27		99.28		99.62
UVAR		0.00		0.00		0.00		0.00
AND		77.06		66.53		74.72		64.45
PYR		0.91		0.66		1.03		0.64
SPES		8.13		2.26		6.85		2.11
GROS		11.27		30.55		14.36		32.46
ALM		2.62		0.00		3.04		0.34
CAT. RESIDUALS								
SI		0.005		0.013		0.000		0.001
TI		0.006		0.001		0.005		0.002
AL		0.000		0.000		0.001		0.000
CR		0.000		0.000		0.000		0.000
FE+3		0.000		0.000		0.000		0.000
FE+2		0.011		0.010		0.005		0.003
MN		0.000		0.000		0.000		0.000
MG		0.000		0.000		0.000		0.000
CA		0.000		0.004		0.000		0.000
NA		0.000		0.000		0.000		0.000
FE+2/MG		4.793		2.429		3.758		1.286

ANALYSIS # 28 ADDITION OF 1.522 MOLES SI WOULD BALANCE THE CATION TO ANION RATIO. THIS EQUALS 18.42 WT.% SIO2
ANALYSIS # 29 ADDITION OF 1.308 MOLES SI WOULD BALANCE THE CATION TO ANION RATIO. THIS EQUALS 16.13 WT.% SIO2
ANALYSIS # 30 ADDITION OF 1.484 MOLES SI WOULD BALANCE THE CATION TO ANION RATIO. THIS EQUALS 17.99 WT.% SIO2
ANALYSIS # 31 ADDITION OF 1.284 MOLES SI WOULD BALANCE THE CATION TO ANION RATIO. THIS EQUALS 15.90 WT.% SIO2

Appendix VIII. Listing of XYPLOT

```
500 PRINT'XYPLOT.BAS'\REM X-Y POINT PLOTTING
505 REM OLD VERSION XYPLT1 MODELLED AFTER J.MCGEE'S IPLOT(ARLEB)
510 REM AND J.S.HUEBNER'S QUAD.BAS(MINC)!M.FLOHR 4/83
515 REM MODIFIED 7/28 BY FLOHR RENAMED XYPLOT
520 A$='@!!'\U$='U'\B$='H A LO'\D$='D'
530 H$='H'\8$=' '\C$=','\H2$='_ H'\D4$='_ '
550 REM X9,Y9=SCALING FACTORS(=1"=200STEPS)
560 X9=1\Y9=1
570 F=1.0\REM NO INPUT OF SCALING FOR NOW
580 S=200*F
590 COUT(,A$,3,0)
600 X1=8*(0+X9)\X1$=STR$(X1)+C$
610 Y1=8*(0+Y9)\Y1$=STR$(Y1)
620 X2=8*(12+X9)\X2$=STR$(X2)+C$
630 Y2=8*(0+Y9)\Y2$=STR$(Y2)
640 X3=X1\X3$=STR$(X3)+C$
650 Y3=S*(8+Y9)\Y3$=STR$(Y3)
660 X4=X2\X4$=STR$(X4)+C$
670 Y4=Y3\Y4$=STR$(Y4)
680 T2$=B$+U$+X1$+Y1$+S$+D$+X2$+Y2$+S$+X4$+Y4$+8$
690 T3$=X3$+Y3$+S$+X1$+Y1$+U$+H$\T1$=T2$+T3$
700 COUT(,T1$,255,0)\REM LABEL AXES,PRINT TITLE
703 PAUSE(5)
705 PRINT'X,Y DIMENSIONS ARE 12,8 INCHES'
710 PRINT'ENTER LABEL FOR X-AXIS: '!\INPUT L4$
715 L$='H A U 800,80 S12 '\P$=L$+L4$+H2$
720 COUT(,A$,3,0)\COUT(,P$,255,0)
725 PRINT'ENTER LABEL FOR Y-AXIS: '!\INPUTL5$
730 L1$='H A U 100,750 S42 '\P1$=L1$+L5$+H2$
740 COUT(,A$,3,0)\COUT(,P1$,255,0)
750 PRINT'ENTER TITLE:'\INPUT L7$
770 L3$='H A U 600,1860 S12 '
790 P3$=L3$+L7$+H2$\COUT(,A$,3,0)
820 COUT(,P3$,255,0)
825 PAUSE(15)\REM NEEDED FOR LONG LABELS
830 PRINT'ENTER MIN,MAX RANGE FOR Y: '!\INPUT R3,R4
840 P5$='H A U 125,200 S12 '\R3$=STR$(R3)\R4$=STR$(R4)
850 P6$='125,1800 S12 '\Q3$=P5$+R3$+D4$+P6$+R4$+H2$
860 COUT(,A$,3,0)\COUT(,Q3$,255,0)
865 PRINT'ENTER MIN,MAX RANGE FOR X: '!\INPUT R5,R6
870 P2$='H A U 175,150 S12 '\REM X-MIN
880 P4$='2560,150 S12 '\REM X-MAX
910 R5$=STR$(R5)\R6$=STR$(R6)
930 Q2$=P2$+R5$+D4$+P4$+R6$+H2$
960 COUT(,A$,3,0)\COUT(,Q2$,255,0)
990 REM PRINT TICK MARKS
995 PAUSE(15)
1000 V1=175\V1$=STR$(V1)
1010 V2=200\V2$=STR$(V2)
1020 V3=1800\V3$=STR$(V3)
1030 V4=1825\V4$=STR$(V4)
1040 V5=2600\V5$=STR$(V5)
1050 V6=2625\V6$=STR$(V6)
1060 COUT(,B$,10,0)\REM BOTTOM
1070 FOR V7=400TO2400 STEP200
1080 V7$=STR$(V7)+C$
1090 V8$=V7$+V1$+S$+D$+V7$+V2$+S$+U$
1100 COUT(,V8$,255,0)\NEXTV7
1110 COUT(,A$,3,0)\COUT(,B$,10,0)\REM LEFT
1120 FOR V9=400TO1600 STEP200
1130 V9$=C$+STR$(V9)
1140 Z1$=V1$+V9$+S$+D$+V2$+V9$+S$+U$
1150 COUT(,Z1$,255,0)\NEXTV9
1160 COUT(,A$,3,0)\COUT(,B$,10,0)\REM TOP
1170 FOR Z2=400 TO 2400 STEP 200
1180 Z2$=STR$(Z2)+C$
1190 Z3$=Z2$+V4$+S$+D$+Z2$+V3$+S$+U$
1200 COUT(,Z3$,255,0)\NEXT Z2
1210 COUT(,A$,3,0)\COUT(,B$,10,0)\REM RIGHT
1220 FOR Z5=400 TO 1600 STEP 200
1230 Z5$=C$+STR$(Z5)
1240 Z4$=V6$+Z5$+S$+D$+V5$+Z5$+S$+U$
1250 COUT(,Z4$,255,0)\NEXT Z5
1260 COUT(,H$,255,0)
1280 END
```

Appendix IX. Listing of TERPLT

```
500 PRINT'TERPLT.BAS'\REM TERNARY POINT PLOTTING
505 REM OLD VERSION TERPL1 MODELLED AFTER J.MCGEE'S IPLOT(ARLEB)
510 REM AND J.S.HUEBNER'S QUAD.BAS(MINC)!M.FLOHR 4/83
515 REM MODIFIED 7/27/83 BY FLOHR-RENAMED TERPLT
520 A$='@!!'\D$='D'\S$=' '\U$='U'\C$=','
530 B$='M A LO'\H$='H'\H2$='_ H'\D4$='_ '
550 PRINT'TRIANGLE HAS SIDES OF 10 INCHES, HEIGHT OF 8.66 INCHES'
560 REM X9,Y9 SCALING FACTORS(=1"=200STEPS)
570 X9=1\Y9=1
580 F=1.0\REM F=SCALING FACTOR-NO INPUT FOR NOW
590 S=200*F
600 COUT(,A$,3,0)
610 X1=S*(0+X9)\X1$=STR$(X1)+C$
620 Y1=S*(0+Y9)\Y1$=STR$(Y1)
630 X2=S*(10+X9)\X2$=STR$(X2)+C$
640 Y2=S*(0+Y9)\Y2$=STR$(Y2)
650 X3=S*(5+X9)\X3$=STR$(X3)+C$
660 Y3=S*(8.66+Y9)\Y3$=STR$(Y3)
670 T1$=B$+U$+X1$+Y1$+S$+D$+X2$+Y2$+S$+X3$+Y3$+S$+X1$+Y1$+U$+H$
680 COUT(,T1$,255,0)
700 L$='M A U 125,125 S13 '\REM LEFT
710 L1$='M A U 1150,1940 S13 '\REM TOP
720 L2$='2200,125 S13 '\REM RIGHT
730 L3$='M A U 1600,1800 S12 '\REM TITLE
740 PRINT'ENTER LABEL FOR LEFT'!\INPUT L4$
750 PRINT'ENTER LABEL FOR RIGHT'!\INPUT L5$
755 P$=L$+L4$+D4$+L2$+L5$+H2$\COUT(,A$,3,0)\COUT(,P$,255,0)
760 PRINT'ENTER LABEL FOR TOP'!\INPUT L6$
770 PRINT'ENTER TITLE (MAX OF 30 CHARACTERS)'
780 INPUT L7$
790 P1$=L1$+L6$+H2$
800 P3$=L3$+L7$+H2$
810 COUT(,A$,3,0)\COUT(,P1$,255,0)
850 COUT(,A$,3,0)\COUT(,P3$,255,0)
855 PAUSE(10)
860 REM PRINT TICK MARKS-BASE FIRST
870 Y4=175\Y4$=STR$(Y4)\Y5=200\Y5$=STR$(Y5)
880 COUT(,A$,3,0)\COUT(,B$,10,0)
890 FOR X4=400TO2000 STEP200
900 X4$=STR$(X4)+C$
910 T2$=X4$+Y4$+S$+D$+X4$+Y5$+S$+U$
920 COUT(,T2$,255,0)
930 NEXT X4
940 COUT(,A$,3,0)\COUT(,B$,10,0)
950 X7=-25\X8=25\REM LEFT SIDE
960 FOR Z4=400TO2000 STEP 200
970 X5=200+((Z4-200)*.5)\Y6=200+((Z4-200)*.866)
1000 Y6$=STR$(Y6)\X5$=STR$(X5)+C$
1020 X9=X7+X5\X9$=STR$(X9)+C$
1050 T3$=X9$+Y6$+S$+D$+X5$+Y6$+S$+U$
1060 COUT(,T3$,255,0)\NEXTZ4
1080 COUT(,H$,255,0)\REM RIGHT SIDE
1090 COUT(,A$,3,0)\COUT(,B$,10,0)
2000 FOR Z5=400 TO 2000 STEP 200
2010 Y8=200+((Z5-200)*.866)\X6=2200-((Z5-200)*.5)
2020 X6$=STR$(X6)+C+\Y8$=STR$(Y8)
2030 W4=X8+X6\W4$=STR$(W4)+C$
2040 T4$=X6$+Y8$+S$+D$+W4$+Y8$+S$+U$
2050 COUT(,T4$,255,0)\NEXTZ5
2060 COUT(,H$,255,0)
2090 END
```

Appendix X. Listing of TETPLT

```
100REM TETPLT.BAS DRAWS TETRAHEDRON M.FLOHR 7/83
110REM USES F.SPEAR ALGORITHM(AM. MIN.65,1980)
120 A$='@;:'\D$='D'\S$=' '\U$='U'
130 C$=','\H$='H'\R$='R'\N$='_'
135 Q1=1200\Q2=800\REM CENTER OF BARYCENTRIC SYSTEM
140 REM CALCULATE X,Y,Z OF APICES; TOP (0,0,0,1)
150 T4=1-.25\T5=0-0.25
160 X=(T5+(T5/2)+(T4/2))/0.8165
170 Y=(T5+(T4/3))/0.9428\Z=T4
180 REM BOTTOM (1,0,0,0)
190 X1=(T4+(T5/2)+(T5/2))/0.8165
200 Y1=(T5+(T5/3))/0.9428\Z1=T5
210 REM RIGHT (0,1,0,0)
220 X2=(T5+(T4/2)+(T5/2))/0.8165\Z2=T5
230 Y2=(T4+(T5/3))/0.9428
240 REM LEFT (0,0,1,0)
250 X3=(T5+(T5/2)+(T5/2))/0.8165
260 Y3=(T5+(T5/3))/0.9428\Z3=T5
270 REM ALPHA ROTATES AROUND Z-AXIS; THETA AROUND X-AXIS
280 PRINT'ENTER ANGLE ALPHA IN DEGREES: '#\INPUT A
290 PRINT'ENTER ANGLE THETA IN DEGREES: '#\INPUT T
295 PRINT'ENTER E (PERSPECTIVE): '#\INPUT E
300 PRINT'ENTER S (SCALING FACTOR): '#\INPUT S
305 REM CONVERT DEGREES TO RADIANS
310 A=A/57.29578\T=T/57.29578
340 REM CALCULATE Z' (HERE V) FOR EACH APEX POINT
360 R1=COS(A)\R2=COS(T)
370 R3=SIN(A)\R4=SIN(T)\W=R3*R4\W2=R1*R4
380 V=(-X*W)-(Y*W2)+(Z*R2)\V1=(-X1*W)-(Y1*W2)+(Z1*R2)
390 V2=(-X2*W)-(Y2*W2)+(Z2*R2)\V3=(-X3*W)-(Y3*W2)+(Z3*R2)
400 REM CALCULATE X,Y COORD. OF APICES
410 REM U1,U2=X,Y TOP; U3,U4 - BOTTOM
420 REM U5,U6 - RIGHT; U7,U8 - LEFT
430 U1=INT(((X*R1)-(Y*R3))*(E/(E-V))*S)+Q1
440 U2=INT(((X*R3*R2)+(Y*R1*R2)+(Z*R4))*(E/(E-V))*S)+Q2
450 U3=INT(((X1*R1)-(Y1*R3))*(E/(E-V1))*S)+Q1
460 U4=INT(((X1*R3*R2)+(Y1*R1*R2)+(Z1*R4))*(E/(E-V1))*S)+Q2
470 U5=INT(((X2*R1)-(Y2*R3))*(E/(E-V2))*S)+Q1
480 U6=INT(((X2*R3*R2)+(Y2*R1*R2)+(Z2*R4))*(E/(E-V2))*S)+Q2
490 U7=INT(((X3*R1)-(Y3*R3))*(E/(E-V3))*S)+Q1
500 U8=INT(((X3*R3*R2)+(Y3*R1*R2)+(Z3*R4))*(E/(E-V3))*S)+Q2
510 PRINT'ENTER LABEL FOR TOP'\INPUT L1$
520 PRINT'ENTER LABEL FOR BOTTOM'\INPUTL2$
530 PRINT'ENTER LABEL FOR RIGHT'\INPUT L3$
540 PRINT'ENTER LABEL FOR LEFT'\INPUT L4$
550 PRINT'ENTER TITLE (MAX 30 CHARACTERS)'\INPUT L5$
560 COUT(,A$,3,0)\B2$='H A L0'\COUT(,B2$,20,0)
590 U1$=STR$(U1)+C$\U2$=STR$(U2)
600 U3$=STR$(U3)+C$\U4$=STR$(U4)
610 U5$=STR$(U5)+C$\U6$=STR$(U6)
620 U7$=STR$(U7)+C$\U8$=STR$(U8)
630 B6$=U3$+U4$+S$+D$+U5$+U6$+S$+U1$+U2$+S$
640 B3$=U3$+U4$+S$+U7$+U8$+S$+U1$+U2$+S$+U$
650 B4$=U$+B6$+B3$
660 COUT(,B4$,255,0)\COUT(,H$,10,0)
670 COUT(,A$,3,0)
680 B7$='H A L5'\B8$=B7$+U$+U7$+U8$+S$+D$+U5$+U6$+S$+U$
690 COUT(,B8$,255,0)
700 COUT(,H$,10,0)
710 PAUSE(10)\REM LABEL DIAGRAM
720 COUT(,A$,3,0)\S3$='S13 '
730 J=U4-60\J$=STR$(J)
740 J2=U5-U3-40\J3=U6-U4
750 J2$=STR$(J2)+C$\J3$=STR$(J3)
760 J5=U2+20
770 J5$=STR$(J5)\J6=U7-60
780 J6$=STR$(J6)+C$\J7$=STR$(J7)
800 K2$=H$+R$+U$+U3$+J$+S$+S3$+L2$+N$+S$+U$
810 K3$=J2$+J3$+S$+S3$+L3$+N$+S$+U$+H$
840 K6$=K2$+K3$\REM BOTTOM,RIGHT LABELS
850 COUT(,K6$,255,0)\COUT(,A$,3,0)
855 J8=U8-60\J8$=STR$(J8)
860 K9$='H A U '\J9$=K9$+J6$+J8$+S$+S3$+L4$+N$+S$+U$+H$
870 COUT(,J9$,255,0)\REM LEFT LABEL
873 K4$=K9$+U1$+J5$+S$+S3$+L1$+N$+S$+U$+H$
876 COUT(,A$,3,0)\COUT(,K4$,255,0)\REM TOP LABEL
880 K7$='H A U 800,1800 S12 '
890 K8$=K7$+L5$+N$+S$+U$+H$\REM TITLE
900 COUT(,K8$,255,0)
1000 END
```

Appendix XI. Listing of PLAMPH

```
100 REM PLAMPH.BAS READS CATION DATA FILE CREATED BY AMPHI.BAS AND
110 REM PLOTS X-Y GRAPHS-M.FLOHR 6/83
130 A$='@!!'\H$='H'\D$='D'\U$='U'\S$=' '
140 C$='####'\C3$='##.###'\H3$='H A '
150 OPEN 'LP:' FOR OUTPUT AS FILE#2
160 REM NO TERNARY OPTIONS FOR NOW - R=2 FOR XY PLOTS
170 R=2\Q4=0\REM Q4 IS COUNTER FOR PRINTING PLOTTER INFORMATION
175 DIM D(20)
178 PRINT#2
180 PRINT'ENTER NAME OF DATA FILE AS SYN:XXXXXX.YYY'\INPUTN$
190 OPEN N$ FOR INPUT AS FILE#1
200 PRINT'ENTER CODE FOR PLOTTING SYMBOL: 0,1,2,3,4 OR 5'\INPUTS
210 IF S=0 THEN P3$=' M20 D U'
215 IF S=1 THEN P3$=' M21 D U'
220 IF S=2 THEN P3$=' M22 D U'
225 IF S=3 THEN P3$=' M23 D U'
230 IF S=4 THEN P3$=' M24 D U'
235 IF S=5 THEN P3$=' M25 D U'
240 INPUT#1,A9$
260 PRINT#2\Z2=0\REM COUNTER FOR RELATIVE PLOTTING
280 IF Q4<>0 GOTO565
290 PRINT'X-Y PLOTTING OPTIONS: (ALVI + FE+3 + 2TI)=1'
295 PRINT'(CA+NA IN M4)=2; (FE+2 + MN)/(FE+2 + MN + MG)=3'
300 PRINT'(NA IN M4)=4; (ALIV)=5; (ALVI + FE+3 + TI + CR)=6'
305 PRINT'(NA IN A + K)=7; (100NA)/(CA+NA)=8; (100AL)/(SI+AL)=9'
310 PRINT'(ALVI)=10; (TI)=11; (K)=12; (NA IN A)=13; MN=14; (MG/MG + FE+2)=15'
450 PRINT'CHOOSE X,Y'\INPUT X,Y
460 PRINT#2,'X-Y OPTIONS ARE: (ALVI + FE+3 + 2TI)=1; (CA+NA IN M4)=2; ';
470 PRINT#2,'(FE+2 + MN)/(FE+2 + MN + MG)=3;  (NA IN M4)=4;  ';
480 PRINT#2,'(ALIV)=5;'
490 PRINT#2,'(ALVI + FE+3 + TI + CR)=6; (NA IN A + K)=7;  ';
500 PRINT#2,'(100NA)/(CA+NA)=8;  (100AL)/(AL+SI)=9;  ';
505 PRINT#2,'(ALVI)=10;  (TI)=11;  (K)=12;'
507 PRINT#2,'(NA IN A)=13;  (MN)=14;  (MG/MG + FE+2)=15'
510 PRINT#2\PRINT#2,'FOR THIS PLOT: X= ';X
520 PRINT#2,'                Y= ';Y\PRINT#2
530 Q4=Q4+1
550 PRINT'INPUT X MIN,MAX'\INPUT P3,P4
560 PRINT'INPUT Y MIN,MAX'\INPUT P5,P6
565 PRINT#2,'DATA FILE ACCESSED: '!N$
570 PRINT#2,'DATA FILE ACCESSED BY AMPHI.BAS: '!A9$\PRINT#2
580 IF END#1 GOTO 4000
590 INPUT#1,N2
592 P1=0\P2=0
595 FOR J=1TO20\INPUT#1,D(J)\NEXTJ
610 C1=D(3)+D(4)+2*D(7)\C2=D(13)+D(14)
620 C3=(D(5)+D(11)+D(8)+D(10))/(D(5)+D(11)+D(8)+D(10)+D(12)+D(6))
630 C4=D(14)\C5=D(2)\C6=D(3)+D(4)+D(7)+D(9)
635 C7=D(15)+D(16)
640 C8=(100*(D(14)+D(15)))/(D(13)+D(17)+D(14)+D(15))
650 C9=(100*(D(2)+D(3)))/(D(1)+D(2)+D(3))
655 H1=D(3)\H2=D(7)\H3=D(16)\H4=D(15)
660 H5=D(8)+D(10)\H6=(D(6)+D(12))/(D(6)+D(12)+D(5)+D(11))
700 REM P1=X COORD;P2=Y COORD
710 IF X=1 THEN P1=C1
715 IF Y=1 THEN P2=C1
720 IF X=2 THEN P1=C2
725 IF Y=2 THEN P2=C2
730 IF X=3 THEN P1=C3
735 IF Y=3 THEN P2=C3
740 IF X=4 THEN P1=C4
745 IF Y=4 THEN P2=C4
750 IF X=5 THEN P1=C5
755 IF Y=5 THEN P2=C5
760 IF X=6 THEN P1=C6
763 IF Y=6 THEN P2=C6
765 IF Y=7 THEN P2=C7
770 IF X=7 THEN P1=C7
780 IF X=8 THEN P1=C8
785 IF Y=8 THEN P2=C8
790 IF X=9 THEN P1=C9
795 IF Y=9 THEN P2=C9
800 IF X=10 THEN P1=H1
805 IF Y=10 THEN P2=H1
810 IF X=11 THEN P1=H2
815 IF Y=11 THEN P2=H2
820 IF X=12 THEN P1=H3
825 IF Y=12 THEN P2=H3
830 IF X=13 THEN P1=H4
835 IF Y=13 THEN P2=H4
```

```
840 IF X=14 THEN P1=H5
845 IF Y=14 THEN P2=H5
850 IF X=15 THEN P1=H6
855 IF Y=15 THEN P2=H6
900 REM PLOT POINT
910 R1=(P1-P3)/(P4-P3)\R2=(P2-P5)/(P6-P5)
920 P7=INT(200+2400*R1)\P8=INT(200+1600*R2)
925 IF Z2<>0GOTO960
930 P7$=STR$(P7)\P8$=STR$(P8)\P9$='H A '+P7$+S$+P8$+P3$
950 COUT(,A$,3,0)\COUT(,P9$,255,0)
955 X4=P7\Y4=P8\Z2=Z2+1\GOTO977
960 X3=P7-X4\Y3=P8-Y4\REM DEFINE RELATIVE COORD. OF NEXT PT.
965 X3$=STR$(X3)\Y3$=STR$(Y3)\X7$='R U '+X3$+S$+Y3$+P3$
970 COUT(,A$,3,0)\COUT(,X7$,255,0)
975 X4=P7\Y4=P8\GOTO990
977 PRINT#2,'SYMBOL CODE:';\PRINT#2,USING"##",S
980 PRINT#2,'ANALYSIS      X        Y'\PRINT#2
990 PRINT#2,USING C$,N2;\PRINT#2,'        ';
1000 PRINT#2,USING C3$,P1;\PRINT#2,'     ';
1010 PRINT#2,USING C3$,P2
1030 GOTO580
4000 CLOSE#1\COUT(,A$,3,0)\COUT(,H3$,10,0)
4010 PRINT'DO YOU WANT TO PLOT DATA FROM ANOTHER FILE ON THIS GRAPH?'
4015 PRINT'(Y OR N)'\INPUT R$
4018 PRINT#2\PRINT#2
4020 IF ASC(R$)=89 GOTO 180
4030 CLOSE#2
5000 END
```

Appendix XII. Listing of MICAPL

```
100 REM MICAPL.BAS READS CATION DATA FILE CREATED BY MICA.BAS AND
110 REM PLOTS X-Y GRAPHS M.FLOHR 6/83
130 A$='@!!'\H$='H'\D$='D'\U$='U'\B$=' !\M2$='  '
140 C$='####'\C3$='#.###'\M3$='H A '
150 OPEN 'LP:' FOR OUTPUT AS FILE#2
160 REM NO TERNARY OPTIONS FOR NOW - R=2 FOR XY PLOTS
170 R=2\Q4=0\REM Q4 COUNTER FOR PRINTING PLOTTING INFO.
175 DIM D(15)
177 PRINT#2
180 PRINT'ENTER NAME OF DATA FILE AS SYN:XXXXXX.YYY'\INPUTN$
185 OPEN N$ FOR INPUT AS FILE#1
190 PRINT'ARE YOU SORTING?(Y OR N) '!\INPUTS$
193 IF ASC(S$)<>89 GOTO200
195 PRINT'ENTER SORT STRING: '!\INPUTM2$
200 PRINT'ENTER CODE FOR PLOTTING SYMBOL: 0,1,2,3,4 OR 5'\INPUTS
210 IF S=0 THEN P3$=' M20 D U'
215 IF S=1 THEN P3$=' M21 D U'
220 IF S=2 THEN P3$=' M22 D U'
225 IF S=3 THEN P3$=' M23 D U'
230 IF S=4 THEN P3$=' M24 D U'
235 IF S=5 THEN P3$=' M25 D U'
240 INPUT#1,A9$
260 Z2=0\REM COUNTER FOR RELATIVE PLOTTING
280 IF Q4<>0 GOTO565
290 PRINT'X-Y PLOTTING OPTIONS: (ALIV)=1! (ALVI)=2!'
295 PRINT'(FE+2 + MG + TI + (SI-3))=3! (ALIV + ALVI + FE+3)=4  '
300 PRINT'((FE+2 + MN)/(FE+2 + MN + MG))=5! (TI)=6'
450 PRINT'CHOOSE X,Y'\INPUT X,Y
460 PRINT#2,'X-Y OPTIONS ARE: (ALIV)=1! (ALVI)=2!  '!
470 PRINT#2,'(FE+2 + MG + TI + (SI-3))=3!  (ALIV + ALVI + FE+3)=4'
480 PRINT#2,'(FE+2 + MN)/(FE+2 + MN + MG)=5!  (TI)=6'
510 PRINT#2\PRINT#2,'FOR THIS PLOT: X= '!X
520 PRINT#2,'                Y= '!Y\PRINT#2
530 Q4=Q4+1
550 PRINT'INPUT X MIN,MAX'\INPUT P3,P4
560 PRINT'INPUT Y MIN,MAX'\INPUT P5,P6
565 PRINT#2,'DATA FILE ACCESSED: '!N$
570 PRINT#2,'DATA FILE ACCESSED BY MICA.BAS: '!A9$\PRINT#2
580 IF END#1 GOTO 4000
590 INPUT#1,M2,M$
592 P1=0\P2=0
595 FOR J=1TO15\INPUT#1,D(J)\NEXTJ
600 IF ASC(S$)<>89 GOTO610
605 IF M$<>M2$ GOTO580
610 C1=D(2)\C2=D(3)\C3=D(5)+D(6)+D(7)+(D(1)-3)
630 C4=D(2)+D(3)+D(4)\C5=(D(5)+D(8))/(D(5)+D(8)+D(6))
640 C6=D(7)
700 REM P1=X COORD!P2=Y COORD
710 IF X=1 THEN P1=C1
715 IF Y=1 THEN P2=C1
720 IF X=2 THEN P1=C2
725 IF Y=2 THEN P2=C2
730 IF X=3 THEN P1=C3
735 IF Y=3 THEN P2=C3
740 IF X=4 THEN P1=C4
745 IF Y=4 THEN P2=C4
750 IF X=5 THEN P1=C5
755 IF Y=5 THEN P2=C5
760 IF X=6 THEN P1=C6
765 IF Y=6 THEN P2=C6
900 REM PLOT POINT
910 R1=(P1-P3)/(P4-P3)\R2=(P2-P5)/(P6-P5)
920 P7=INT(200+2400*R1)\P8=INT(200+1600*R2)
925 IF Z2<>0 GOTO960
930 P7$=STR$(P7)\P8$=STR$(P8)\P9$='H A '+P7$+S$+P8$+P3$
950 COUT(,A$,3,0)\COUT(,P9$,255,0)
955 X4=P7\Y4=P8\Z2=Z2+1\GOTO983
960 X3=P7-X4\Y3=P8-Y4\REM DEFINE RELATIVE COORD. OF NEXT PT.
965 X3$=STR$(X3)\Y3$=STR$(Y3)\X7$='R U '+X3$+S$+Y3$+P3$
975 COUT(,A$,3,0)\COUT(,X7$,255,0)
980 X4=P7\Y4=P8\Z2=Z2+1\GOTO995
983 PRINT#2,'SYMBOL CODE: '!\PRINT#2,USING"##",S
985 PRINT#2,'ANALYSIS     X       Y'
990 PRINT#2,USING"LL",M2$
995 PRINT#2,USING C$,N2!\PRINT#2,'       '!
1000 PRINT#2,USING C3$,P1!\PRINT#2,'     '!
1010 PRINT#2,USING C3$,P2
1030 GOTO580
4000 PRINT#2\COUT(,A$,3,0)\COUT(,H3$,10,0)
4005 PRINT'DO YOU WANT TO PLOT MORE DATA FROM THIS FILE ON THIS GRAPH?'
4010 PRINT'(Y OR N)'\INPUT R5$
4020 IF ASC(R5$)<>89 GOTO4040
4030 RESTORE#1\GOTO195
4040 CLOSE#1
4050 PRINT'DO YOU WANT TO PLOT DATA FROM ANOTHER FILE ON THIS GRAPH?'
4060 PRINT'(Y OR N)'\INPUT R$
4070 PRINT#2\PRINT#2
4080 IF ASC(R$)=89 GOTO180
4090 CLOSE#2
5000 END
```

Appendix XIII. Listing of CHPLOT

```
100 REM CHPLOT.BAS READS CATION DATA FILE CREATED BY CHLOR.BAS AND
110 REM PLOTS X-Y GRAPHS!M.FLOHR 6/83
130 A$='@!!'\H$='H'\D$='D'\U$='U'\S$=' '
140 C$='####'\C3$='#.###'\H3$='H A '
150 OPEN 'LP:' FOR OUTPUT AS FILE#2
160 REM NO TERNARY OPTIONS FOR NOW - R=2 FOR XY PLOT?
170 R=2\Q4=0\REM Q4 COUNTER FOR PRINTING PLOTTING INFORMATION
175 DIM D(15)
177 PRINT#2
180 PRINT'ENTER NAME OF DATA FILE AS SYN:XXXXXX.YYY'\INPUTN$
190 OPEN N$ FOR INPUT AS FILE#1
200 PRINT'ENTER CODE FOR PLOTTING SYMBOL: 0,1,2,3,4 OR 5'\INPUTS
210 IF S=0 THEN P3$=' M20 D U'
215 IF S=1 THEN P3$=' M21 D U'
220 IF S=2 THEN P3$=' M22 D U'
225 IF S=3 THEN P3$=' M23 D U'
230 IF S=4 THEN P3$=' M24 D U'
235 IF S=5 THEN P3$=' M25 D U'
240 INPUT#1,A9$
260 Z2=0\REM COUNTER FOR RELATIVE POINT PLOTTING
280 IF Q4<>0 GOTO565
290 PRINT'X-Y PLOTTING OPTIONS: (ALIV)=1; (ALVI)=2; (MN)=3;'
300 PRINT'(MG/MG + FE+2))=4; (TOTAL AL)=5; (MN+FE+MG)=6; (SI)=7'
450 PRINT'CHOOSE X,Y'\INPUT X,Y
460 PRINT#2,'X-Y OPTIONS ARE: (ALIV)=1; (ALVI)=2;  (MN)=3; ';
470 PRINT#2,'(MG/(MG + FE+2))=4;  (TOTAL AL)=5;  (MN+FE+MG)=6;  (SI)=7'
510 PRINT#2\PRINT#2,'FOR THIS PLOT: X= '!X
520 PRINT#2,'                Y= '!Y\PRINT#2
530 Q4=Q4+1
550 PRINT'INPUT X MIN,MAX'\INPUT P3,P4
560 PRINT'INPUT Y MIN,MAX'\INPUT P5,P6
565 PRINT#2,'DATA FILE ACCESSED: '!N$
570 PRINT#2,'DATA FILE ACCESSED BY CHLOR.BAS: '!A9$\PRINT#2
580 IF END#1 GOTO 4000
590 INPUT#1,N2
592 P1=0\P2=0\REM P1=X COORD;P2=Y COORD.
595 FOR J=1TO13\INPUT#1,D(J)\NEXTJ
610 C1=D(2)\C2=D(3)\C3=D(9)\C4=D(6)/(D(6)+D(5))\C5=C1+C2
620 C6=D(5)+D(6)+D(9)\C7=D(1)
710 IF X=1 THEN P1=C1
715 IF Y=1 THEN P2=C1
720 IF X=2 THEN P1=C2
725 IF Y=2 THEN P2=C2
730 IF X=3 THEN P1=C3
735 IF Y=3 THEN P2=C3
740 IF X=4 THEN P1=C4
745 IF Y=4 THEN P2=C4
750 IF X=5 THEN P1=C5
755 IF Y=5 THEN P2=C5
760 IF X=6 THEN P1=C6
765 IF Y=6 THEN P2=C6
770 IF X=7 THEN P1=C7
775 IF Y=7 THEN P2=C7
850 REM PLOT POINT
855 R1=(P1-P3)/(P4-P3)\R2=(P2-P5)/(P6-P5)
860 P7=INT(200+2400*R1)\P8=INT(200+1600*R2)
865 IF Z2<>0 GOTO890\REM ONLY 1ST PT. PLOTTED IN ABSOLUTE COORD.
870 P7$=STR$(P7)\P8$=STR$(P8)\P9$='H A '+P7$+S$+P8$+P3$
880 COUT(,A$,3,0)\COUT(,P9$,255,0)
885 X4=P7\Y4=P8\Z2=Z2+1\GOTO920
890 X3=P7-X4\Y3=P8-Y4\REM DEFINE RELATIVE COORD. OF NEXT PT.
895 X3$=STR$(X3)\Y3$=STR$(Y3)\X7$='R U '+X3$+S$+Y3$+P3$
900 COUT(,A$,3,0)\COUT(,X7$,255,0)
910 X4=P7\Y4=P8\Z2=Z2+1\GOTO990
920 PRINT#2,'SYMBOL CODE:'!\PRINT#2,USING'##',S
980 PRINT#2,'ANALYSIS      X        Y'\PRINT#2
990 PRINT#2,USING C$,N2!\PRINT#2,'       ';
1000 PRINT#2,USING C3$,P1!\PRINT#2,'     ';
1010 PRINT#2,USING C3$,P2
1030 GOTO580
4000 PRINT#2\CLOSE#1\COUT(,A$,3,0)\COUT(,H3$,10,0)
4030 PRINT'DO YOU WANT TO PLOT DATA FROM ANOTHER FILE ON THIS GRAPH?'
4035 PRINT'(Y OR N)'\INPUT R$
4040 PRINT#2\PRINT#2
4045 IF ASC(R$)=89 GOTO190
4050 CLOSE#2
5000 END
```

Appendix XIV. Listing of GRNPLT

```
100 PRINT'GRNPLT.BAS FOR TERNARY PLOTTING OF GARNET DATA'
105 REM READS IN OXIDES,CATIONS AS CALCULATED BY GARN.BAS,PLOTS POINTS
107 REM ON TERNARY GRAPH! M.FLOHR 6/30/83
110 A$='@!'\D$='D'\S$=' '
120 U$='U'\C$=','\H$='H'\H3$='H A '
125 F$='@@@@.@@@'\F2$="'RRRRRRR"
130 DIM B(10),Z(10),Z1(10),T$(48)
140 FOR I=1TO10\READ B(I)
150 DATA 60.09,79.9,101.96,151.99,159.69,71.85,70.94,40.31,
160 DATA 56.08,61.98
165 NEXT I
180 REM ORDER: SI,TI,AL,CR,FC,FE,MN,MG,CA,NA
210 OPEN'LP:' FOR OUTPUT AS FILE@2
220 PRINT@2,DAT$,CLK$\PRINT@2
230 PRINT'CHOOSE PLOT: AFM=1!ACFm=2!MN-FE+2-MG=3!AKFm=4'\INPUTM1
250 PRINT'ENTER NAME OF DATA FILE AS SYN:XXXXXX.YYY '\INPUTN$
300 PRINT'CHOOSE PLOTTING SYMBOL 0,1,2,3,4 OR 5 '!\INPUTN5
320 PRINT@2,'DATA FILE ACCESSED FOR THIS PLOT: '!N$
340 C4=0\REM COUNTER TO PRINT HEADINGS
345 Z2=0\REM COUNTER FOR RELATIVE PLOTTING
350 OPEN N$ FOR INPUT AS FILE@1
360 C3=0\REM COUNTER TO PRINT PLOTTING SYMBOL
365 INPUT@1,A5$\PRINT@2,'DATA FILE ACCESSED BY GARN.BAS: '!A5$
367 PRINT@2
370 IF END@1 GOTO790
375 INPUT@1,A2
420 FOR J=1TO10\INPUT@1,Z(J),Z1(J)\NEXTJ
500 IF M1=1 THEN GOSUB 860
510 IF M1=1 GOTO 650
520 IF M1=2 THEN GOSUB 2000
525 IF M1=2 GOTO650
530 IF M1=3 THEN GOSUB3050
535 IF M1=3 GOTO 650
540 IF M1=4 THEN GOSUB2000
650 REM PLOT POINT
660 L8=INT(200*(10-L4*10)-.5774*1733.4*A4+200)\L8$=STR$(L8)
670 A7=INT(1733.4*A4+200)\A7$=STR$(A7)
675 IF Z2<>0 GOTO685
680 P2$='H A '+L8$+S$+A7$
683 X4=L8\Y4=A7\Z2=Z2+1\GOTO690
685 X3=L8-X4\Y3=A7-Y4\REM DEFINE RELATIVE COORD. OF NEXT PT.
687 X3$=STR$(X3)\Y3$=STR$(Y3)\P2$='R U '+X3$+S$+Y3$
688 X4=L8\Y4=A7\Z2=Z2+1
690 IF N5=0 THEN P3$=' M20 D U'
700 IF N5=1 THEN P3$=' M21 D U'
710 IF N5=2 THEN P3$=' M22 D U'
720 IF N5=3 THEN P3$=' M23 D U'
730 IF N5=4 THEN P3$=' M24 D U'
740 IF N5=5 THEN P3$=' M25 D U'
750 P4$=P2$+P3$
760 COUT(,A@,3,0)
770 COUT(,P4$,255,0)\PAUSE(3)
780 GOTO 370
790 CLOSE@1\COUT(,A$,3,0)\COUT(,H3$,10,0)
820 PRINT'DO YOU WANT TO PLOT DATA FROM ANOTHER GARNET FILE ON THIS'
825 PRINT'GRAPH? YES=1! NO=2)'!\INPUTN7
828 PRINT@2\PRINT@2
830 IF N7=1 GOTO 250
850 GOTO 9000
860 REM SUBROUTINE FOR AFM
870 IF C4<>0 GOTO 1010
880 PRINT@2,'          '!
890 PRINT@2,USING F2$,'A'!
900 PRINT@2,USING F2$,'F'!
1000 PRINT@2,USING"'RRRRRRR",'M'
1010 IF C3=0 THEN PRINT@2,'SYMBOL CODE'!\PRINT@2,USING"@@",N5
1020 A4=(Z(3)/B(3))+(Z(5)/B(5))-(Z(10)/B(10))
1030 L4=(Z(6)/B(6))+(Z(7)/B(7))
1040 R4=Z(8)/B(8)\T=R4+L4+A4
1050 A4=A4/T\L4=L4/T\R4=R4/T
1055 PRINT@2,'ANALYSIS @ '!\PRINT@2,USING"@@@@",A2!
1060 PRINT@2,USING F$,A4!
1070 PRINT@2,USING F$,L4!\PRINT@2,USING F$,R4
1080 C3=C3+1\C4=C4+1
1090 RETURN
2000 REM SUBROUTINE FOR ACFm,AKFm
2005 IF C4<>0 GOTO 2050
2010 PRINT@2\PRINT@2,'          '!\IF M1=4 GOTO2025
2015 PRINT@2,USING F2$,'A'!\PRINT@2,USING F2$,'C'!
2020 GOTO2030
```

```
2025 PRINT#2,USING F2$,'A';\PRINT#2,USING F2$,'K';
2030 PRINT#2,USING"'RRRRRRRR",'Fe'
2050 IF C3=0 THEN PRINT#2,'SYMBOL CODE';\PRINT#2,USING"##",N5
2060 A4=(Z(3)/B(3))+(Z(5)/B(5))-(Z(10)/B(10))
2070 R4=(Z(6)/B(6))+(Z(7)/B(7))+(Z(8)/B(8))
2075 IF M1<>2 GOTO2090
2080 L4=Z(9)/B(9)\GOTO2095
2090 L4=0\REM NO K REPORTED FOR GARNETS
2095 T=A4+L4+R4\A4=A4/T\L4=L4/T\R4=R4/T
3000 PRINT#2,'ANALYSIS # ';\PRINT#2,USING"####",A2;
3010 PRINT#2,USING F$,A4;\PRINT#2,USING F$,L4;
3020 PRINT#2,USING F$,R4
3030 C3=C3+1\C4=C4+1
3040 RETURN
3050 REM SUBROUTINE FOR MN(TOP)-FE(LEFT)-MG(RIGHT)
3060 IF C4<>0 GOTO4000
3070 PRINT#2\PRINT#2,'              ';
3080 PRINT#2,USING F2$,'MN';\PRINT#2,USING F2$,'FE';
3090 PRINT#2,USING F2$,'MG'
4000 IF C3=0 THEN PRINT#2,'SYMBOL CODE';\PRINT#2,USING"##",N5
4010 A4=Z1(7)\L4=Z1(6)\R4=Z1(8)\T=A4+L4+R4
4020 A4=A4/T\L4=L4/T\R4=R4/T
4030 PRINT#2,'ANALYSIS # ';\PRINT#2,USING"####",A2;
4040 PRINT#2,USING F$,A4;\PRINT#2,USING F$,L4;
4050 PRINT#2,USING F$,R4
4060 C3=C3+1\C4=C4+1
4070 RETURN
9000 CLOSE#2
9050 END
```

Appendix XV. Listing of GAPLOT

```
100 REM GAPLOT.BAS READS GARNET END MEMBER DATA FILE CREATED BY
110 REM GARN.BAS AND PLOTS PTS. ON TERNARY DIAGRAM  M.FLOHR 6/83
120 C$='####'\C3$='######.##'
130 DIM E(6)
140 A$='@!!'\H@='H'\B$=' '\U$='U'\H3$='H A '
150 OPEN 'LP:' FOR OUTPUT AS FILE#2
160 Q4=0\REM COUNTER FOR PRINTING PLOTTING INFORMATION
170 PRINT'TERNARY PLOTTING OPTIONS: UVAR=1; AND=2; PYR=3'
180 PRINT'SPESS=4; GROS=5; ALM=6; ALM+PYR=7'
200 PRINT'CHOOSE COMPONENT FOR: TOP,LEFT,RIGHT'\INPUTA,L,R
220 PRINT'ENTER NAME OF GARNET END MEMBER DATA FILE AS ';
230 PRINT'SYN:XXXXXX:YYY'\INPUT N$
240 OPEN N$ FOR INPUT AS FILE#1
260 PRINT#2,'DATA FILE ACCESSED: ';N$
270 INPUT#1,B$\PRINT#2,'DATA FILE ACCESSED BY GARN.BAS: ';B$
275 PRINT#2\Z2=0\REM COUNTER FOR RELATIVE PLOTTING
280 IF Q4<>0 GOTO 360
290 PRINT#2,'PLOTTING PARAMETERS ARE:  UVAROVITE=1; ANDRADITE=2; PYROPE=3  '
300 PRINT#2,'SPESSARTINE=4; GROSSULAR=5;  ALMANDINE=6; ALMANDINE+PYROPE=7'
320 PRINT#2
325 PRINT#2,'FOR THIS PLOT:   TOP COMPONENT IS :';A
330 PRINT#2,'                LEFT COMPONENT IS :';L
340 PRINT#2,'               RIGHT COMPONENT IS :';R\PRINT#2
350 Q4=Q4+1
360 PRINT'ENTER CODE FOR PLOTTING SYMBOL: 0,1,2,3,4 OR 5';\INPUT M
370 IF M=0 THEN Q$=' M20 D U'
380 IF M=1 THEN Q$=' M21 D U'
390 IF M=3 THEN Q$=' M23 D U'
400 IF M=2 THEN Q$=' M22 D U'
410 IF M=5 THEN Q$=' M25 D U'
420 IF M=4 THEN Q$=' M24 D U'
430 IF END#1 GOTO800
435 INPUT#1,N
440 FOR J=1 TO 6\INPUT#1,E(J)\NEXTJ
450 IF A=1 THEN A4=E(1)
460 IF A=2 THEN A4=E(2)
470 IF A=3 THEN A4=E(3)
480 IF A=4 THEN A4=E(4)
490 IF A=5 THEN A4=E(5)
500 IF A=6 THEN A4=E(6)
510 IF A=7 THEN A4=E(6)+E(3)
520 IF L=1 THEN L4=E(1)
530 IF L=2 THEN L4=E(2)
540 IF L=3 THEN L4=E(3)
550 IF L=4 THEN L4=E(4)
560 IF L=5 THEN L4=E(5)
570 IF L=6 THEN L4=E(6)
580 IF L=7 THEN L4=E(6)+E(3)
590 IF R=1 THEN R4=E(1)
600 IF R=2 THEN R4=E(2)
610 IF R=3 THEN R4=E(3)
620 IF R=4 THEN R4=E(4)
630 IF R=5 THEN R4=E(5)
640 IF R=6 THEN R4=E(6)
650 IF R=7 THEN R4=E(6)+E(3)
660 T=A4+L4+R4
670 A5=A4/T\L5=L4/T\R5=R4/T
680 COUT(,A$,3,0)\REM PLOT POINT
690 L@=INT(200*(10-L5*10)-.5774*1733.4*A5+200)
700 A4=INT(1733.4*A5+200)\L@$=STR$(L@)\A4$=STR$(A4)
705 IF Z2<>0 GOTO720
710 P4$='H A '+L@$+B$+A4$+Q$\COUT(,P4$,255,0)
715 X4=L@\Y4=A4\Z2=Z2+1\GOTO745
720 X3=L@-X4\Y3=A4-Y4\REM DEFINE RELATIVE COORD. OF NEXT PT.
725 X3$=STR$(X3)\Y3$=STR$(Y3)\X7$='R U '+X3$+S$+Y3$+Q$
730 COUT(,A$,3,0)\COUT(,X7$,255,0)
740 X4=L@\Y4=A4\Z2=Z2+1\GOTO760
745 PRINT#2,'SYMBOL CODE: ';\PRINT#2,USING"##",M
750 PRINT#2,'ANALYSIS    TOP      LEFT     RIGHT'\PRINT#2
760 PRINT#2,USING C$,N;\PRINT#2,'    ';
770 PRINT#2,USING C3$,A5*100;\PRINT#2,USING C3$,L5*100;
780 PRINT#2,USING C3$,R5*100
790 GOTO430
800 CLOSE#1\COUT(,A$,3,0)\COUT(,H3$,10,0)
810 PRINT'DO YOU WANT TO PLOT DATA FROM ANOTHER FILE ON THIS GRAPH (Y OR N)'
820 INPUT R$
830 IF ASC(R$)<>89 GOTO 850
840 PRINT#2\PRINT#2\GOTO220
850 CLOSE#2
1000 END
```

Appendix XVI. Example of format of plotting file

```
SY1:86954.DAT
    1
7.774
0.226
0.065
0.000
0.012
4.880
0.040
0.000
0.003
0.039
0.048
0.000
1.913
0.000
0.034
0.012
0.015
0.062
0.126
0.000
    4
6.946
1.054
0.653
0.000
1.760
2.547
0.039
0.000
0.001
0.134
0.073
0.000
1.793
0.000
0.203
0.032
0.043
0.279
0.009
0.013
    5
6.852
1.148
0.761
0.000
1.690
2.507
0.042
0.000
0.000
0.163
0.017
0.000
```

Appendix XVII. Example of output from PLAMPH



X-Y OPTIONS ARE: (ALVI + FE+3 + 2TI)=1; (CA+NA IN M4)=2; (FE+2 + MN)/(FE+2 + MN + MG)=3; (NA IN M4)=4; (ALIV)=5; (ALVI + FE+3 + TI + CR)=6; (NA IN A + K)=7; (100NA)/(CA+NA)=8; (100AL)/(AL+SI)=9; (ALVI)=10; (TI)=11; (K)=12; (NA IN A)=13; (MN)=14; (MG/MG + FE+2)=15

FOR THIS PLOT: X= 6
Y= 13

DATA FILE ACCESSED: SY1:PA6851.DAT
DATA FILE ACCESSED BY AMPHI.BAS: SY1:S6851.DAT

SYMBOL CODE: 0

ANALYSIS	X	Y
9	0.483	0.197
14	0.373	0.146
15	0.551	0.204
16	0.638	0.232
17	0.412	0.170
19	0.544	0.165
20	0.420	0.192
21	0.391	0.132
25	0.452	0.174
26	0.298	0.111
27	0.507	0.171
28	0.560	0.118
29	0.379	0.142
30	0.492	0.166
49	0.813	0.297
50	0.634	0.212
51	0.555	0.215
52	0.348	0.120

DATA FILE ACCESSED: SY1:PA6954.DAT
DATA FILE ACCESSED BY AMPHI.BAS: SY1:S6954.DAT

SYMBOL CODE: 3

ANALYSIS	X	Y
1	0.108	0.034
4	0.693	0.203
5	0.803	0.215
6	0.730	0.270
7	0.452	0.129
8	0.523	0.156
9	0.590	0.184
10	0.545	0.188
11	0.528	0.150
12	0.492	0.151
13	0.593	0.161
14	0.048	0.024
15	0.130	0.049
16	0.087	0.030
17	0.009	0.014
18	0.570	0.209
19	0.448	0.122
20	0.563	0.198
21	0.621	0.224
22	0.452	0.130

Appendix XVII. Example of input for PLAMPH.

The following are the prompts given by program PLAMPH and the responses (indicated by a box ☐), which were given to generate the plot shown in this appendix.

ENTER NAME OF DATA FILE AS SYn:XXXXXX.YYY ? [SY1:PA6951.DAT]

ENTER CODE FOR PLOTTING SYMBOL: 0,1,2,3,4 OR 5 ? [0]

X-Y PLOTTING OPTIONS ARE: (ALIV + FE+3 + 2TI)=1
(CA+NA IN M4)=2; (FE+2 + MN)/(FE+2 + MG + MN)=3
(NA IN M4)=4; (ALIV)=5; (ALVI + FE+3 + TI + CR)=6
(NA IN A + K)=7; (100NA)/(CA+NA)=8; (100AL)/(SI+AL)=9
(ALIV)=10; (TI)=11; (K)=12; (NA IN A)=13; MN=14; (MG/MG + FE+2)=15

CHOOSE X,Y ? [6,13]

INPUT X MIN,MAX ? [0,1]

INPUT Y MIN,MAX ? [0,.3]

DO YOU WANT TO PLOT DATA FROM ANOTHER FILE ON THIS GRAPH?
(Y OR N) ? [Y]

ENTER NAME OF DATA FILE AS SYN:XXXXXX.YYY ? [SY1:PA6954.DAT]

ENTER CODE FOR PLOTTING SYMBOL: 0,1,2,3,4 OR 5 ? [3]

DO YOU WANT TO PLOT DATA FROM ANOTHER FILE ON THIS GRAPH?
(Y OR N) ? [N]

Appendix XVIII. Listing of PLDATA

```
100 PRINT'PLDATA.BAS FOR TERNARY PLOTTING'
105 REM READS IN DATA,SORTS ANALYSES,PLOTS POINTS!M.FLOHR 6/83
110 A$='@!:'\D$='D'\S$=' '\H3$='H A '
120 U$='U'\C$=','\H$='H'\M$=' '\REM IF NOT SORTING M$ IS LEFT BLANK
125 F$='####.###'\F2$="'RRRRRRR"
130 REM B=MOLECULAR WT. FOR CALCULATING MOLEC.PROPORTIONS
140 DIM B(18),Z(18),Z1(18),T$(48)
150 FOR I=1TO18\READ B(I)
160 DATA 60.09,79.9,101.96,159.69,151.99,71.85,81.37,70.94,74.69,40.31,
170 DATA 56.08,103.62,153.34,61.98,94.2,141.94,19,35.45
180 NEXT I
210 OPEN'LP:' FOR OUTPUT AS FILE#2
220 PRINT#2,DAT$,CLK$\PRINT#2
230 PRINT'CHOOSE PLOT: AFM=1;ACF=2;(OR+CN)-AB-AN=3'
240 PRINT'WO-EN-FS=4; AKF=5 '\INPUTM1
250 PRINT'ENTER NAME OF DATA FILE AS SYN:XXXXXX.YYY '\INPUTN$
260 PRINT'KEYPAD EDITOR ERROR CHECKING? (Y OR N)'
280 INPUT R$\N3=0\REM COUNTER FOR PRINTING FILE NAME
285 PRINT'ARE YOU SORTING?(Y OR N) ';\INPUTN8$
287 IF ASC(N8$)=89 GOTO295
290 M$=' '\GOTO300
295 PRINT'ENTER 2 CHARACTER SORT STRING: ';\INPUTM$
300 PRINT'CHOOSE PLOTTING SYMBOL 0,1,2,3,4 OR 5 ';\INPUTN5
310 IF N3<>0 GOTO360
320 PRINT#2,'DATA FILE ACCESSED: ';N$\PRINT#2
340 N3=N3+1\C4=0
350 OPEN N$ FOR INPUT AS FILE#1
360 C3=0\Z2=0\REM C3 COUNTER FOR PRINTING HEADINGS!Z2 FOR RELATIVE PLOTTING
363 INPUT#1,A8
365 FOR J=1TO18\INPUT#1,A5$\NEXTJ
370 IF END#1 GOTO 790
375 INPUT#1,A2,T$\IF ASC(R$)=89 THEN PRINT A2,T$
385 INPUT#1,T5$,C8\IF ASC(N8$)<>89 GOTO420
390 M5$=SEG$(T$,1,2)
400 IF M5$=M$ GOTO420
405 FOR J=1TO18\INPUT#1,H2,H3
410 NEXTJ\GOTO 370
420 FOR J=1TO18\INPUT#1,Z(J),Z1(J)\NEXTJ
430 IF N5=0 THEN P3$=' M20 D U'
440 IF N5=1 THEN P3$=' M21 D U'
450 IF N5=2 THEN P3$=' M22 D U'
460 IF N5=3 THEN P3$=' M23 D U'
470 IF N5=4 THEN P3$=' M24 D U'
480 IF N5=5 THEN P3$=' M25 D U'
500 IF M1=1 THEN GOSUB 860
510 IF M1=1 GOTO 650
520 IF M1=2 THEN GOSUB 2000
530 IF M1=2 GOTO 650
540 IF M1=3 THEN GOSUB 3050
550 IF M1=3 GOTO 650
560 IF M1=4 THEN GOSUB 3050
570 IF M1=4 GOTO650
580 IF M1=5 THEN GOSUB 2000
650 REM PLOT POINT
660 L8=INT(200*(10-L4*10)-.5774*1733.4*A4+200)
670 L9=INT(1733.4*A4+200)\L9$=STR$(L9)\L8$=STR$(L8)
680 IF Z2<>0 GOTO720
690 P2$='H A '+L8$+S$+L9$+P3$\COUT(,A$,3,0)\COUT(,P2$,255,0)
710 X4=L8\Y4=L9\Z2=Z2+1\GOTO370
720 X3=L8-X4\Y3=L9-Y4\REM DEFINE RELATIVE COORD. OF NEXT PT.
730 X3$=STR$(X3)\Y3$=STR$(Y3)\X7$='R U '+X3$+S$+Y3$+P3$
735 COUT(,A$,3,0)\COUT(,X7$,255,0)
740 X4=L8\Y4=L9\Z2=Z2+1\GOTO370
790 IF ASC(N8$)<>89 GOTO830
800 PRINT'DO YOU WANT TO PLOT MORE DATA FROM THIS FILE ON THIS'
805 PRINT'GRAPH? (YES=1; NO=2)';\INPUTN4
810 IF N4=2 GOTO830
815 RESTORE#1
820 COUT(,A$,3,0)\COUT(,H3$,10,0)\GOTO295
830 CLOSE#1\COUT(,A$,3,0)\COUT(,H3$,10,0)
835 PRINT'DO YOU WANT TO PLOT DATA FROM ANOTHER FILE ON THIS'
840 PRINT'GRAPH? YES=1; NO=2)';\INPUTN7
845 PRINT#2\PRINT#2
848 IF N7=1 GOTO 250
850 GOTO 9000
860 REM SUBROUTINE FOR AFM
870 IF C4<>0 GOTO 1010
880 PRINT#2,'            ';
890 PRINT#2,USING F2$,'A';
900 PRINT#2,USING F2$,'F';
```

```
1000 PRINT#2,USING"'RRRRRRR",'M'
1010 IF C3=0 THEN PRINT#2,M$,N5
1020 A4=((Z(3)/B(3))+(Z(4)/B(4))-(Z(14)/B(14))-(Z(15)/B(15)))
1030 L4=((Z(6)/B(6))+(Z(8)/B(8)))
1040 R4=Z(10)/B(10)\T=R4+L4+A4
1050 A4=A4/T\L4=L4/T\R4=R4/T
1055 PRINT#2,'ANALYSIS # '\PRINT#2,USING"####",A2;
1060 PRINT#2,USING F$,A4;
1070 PRINT#2,USING F$,L4;\PRINT#2,USING F$,R4
1080 C3=C3+1\C4=C4+1
1090 RETURN
2000 REM SUBROUTINE FOR ACFm AND AKFm
2010 IF C4<>0 GOTO 2050
2015 PRINT#2\PRINT#2,'            '\IF M1=5 GOTO2030
2020 PRINT#2,USING F2$,'A';\PRINT#2,USING F2$,'C';
2025 PRINT#2,USING"'RRRRRR",'F'\GOTO2050
2030 PRINT#2,USING F2$,'A';\PRINT#2,USING F2$,'K';
2035 PRINT#2,USING"'RRRRRRR",'Fm'
2050 IF C3=0 THEN PRINT#2,M$,N5
2060 A4=((Z(3)/B(3))+(Z(4)/B(4))-(Z(14)/B(14))-(Z(15)/B(15)))
2070 R4=((Z(6)/B(6))+(Z(10)/B(10))+(Z(8)/B(8)))
2075 IFM1<>2 GOTO2085
2080 L4=Z(11)/B(11)\T=A4+L4+R4\GOTO2090
2085 L4=Z(15)/B(15)\T=A4+L4+R4
2090 A4=A4/T\L4=L4/T\R4=R4/T
3000 PRINT#2,'ANALYSIS # '\PRINT#2,USING"####",A2;
3010 PRINT#2,USING F$,A4;\PRINT#2,USING F$,L4;
3020 PRINT#2,USING F$,R4
3030 C3=C3+1\C4=C4+1
3040 RETURN
3050 REM SUBROUTINE FOR AN-AB-OR+CN;WO-EN-FS
3060 IF M1=4 GOTO 4010\IFC4<>0 GOTO 4050
3070 PRINT#2\PRINT#2,'                  '
3080 PRINT#2,USING"'RRRR",'OR+CN';
3090 PRINT#2,USING F2$,'AB';\PRINT#2,USING F2$,'AN'
4000 GOTO 4050
4010 IF C4<>0 GOTO 4050
4020 PRINT#2\PRINT#2,'                  '
4030 PRINT#2,USING"'RRRRRR",'WO';\PRINT#2,USING F2$,'EN';
4040 PRINT#2,USING F2$,'FS'
4050 IF C3=0 THEN PRINT#2,M$,N5
5020 IF M1=4 GOTO 5060
5030 T=Z1(13)+Z1(14)+Z1(15)+Z1(11)
5040 A4=(Z1(13)+Z1(15))/T\R4=Z1(11)/T
5050 L4=Z1(14)/T\GOTO 5080
5060 T=Z1(11)+Z1(10)+Z1(6)
5070 A4=Z1(11)/T\R4=Z1(6)/T\L4=Z1(10)/T
5080 PRINT#2,'ANALYSIS # '\PRINT#2,USING"####",A2;
5090 PRINT#2,USING F$,A4;\PRINT#2,USING F$,L4;
6000 PRINT#2,USING F$,R4
6010 C3=C3+1\C4=C4+1
6020 RETURN
9000 CLOSE#2
9050 END
```

Appendix XVIII. Example of output from PLDATA



26-SEP-83 13:55:46

DATA FILE ACCESSED: SY1:S6752.DAT

		A	F	M
AM	0			
ANALYSIS #	1	0.009	0.337	0.654
ANALYSIS #	2	0.020	0.350	0.630
ANALYSIS #	3	0.027	0.343	0.630
ANALYSIS #	4	0.021	0.357	0.621
ANALYSIS #	5	0.034	0.354	0.612
ANALYSIS #	6	0.010	0.341	0.649
ANALYSIS #	7	0.028	0.349	0.623
BI	1			
ANALYSIS #	19	0.111	0.248	0.641
ANALYSIS #	20	0.098	0.239	0.663
ANALYSIS #	21	0.102	0.232	0.666
ANALYSIS #	22	0.116	0.255	0.629
ANALYSIS #	23	0.126	0.264	0.610
ANALYSIS #	24	0.108	0.242	0.651
CD	3			
ANALYSIS #	15	0.490	0.096	0.414
ANALYSIS #	16	0.497	0.090	0.413
ANALYSIS #	17	0.494	0.093	0.413
ANALYSIS #	26	0.492	0.094	0.413

Appendix XIX. Listing of TETPT

```
100 PRINT'TETPT.BAS FOR TETRAHEDRAL POINT PLOTTING'
105 REM READS IN DATA,CHOOSES ANALYSES,PLOTS POINTS;M.FLOHR 7/83
110 REM USES F.SPEAR ALGORITHM AM.MIN.65,1980
120 A$='@';'\D$='D'\B$=' '\M$=' '
130 U$='U'\C$=','\H$='H'
130 DIM B(18),Z(18),Z1(18),T$(48)
140 FOR I=1TO18\READ B(I)
150 DATA 60.09,79.9,101.96,159.69,151.99,71.85,81.37,70.94,74.69,40.31,
160 DATA 56.08,103.62,153.34,61.98,94.2,141.94,19,35.45
165 NEXT I
170 PRINT'ENTER ANGLE ALPHA IN DEGREES';\INPUTA3
175 PRINT'ENTER ANGLE THETA IN DEGREES';\INPUTT3
180 PRINT'ENTER E (PERSPECTIVE)';\INPUTE
185 PRINT'ENTER S (SCALING FACTOR)';\INPUTS
190 A3=A3/57.29578\T3=T3/57.29578\REM CONVERT DEGREES TO RADIANS
195 REM ALPHA-ROTATATION ANGLE AROUND Z-AXIS;THETA AROUND X-AXIS
200 N4=0\REM C4 COUNTER TO PRINT HEADINGS;C3 TO PRINT SORT STRINGS
205 REM + SYMBOL CODE; N4 IF SORTING;N3 TO OPEN FILE;C5 FOR RELATIVE PLOTTING
210 OPEN'LP:' FOR OUTPUT AS FILE#2
220 PRINT#2,DAT$,CLK$\PRINT#2
230 PRINT'CHOOSE PLOT: AFMK=1;AFMC=2;AF=KC=3'\INPUTM1
250 PRINT'ENTER NAME OF DATA FILE AS SYN:XXXXXX.YYY '\INPUTN$
260 PRINT'KEYPAD EDITOR ERROR CHECKING? (Y OR N)'
265 INPUT R$\N3=0
270 PRINT'ARE YOU SORTING?(Y OR N)';\INPUTN8$
280 IF ASC(N8$)<>89 GOTO300
290 PRINT'ENTER 2 CHARACTER SORT STRING: ';\INPUTM$
295 C5=0
300 PRINT'CHOOSE PLOTTING SYMBOL 0,1,2,3,4 OR 5 ';\INPUTN5
310 IF N3<>0 GOTO360
320 PRINT#2,'DATA FILE ACCESSED: ';N$\PRINT#2
340 N3=N3+1\C4=0
350 OPEN N$ FOR INPUT AS FILE#1
360 C3=0
363 INPUT#1,A8
365 FOR J=1TO18\INPUT#1,A5$\NEXTJ
370 IF END#1 GOTO 2400
375 INPUT#1,A2,T$\IF ASC(R$)=89 THEN PRINT A2,T$
385 INPUT#1,T5$,C8\IF ASC(N8$)<>89 GOTO420
390 M5$=SEG$(T$,1,2)
400 IF M5$=M$ GOTO420
405 FOR J=1TO18\INPUT#1,H2,H3
410 NEXTJ\GOTO 370
420 FOR J=1TO18\INPUT#1,Z(J),Z1(J)\NEXTJ
500 A=(Z(3)/B(3))+(Z(4)/B(4))-(Z(15)/B(15))-(Z(14)/B(14))
510 F=(Z(6)/B(6))+(Z(8)/B(8))\M=Z(10)/B(10)
520 F2=F+M\K=Z(15)/B(15)\C=Z(11)/B(11)
530 IF M1=1 GOTO600\REM AFMK
540 IF M1=2 GOTO660\REM AFMC
550 IF M1=3 GOTO710\REM AF=KC
600 REM CALCULATE (X,Y,Z) OF PT.
610 T=A+F+M+K\A=A/T\F=F/T
620 M=M/T\K=K/T
630 X=((F-.25)+(.5*(M-.25))+(.5*(A-.25)))/.8165
640 Y=((M-.25)+((A-.25)/3))/.9428\Z=A-.25
650 GOTO750
660 T=A+F+M+C\A=A/T\F=F/T
670 M=M/T\C=C/T
680 X=((F-.25)+(.5*(M-.25))+(.5*(A-.25)))/.8165
690 Y=((M-.25)+((A .25)/3))/.9428\Z=A-.25
700 GOTO 750
710 T=A+F2+K+C\A=A/T\F2=F2/T
720 K=K/T\C=C/T
730 X=((F2-.25)+(.5*(K-.25))+(.5*(A-.25)))/.8165
740 Y=((K-.25)+((A-.25)/3))/.9428\Z=A-.25
750 R1=COS(A3)\R2=COS(T3)\R3=SIN(A3)
760 R4=SIN(T3)\REM CALCULATE Z'(HERE V) FOR POINT (X,Y,Z)
770 V=(-X*R3*R4)-(Y*R1*R4)+(Z*R2)
780 REM CALCULATE X,Y FROM PERSPECTIVE E
790 Q1=1200\Q2=800\REM DEFINE CENTER
800 X2=INT(((X*R1)-(Y*R3))*(E/(E-V))*S)+Q1
810 Y2=INT(((X*R3*R2)+(Y*R1*R2)+(Z*R4))*(E/(E-V))*S)+Q2
820 X2$=STR$(X2)\Y2$=STR$(Y2)
830 REM PRINT COORDINATES AT LINE PRINTER
840 IF C4<>0 GOTO880
850 IF M1=1 THEN PRINT#2,'ANALYSIS#     A       F       M       K'
860 IF M1=2 THEN PRINT#2,'ANALYSIS#     A       F       M       C'
870 IF M1=3 THEN PRINT#2,'ANALYSIS#     A       F=      K       C'
880 IF ASC(N8$)<>89 GOTO 900
```

```
890 IF C3=0 THEN PRINT@2,M@;'    ';N5
900 PRINT@2,USING"@@@@",A2;\PRINT@2,'    ';
910 F@='@@@@.@@@'
920 IF M1=1 GOTO970
930 IF M1=2 GOTO1000
940 IF M1=3 THEN PRINT@2,USING F@,A;
950 PRINT@2,USING F@,F2;\PRINT@2,USING F@,K;
960 PRINT@2,USING F@,C\GOTO1020
970 PRINT@2,USING F@,A;\PRINT@2,USING F@,F;
980 PRINT@2,USING F@,M;\PRINT@2,USING F@,K\GOTO1020
1000 PRINT@2,USING F@,A;\PRINT@2,USING F@,F;
1010 PRINT@2,USING F@,M;\PRINT@2,USING F@,C
1020 C3=C3+1\C4=C4+1\REM PLOT POINT(X2,Y2)
1040 IF N5=0 THEN P3@=' M20'
1050 IF N5=1 THEN P3@=' M21'
1060 IF N5=2 THEN P3@=' M22'
1070 IF N5=3 THEN P3@=' M23'
1080 IF N5=4 THEN P3@=' M24'
1090 IF N5=5 THEN P3@=' M25'
2000 IF C5<>0 GOTO 2040
2010 P2@='M A '+X2@+S@+Y2@+P3@+S@+U@
2020 COUT(,A@,3,0)\COUT(,P2@,255,0)
2030 X4=X2\Y4=Y2\C5=C5+1\GOTO370
2040 X3=X2-X4\Y3=Y2-Y4\REM DEFINE RELATIVE COORD. OF NEXT POINT
2050 X3@=STR@(X3)\Y3@=STR@(Y3)
2060 P2@='R U '+X3@+S@+Y3@+P3@+S@+U@
2070 COUT(,A@,3,0)\COUT(,P2@,255,0)
2080 X4=X2\Y4=Y2\C5=C5+1\GOTO370
2100 IF ASC(N8@)<>89 THEN GOTO 3000
2110 PRINT'DO YOU WANT TO PLOT MORE DATA FROM THIS FILE ON THIS'
2120 PRINT'GRAPH? (YES=1; NO=2)';\INPUTN4
2130 IF N4=2 GOTO3000
2140 RESTORE@1\GOTO 290
3000 CLOSE@1
3010 PRINT'DO YOU WANT TO PLOT DATA FROM ANOTHER FILE ON THIS'
3020 PRINT'GRAPH? YES=1; NO=2)';\INPUTN7
3030 PRINT@2\PRINT@2
3040 IF N7=1 GOTO 250
3060 P7@='M A '
9000 COUT(,A@,3,0)\COUT(,P7@,10,0)
9010 CLOSE@2
9050 END
```

Appendix XIX Example of output from TETPT



26-SEP-83 15:02:22

DATA FILE ACCESSED: SY1:S6851.DAT

ANALYSIS#	A	F	M	C
AM 0				
9	0.072	0.262	0.402	0.265
14	0.055	0.256	0.419	0.270
15	0.081	0.278	0.376	0.265
16	0.094	0.285	0.355	0.266
17	0.060	0.254	0.420	0.266
19	0.075	0.262	0.403	0.260
20	0.067	0.266	0.401	0.266
21	0.064	0.255	0.425	0.256
25	0.065	0.259	0.405	0.271
26	0.044	0.246	0.444	0.266
27	0.070	0.268	0.398	0.265
28	0.073	0.263	0.401	0.263
29	0.053	0.241	0.439	0.267
30	0.067	0.239	0.426	0.267
49	0.117	0.279	0.342	0.262
50	0.093	0.264	0.376	0.267
51	0.085	0.262	0.389	0.264
52	0.053	0.238	0.447	0.262
BI 1				
2	0.094	0.388	0.517	0.001
3	0.089	0.398	0.513	0.000
4	0.092	0.396	0.511	0.001
5	0.097	0.400	0.502	0.001
31	0.087	0.414	0.498	0.002
32	0.091	0.406	0.499	0.004
33	0.091	0.406	0.501	0.003
48	0.095	0.408	0.494	0.003
PX 4				
10	0.002	0.177	0.340	0.481
11	0.000	0.162	0.354	0.484
12	0.008	0.189	0.326	0.477
13	0.004	0.180	0.335	0.482
PL 2				
34	0.501	0.000	0.000	0.499
35	0.503	0.000	0.000	0.497
36	0.500	0.000	0.000	0.500
37	0.505	0.000	0.000	0.495
38	0.497	0.000	0.000	0.503
39	0.504	0.000	0.000	0.496
40	0.508	0.000	0.000	0.492
41	0.510	0.000	0.000	0.490
42	0.504	0.000	0.000	0.496
43	0.498	0.000	0.000	0.502
44	0.503	0.000	0.000	0.497

Figure 1. Flow chart of programs and their options.



Table 1. Order of data in plotting files.

PLAMPH		MICAPL		CHPLOT		PLCORD		GRNPLT	GAPLOT
(T site)	Si	(T site)	Si	(T site)	Si	(T or [X])	Si	(oxide, cation pairs)	(garnet end members)
	Al IV		Al IV		Al IV		Al IV	Si	Uvar
(M_1-M_3)	Al VI	(M sites)	Al VI	(Oct.)	Al VI	([Y]4)	Al VI	Al	And
	Fe^{+3}		Fe^{+3}		Fe^{+3}		Ti	Ti	Pyr
	Fe^{+2}		Fe^{+2}		Fe^{+2}		Fe^{+3}	Cr	Spes
	Mg		Mg		Mg	([XY]6)	Mg	Fe^{+3}	Gros
	Ti		Ti		Mn		Fe^{+2}	Fe^{+2}	Alm
	Mn		Mn		Cr		Mn	Mn	
	Cr		Cr		Ca		Ca	Mg	
(M_4)	Mn	(A site)	Ca		Na		Na	Ca	
	Fe^{+2}		Na		K		K	Na	
	Mg		K						
	Ca		Ba						
	Na	(anions)	F						
(A)	Na		Cl						
	K								
	Ca								
	Sum of A site								
(anions)	F								
	Cl								